


Asymptote: the Vector Graphics Language

For version 1.18



This file documents **Asymptote**, version 1.18.

<http://asymptote.sourceforge.net>

Copyright © 2004-6 Andy Hammerlindl, John Bowman, and Tom Prince.

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU General Public License. On Debian systems, the GPL can be found at `/usr/share/common-licenses/GPL`.

Table of Contents

1	Description	1
2	Installation	2
2.1	UNIX binary distributions	2
2.2	MacOS X binary distributions	2
2.3	Microsoft Windows	3
2.4	Configuring	3
2.5	Search paths	4
2.6	Compiling from UNIX source	5
2.7	Editing modes	5
2.8	Subversion	6
2.9	Uninstall	6
3	Tutorial	7
4	Drawing commands	13
4.1	draw	13
4.2	fill	15
4.3	clip	16
4.4	label	16
5	Programming	20
5.1	Data types	20
5.2	Guides and paths	25
5.3	Pens	30
5.4	Transforms	37
5.5	Frames and pictures	38
5.6	Files	43
5.7	Structures	45
5.8	Operators	47
5.8.1	Arithmetic & logical operators	47
5.8.2	Self & prefix operators	48
5.8.3	User-defined operators	48
5.9	Implicit scaling	49
5.10	Functions	49
5.10.1	Default arguments	51
5.10.2	Named arguments	51
5.10.3	Rest arguments	52
5.10.4	Mathematical functions	54
5.11	Arrays	55
5.12	Casts	60
5.13	Import	61
5.14	Static	63

6	LaTeX usage	65
7	Base modules	70
7.1	plain	70
7.2	simplex	70
7.3	math	70
7.4	geometry	70
7.5	stats	71
7.6	patterns	71
7.7	tree	71
7.8	drawtree	71
7.9	feynman	71
7.10	roundedpath	71
7.11	slide	71
7.12	MetaPost	73
7.13	unicode	73
7.14	latin1	73
7.15	babel	73
7.16	embed	73
7.17	annotate	74
7.18	CAD	74
7.19	graph	74
7.20	palette	97
7.21	three	102
7.22	light	107
7.23	graph3	107
7.24	solids	112
7.25	featpost3D	113
7.26	flowchart	113
7.27	contour	115
7.28	slopefield	120
8	Options	122
9	Interactive mode	125
10	Graphical User Interface	126
11	PostScript to Asymptote	127
12	Help	128
13	Debugger	129
14	Acknowledgments	130

Index	131
-------------	-----

1 Description

Asymptote is a powerful descriptive vector graphics language that provides a mathematical coordinate-based framework for technical drawings. Labels and equations are typeset with **LaTeX**, for overall document consistency, yielding the same high-quality level of typesetting that **LaTeX** provides for scientific text. By default it produces **PostScript** output, but it can also generate any format that the **ImageMagick** package can produce.

A major advantage of **Asymptote** over other graphics packages is that it is a high-level programming language, as opposed to just a graphics program: it can therefore exploit the best features of the script (command-driven) and graphical-user-interface (GUI) methods for producing figures. The rudimentary GUI **xasy** included with the package allows one to move script-generated objects around. To make **Asymptote** accessible to the average user, this GUI is currently being developed into a full-fledged interface that can generate objects directly. However, the script portion of the language is now ready for general use by users who are willing to learn a few simple **Asymptote** graphics commands (see [Chapter 4 \[Drawing commands\]](#), page 13).

Asymptote is mathematically oriented (e.g. one can use complex multiplication to rotate a vector) and uses **LaTeX** to do the typesetting of labels. This is an important feature for scientific applications. It was inspired by an earlier drawing program (with a weaker syntax & capabilities) called **MetaPost**.

Many of the features of **Asymptote** are written in the **Asymptote** language itself. While the stock version of **Asymptote** is designed for mathematics typesetting needs, one can write **Asymptote** modules that tailor it to specific applications. A scientific graphing module has already been written (see [Section 7.19 \[graph\]](#), page 74). Examples of **Asymptote** code and output, including animations, are available at

<http://asymptote.sourceforge.net/gallery/>.

The **Asymptote** vector graphics language provides:

- a natural coordinate-based framework for technical drawings, inspired by **MetaPost**, with a much cleaner, powerful C++-like programming syntax;
- **LaTeX** typesetting of labels, for overall document consistency;
- compilation of figures into virtual machine code for speed, without sacrificing portability;
- the power of a script-based language coupled to the convenience of a GUI;
- customization using its own C++-like graphics programming language;
- sensible defaults for graphical features, with the ability to override;
- a high-level mathematically oriented interface to the **PostScript** language for vector graphics, including affine transforms and complex variables;
- functions that can create new (anonymous) functions;
- deferred drawing that uses the simplex method to solve overall size constraint issues between fixed-sized objects (labels and arrowheads) and objects that should scale with figure size;
- a standard for typesetting mathematical figures, just as **T_EX/LaTeX** is the de-facto standard for typesetting equations.

2 Installation

After following the instructions for your specific distribution, please see also [Section 2.4 \[Configuring\]](#), page 3.

We recommend subscribing to new release announcements at

<http://freshmeat.net/subscribe/50750>

Users may also wish to monitor the **Asymptote** forum:

http://sourceforge.net/forum/monitor.php?forum_id=409349

and provide guidance to others by rating the **Asymptote** project:

<http://freshmeat.net/projects/asy>

2.1 UNIX binary distributions

We now release both RPM and **tar.gz** binary distributions of **Asymptote**. Here are the commands that the root user can use to install the Linux i386 binary distribution of version x.xx of **Asymptote** in `/usr/local`. The executable file will be `/usr/local/bin/asy` (the optional **texhash** command installs LaTeX style files):

```
tar -C / -zxf asymptote-x.xx.i386.tar.gz
texhash
```

Example code will be installed by default in `/usr/share/doc/asymptote/examples`.

Alternatively, Debian users can install Hubert Chan's **Asymptote** prebuilt binary from one of the following sites:

<http://www.uhoreg.ca/programming/debian/pool/main/a/asymptote>

<http://packages.debian.org/asymptote>

Fedora Core 4 (and later) users can easily install **Asymptote** from Fedora Extras mirrors:

```
yum install asymptote
```

2.2 MacOS X binary distributions

MacOS users can either compile the UNIX source code (see [Section 2.6 \[Compiling from UNIX source\]](#), page 5) or install the contributed **Asymptote** binary available at

http://www.hmug.org/pub/MacOS_X/BSD/Applications/Publishing/asymptote/

Because these these preconfigured binary distributions have strict architecture and library dependencies that many installations do not satisfy, we recommend installing **Asymptote** directly from the official source:

http://sourceforge.net/project/showfiles.php?group_id=120000

Note that many MacOS X (and FreeBSD) systems inexplicably ship with an extremely old GNU **readline** version (4.1, dated 21 March 2000). For full interactive functionality, **readline** version 4.2 or later (16 April 2001) is required.

2.3 Microsoft Windows

Users of the Microsoft Windows operating system can install the self-extracting **Asymptote** executable `asymptote-install-x.xx.exe`. The recommended installation directory is `c:\Program Files\Asymptote`. If you wish to install **Asymptote** in another directory, say `d:\Asymptote`, you can tell **Asymptote** where to find `plain.asy` and its other base modules by creating a file `config.asy` in a new folder `%USERPROFILE%\`.`asy` (by default this means `C:\Documents and Settings\%USERNAME%\`.`asy`, where `%USERNAME%` denotes your user name) containing the lines

```
import settings;
dir="d:\Asymptote";
```

A working TeX implementation (such as the one available at <http://www.miktex.org>) will be required to typeset labels. The Python interpreter from <http://www.python.org> is only required if you wish to try out the graphical user interface (see [Chapter 10 \[GUI\]](#), [page 126](#)).

Example code will be installed by default in `c:\Program Files\Asymptote\examples`.

2.4 Configuring

In interactive mode, or when given the `-V` option (the default when running **Asymptote** on a single file under MSDOS), **Asymptote** will automatically invoke the PostScript viewer `gv` (under UNIX) or `gsview` (under MSDOS; available from <http://www.cs.wisc.edu/~ghost/gsview/>) to display graphical output. These defaults may be overridden with the configuration variable `psviewer`.

Configuration variables are most easily set as **Asymptote** variables in the configuration file (by default, `.asy/config.asy` in the user's home directory or `%USERPROFILE%\`.`asy\config.asy` under MSDOS); see [\[configuration file\]](#), [page 123](#). Here are the default values of several important configuration variables.

Under UNIX:

```
import settings;
psviewer="gv";
pdfviewer="xpdf";
gs="gs";
python="";
```

Under MSDOS:

```
import settings;
psviewer="c:\Program Files\Ghostgum\gsview\gsview32.exe";
pdfviewer="c:\Program Files\Adobe\Acrobat 7.0\Reader\AcroRd32.exe";
gs="c:\Program Files\gs\gs8.54\bin\gswin32c.exe";
python="c:\Python24\python.exe";
```

For PDF format output, the `gs` setting specifies the location of the PostScript-to-PDF processor `gs` and `pdfviewer` specifies the location of the PDF viewer. The graphical user interface may also require setting the variable `python` if Python is installed in a nonstandard location.

The configuration variable `dir` can be used to adjust the search path (see [Section 2.5 \[Search paths\]](#), [page 4](#)).

By default, **Asymptote** attempts to center the figure on the page, assuming that the paper type is **letter**. The default paper type may be changed to **a4** with the configuration variable **papertype**. Alignment to other paper sizes can be obtained by setting the configuration variables **paperwidth** and **paperheight**.

The following configuration variables normally do not require adjustment:

```
texpath
dvips
convert
display
animate
xasy
```

Configuration variables may also be set or overwritten with a command line option:

```
asy -psviewer=gsview -V venn
```

Alternatively, system environment versions of these configuration variables may be set in the conventional way. The corresponding environment variable name is obtained by converting the configuration variable name to upper case and prepending **ASYMPTOTE_**: For example, to set the environment variable

```
ASYMPTOTE_PSVIEWER="c:\Program Files\Ghostgum\gsview\gsview32.exe";
```

under Microsoft Windows XP:

1. Click on the **Start** button;
2. Right-click on **My Computer**;
3. Choose **Properties** from the popup menu;
4. Click the **Advanced** tab;
5. Click the **Environment Variables** button.

The **PostScript** viewer should be capable of automatically redrawing whenever the output file is updated. The default UNIX **PostScript** viewer **gv** supports this (via a **SIGHUP** signal). Upgrading to **gv-3.6.2** (from <http://ftp.gnu.org/gnu/gv/>) is required for interactive mode to work properly.

Users of **ggv** will need to enable **Watch file** under **Edit/Postscript Viewer Preferences**

Users of **gsview** will need to enable **Options/Auto Redisplay** (however, under MSDOS it is still necessary to click on the **gsview** window; under UNIX one must manually redisplay by pressing the **r** key).

2.5 Search paths

In looking for **Asymptote** system files, **asy** will search the following paths, in the order listed:

1. The current directory;
2. The directory **.asy** in the user's home directory (**%USERPROFILE%**.asy under MSDOS);
3. A list of one or more directories specified by the configuration variable **dir** (separated by **:** under UNIX and **;** under MSDOS);
4. The **Asymptote** system directory (by default, **/usr/share/asymptote** under UNIX and **c:\Program Files\Asymptote** under MSDOS).

2.6 Compiling from UNIX source

To compile and install a UNIX executable from a source release `x.xx`, first execute the commands:

```
tar -zxf asymptote-x.xx.tar.gz
cd asymptote-x.xx
```

Then put http://www.hpl.hp.com/personal/Hans_Boehm/gc/gc_source/gc6.8.tar.gz in the current directory and

```
./configure
make all
make install
```

If you get errors from a broken `pdftex` installation, simply put

<http://asymptote.sourceforge.net/asymptote.pdf>

in the directory `doc` and repeat the command `make all`.

For a (default) system-wide installation, the last command should be done as root. To install without root privileges:

```
./configure --prefix=$HOME/asymptote
```

The above steps will compile an optimized single-threaded static version of the Boehm garbage collector (http://www.hpl.hp.com/personal/Hans_Boehm/gc/). Alternatively, one can request use of a (presumably multithreaded and therefore slower) system version of the Boehm garbage collector by configuring instead with `./configure --enable-gc=system` (provided of course that your system has the `libgc` and `libgccpp` libraries). One can disable use of the garbage collector by configuring with `./configure --disable-gc`. For a list of other configuration options, say `./configure --help`. For example, one can tell configure to look for header files and libraries in nonstandard locations:

```
./configure CFLAGS=-I/opt/usr/include LDFLAGS=-L/opt/usr/lib
```

If you are compiling `Asymptote` with `gcc`, you will need a relatively recent version (e.g. 3.2 or later). For full interactive functionality, you will need version 4.2 or later of the GNU `readline` library.

The `FFTW` library is only required if you want `Asymptote` to be able to take Fourier transforms of data (say, to compute an audio power spectrum). The `GSL` library is only required if you require the special functions that it supports.

If you don't want to install `Asymptote` system wide, just make sure the compiled binary `asy` and GUI script `xasy` are in your path and set the configuration variable `dir` to point to the directory `base` (in the top level directory of the `Asymptote` source code).

2.7 Editing modes

Users of `emacs` can edit `Asymptote` code with the mode `asy-mode`, which is installed and enabled by default in the Debian package.

Particularly useful key bindings in this mode are `C-c C-c`, which compiles and displays the current buffer, and the key binding `C-c ?`, which shows the available function prototypes for the command at the cursor. For full functionality you should also install the Apache Software Foundation package `two-mode-mode`: <http://www.dedasy.com/freesoftware/files/two-mode-mode.el>

Fans of vim can customize vim for Asymptote with

```
cp @value{Datadir}/doc/asymptote/examples/asy.vim.gz ~/.vim/syntax/asy.vim.gz
gunzip ~/.vim/syntax/asy.vim.gz
```

and add the following to their ~/.vimrc file:

```
augroup filetypedetect
au BufNewFile,BufRead *.asy      setf asy
augroup END
filetype plugin on
```

If any of these directories or files don't exist, just create them. To set vim up to run the current asymptote script using :make just add to ~/.vim/ftplugin/asy.vim:

```
setlocal makeprg=asy\ %
setlocal errorformat=%f:\ %l.%c:\ %m
```

2.8 Subversion

The following commands are needed to install the latest development version of Asymptote using Subversion:

```
svn co https://svn.sourceforge.net/svnroot/asymptote/trunk/asymptote
```

```
cd asymptote
./autogen.sh
wget http://www.hpl.hp.com/personal/Hans_Boehm/gc/gc_source/gc6.7.tar.gz
./configure
make all
make install
```

To compile without optimization, use the command `make CFLAGS=-g`.

2.9 Uninstall

To uninstall an Linux i386 binary distribution, use the commands

```
tar -zxvf asymptote-x.xx.i386.tar.gz | xargs --replace=% rm %/
texhash
```


To uninstall all Asymptote files installed from a source distribution, use the command

```
make uninstall
```

3 Tutorial

To draw a line from coordinate (0,0) to coordinate (100,100) using **Asymptote's** interactive mode, type at the command prompt:

```
asy
draw((0,0)--(100,100));
```



The units here are **PostScript** "big points" (1 bp = 1/72 inch); -- means join with a linear segment. In **Asymptote** coordinates like (0,0) and (1000,100) are called *pairs*.

At this point you can type in further draw commands, which will be added to the displayed figure, or type **quit** to exit interactive mode. You can use the arrow keys in interactive mode to edit previous lines (assuming that you have support for the GNU **readline** library enabled). The tab key will automatically complete unambiguous words; otherwise, hitting tab again will show the possible choices. Further commands specific to interactive mode are described in [Chapter 9 \[Interactive mode\]](#), page 125.

In batch mode, **Asymptote** reads commands directly from a file. To try this out, type

```
draw((0,0)--(100,100));
```

into a file, say test.asy. Then execute this file with the **MSDOS** or **UNIX** command


```
asy -V test
```

MSDOS users can drag and drop the file onto the Desktop **asy** icon or make **Asymptote** the default application for files with the extension **asy**.

The **-V** option opens up a **PostScript** viewer window so you can immediately view the encapsulated **PostScript** output. By default the output will be written to the file **test.eps**; the prefix of the output file may be changed with the **-o** command line option.

One can draw a line with more than two points and create a cyclic path like this square:

```
draw((0,0)--(100,0)--(100,100)--(0,100)--cycle);
```




It is often inconvenient to work directly with **PostScript** coordinates. The next example draws a unit square scaled to width 101 bp and height 101 bp. The output is identical to that of the previous example.

```
size(101,101);
draw((0,0)--(1,0)--(1,1)--(0,1)--cycle);
```

For convenience, the path `(0,0)--(1,0)--(1,1)--(0,1)--cycle` may be replaced with the predefined variable `unitsquare`, or equivalently, `box((0,0),(1,1))`.

One can also specify the size in **pt** ($1\text{ pt} = 1/72.27\text{ inch}$), **cm**, **mm**, or **inches**. If 0 is given as a size argument, no restriction is made in that direction; the overall scaling will be determined by the other direction (see [\[size\]](#), page 38):

```
size(0,3cm);
draw(unitsquare);
```



To make the user coordinates represent multiples of exactly 1cm:


```
draw(unitsquare);
shipout(unitsize=1cm);
```

One can also specify different x and y unit sizes:

```
shipout(xunitsize=1cm,yunitsize=2cm);
```

Adding labels is easy in **Asymptote**; one specifies the label as a double-quoted **LaTeX** string, a coordinate, and an optional alignment direction:


```
size(0,3cm);
draw(unitsquare);
label("$A$", (0,0), SW);
label("$B$", (1,0), SE);
label("$C$", (1,1), NE);
label("$D$", (0,1), NW);
```



`Asymptote` uses the standard compass directions $E=(1,0)$, $N=(0,1)$, $NE=\text{unit}(N+E)$, and $ENE=\text{unit}(E+NE)$, etc., which along with the directions `up`, `down`, `right`, and `left` are defined as pairs in the `Asymptote` base module `plain`.

This example draws a path that approximates a quarter circle:

```
size(100,0);
draw((1,0){up}..{left}(0,1));
```



In general, a path is specified as a list of pairs (or other paths) interconnected with `--`, which denotes a straight line segment, or `..`, which denotes a cubic spline. Specifying a final node `cycle` creates a cyclic path that connects smoothly back to the initial node, as in this approximation (accurate to within 0.06%) of a unit circle:


```
path unitcircle=E..N..W..S..cycle;
```

Each interior node of a cubic spline may be given a direction prefix or suffix `{dir}`: the direction of the pair `dir` specifies the direction of the incoming or outgoing tangent, respectively, to the curve at that node. Exterior nodes may be given direction specifiers only on their interior side.

A cubic spline between the node z_0 , with postcontrol point c_0 , and the node z_1 , with precontrol point c_1 , is computed as the Bezier curve

$$(1-t)^3 z_0 + 3t(1-t)^2 c_0 + 3t^2(1-t) c_1 + t^3 z_1 \quad 0 \leq t \leq 1.$$

As illustrated in the diagram below, the third-order midpoint (m_5) constructed from two endpoints z_0 and z_1 and two control points c_0 and c_1 , lies on the Bezier curve formed by the quadruple (z_0, c_0, c_1, z_1) . This allows one to recursively construct the desired curve, by using the newly extracted third-order midpoint as an endpoint and the respective second- and first-order midpoints as control points:



Here m_0 , m_1 and m_2 are the first-order midpoints, m_3 and m_4 are the second-order midpoints, and m_5 is the third-order midpoint. The curve is then constructed by recursively applying the algorithm to (z_0, m_0, m_3, m_5) and (m_5, m_4, m_2, z_1) .

The curve constructed in this manner has the following properties:

- It is entirely contained in the convex hull of the given four points.
- It starts heading from the first endpoint to the first control point and finishes heading from the second control point to the second endpoint.

The user can specify explicit control points between two nodes like this:

```
draw((0,0)..controls (0,100) and (100,100)..(100,0));
```

However, it is usually more convenient to just use the `..` operator, which tells **Asymptote** to choose its own control points using the algorithms described in Donald Knuth's monograph, *The MetaFontbook*, chapter 14. The user can still customize the guide (or path) by specifying direction, tension, and curl values.

The higher the tension, the straighter the curve is, and the more it approximates a straight line. One can change the spline tension from its default value of 1 to any real value greater than or equal to 0.75 (cf. John Hobby, *Discrete and Computational Geometry* 1, 1986):

```
draw((100,0)..tension 2 ..(100,100)..(0,100));
draw((100,0)..tension 2 and 1 ..(100,100)..(0,100));
draw((100,0)..tension atleast 1 ..(100,100)..(0,100));
```

The curl parameter specifies the curvature at the endpoints of a path (0 means straight; the default value of 1 means approximately circular):

```
draw((100,0){curl 0}..(100,100)..{curl 0}(0,100));
```

The **MetaPost** `... path` connector, which requests, when possible, an inflection-free curve confined to a triangle defined by the endpoints and directions, is implemented in **Asymptote** as the convenient abbreviation `::` for `..tension atleast 1 ..` (the ellipsis `...` is used in **Asymptote** to indicate a variable number of arguments; see [Section 5.10.3 \[Rest arguments\]](#), page 52). For example, compare

```
draw((0,0){up}..(100,25){right}..(200,0){down});
```



with

```
draw((0,0){up}::(100,25){right}::(200,0){down});
```




The `---` connector is an abbreviation for `..tension atleast infinity..` and the `&` connector concatenates two paths which meet at a common point.

An **Asymptote** path, being connected, is equivalent to a **Postscript** `subpath`. The `^^` binary operator, which requests that the pen be moved (without drawing or affecting endpoint

curvatures) from the final point of the left-hand path to the initial point of the right-hand path, may be used to group several `Asymptote` paths into a `path[]` array (equivalent to a PostScript path):

```
size(0,100);
path unitcircle=E..N..W..S..cycle;
path g=scale(2)*unitcircle;
filldraw(unitcircle^^g,evenodd+yellow,black);
```



The PostScript even-odd fill rule here specifies that only the region bounded between the two unit circles is filled (see [\[fillrule\]](#), page 33). In this example, the same effect can be achieved by using the default zero winding number fill rule, if one is careful to alternate the orientation of the paths:

```
filldraw(unitcircle^^reverse(g),yellow,black);
```

The ^^ operator is used by the `box3d` function in `three.asy` to construct a two-dimensional projection of the edges of a 3D cube, without retracing steps:

```
import three;
```

```
size(0,100);
```

```
currentprojection=obliqueX;
```

```
draw(unitcube);
```


```
dot(unitcube,red);
```

```
label("$O$", (0,0,0),NW);
```

```
label("(1,0,0)", (1,0,0),S);
```

```
label("(0,1,0)", (0,1,0),E);
```

```
label("(0,0,1)", (0,0,1),N);
```



See section [Section 7.19 \[graph\]](#), page 74 (or the online `Asymptote` gallery at <http://asymptote.sourceforge.net>) for further examples, including two-dimensional scientific graphs. Additional examples have been posted by Philippe Ivaldi at <http://home.tele2.fr/phivaldi/asymptote>.

4 Drawing commands

All of *Asymptote*'s graphical capabilities are based on four primitive commands. The three *PostScript* drawing commands `draw`, `fill`, and `clip` add objects to a picture in the order in which they are executed, with the most recently drawn object appearing on top. The labeling command `label` can be used to add text labels and external EPS images, which will appear on top of the *PostScript* objects (since this is normally what one wants), but again in the relative order in which they were executed. After drawing objects on a picture, the picture can be output with the `shipout` function (see [\[shipout\]](#), page 39).

If you wish to draw *PostScript* objects on top of labels (or verbatim `tex` commands; see [\[tex\]](#), page 43), the `layer` command may be used to start a new *PostScript*/*LaTeX* layer:

```
void layer(picture pic=currentpicture);
```

The `layer` function gives one full control over the order in which objects are drawn. Layers are drawn sequentially, with the most recent layer appearing on top. Within each layer, labels, images, and verbatim `tex` commands are always drawn after the *PostScript* objects in that layer.

While some of these drawing commands take many options, they all have sensible default values (for example, the picture argument defaults to `currentpicture`).

4.1 draw

```
void draw(picture pic=currentpicture, Label L="", path g,
          align align=NoAlign, pen p=currentpen,
          arrowbar arrow=None, arrowbar bar=None, margin margin=NoMargin,
          Label legend="", marker marker=nomarker);
```

Draw the path `g` on the picture `pic` using pen `p` for drawing, with optional drawing attributes (Label `L`, explicit label alignment `align`, arrows and bars `arrow` and `bar`, margins `margin`, legend, and markers `marker`). Only one parameter, the path, is required. For convenience, the arguments `arrow` and `bar` may be specified in either order. The argument `legend` is a Label to use in constructing an optional legend entry.

Bars are useful for indicating dimensions. The possible values of `bar` are `None`, `BeginBar`, `EndBar` (or equivalently `Bar`), and `Bars` (which draws a bar at both ends of the path). Each of these bar specifiers (except for `None`) will accept an optional real argument that denotes the length of the bar in *PostScript* coordinates. The default bar length is `barsize(p)`.

The possible values of `arrow` are `None`, `Blank` (which draws no arrows or path), `BeginArrow`, `MidArrow`, `EndArrow` (or equivalently `Arrow`), and `Arrows` (which draws an arrow at both ends of the path). These arrow specifiers (except for `None` and `Blank`) may be given the optional arguments real `size` (arrowhead size in *PostScript* coordinates), real `angle` (arrowhead angle in degrees), `FillDraw`, `Fill`, `NoFill`, `Draw`, and (except for `MidArrow` and `Arrows`) a relative real position along the path (an `arctime`) where the tip of the arrow should be placed. The default arrowhead size is `arrowheadsize(p)`. There are also arrow versions with slightly modified default values of `size` and `angle` suitable for curved arrows: `BeginArcArrow`, `EndArcArrow` (or equivalently `ArcArrow`), `MidArcArrow`, and `ArcArrows`.

Margins can be used to shrink the visible portion of a path by `labelmargin(p)` to avoid overlap with other drawn objects. Typical values of `margin` are `NoMargin`, `BeginMargin`, `EndMargin` (or equivalently `Margin`), and `Margins` (which leaves a margin at both ends of the path). One may use `Margin(real begin, real end)` to specify the size of the beginning and ending margin, respectively, in multiples of the units `labelmargin(p)` used for aligning labels. Alternatively, `BeginPenMargin`, `EndPenMargin` (or equivalently `PenMargin`), `PenMargins`, `PenMargin(real begin, real end)` specify a margin in units of the pen linewidth, taking account of the pen linewidth when drawing the path or arrow. For example, use `DotMargin`, an abbreviation for `PenMargin(-0.5, 0.5*dotfactor)`, to draw from the usual beginning point just up to the boundary of an end dot of width `dotfactor*linewidth(p)`. The qualifiers `BeginDotMargin`, `EndDotMargin`, and `DotMargins` work similarly. The qualifier `TrueMargin(real begin, real end)` allows one to specify a margin directly in PostScript units, independent of the pen linewidth.

The use of arrows, bars, and margins is illustrated by the examples `Pythagoras.asy`, `sqrtx01.asy`, and `triads.asy`.

The legend for a picture `pic` can be fit and aligned to a frame with the routine (see [\[filltype\]](#), page 39):

```
frame legend(picture pic=currentpicture, int perline=1,
             real xmargin=legendmargin, real ymargin=xmargin,
             real linelength=legendlinelength,
             real hskip=legendhskip, real vskip=legendvskip,
             real maxwidth=0, pen p=currentpen);
```

Here `xmargin` and `ymargin` specify the surrounding x and y margins, `perline` specifies the number of entries per line (default 1; 0 means choose this number automatically), `linelength` specifies the length of the path lines, `hskip` and `vskip` specify the line skip (as a multiple of legend entry size), `maxwidth` specifies an optional upper limit on the width of the resulting legend (0 means unlimited), and `p` specifies the pen used to draw the bounding box. The legend frame can then be added and aligned about a point on a picture `dest` using `add` or `attach` (see [\[add about\]](#), page 41).

To draw a dot, simply draw a path containing a single point. The `dot` command defined in the module `plain` draws a dot having a diameter equal to an explicit pen linewidth or the default linewidth magnified by `dotfactor` (6 by default):

```
void dot(picture pic=currentpicture, pair z, pen p=currentpen);
void dot(picture pic=currentpicture, pair[] z, pen p=currentpen);
void dot(picture pic=currentpicture, Label L, pair z, align align=NoAlign,
         string format=defaultformat, pen p=currentpen)
void dot(picture pic=currentpicture, Label L, pen p=currentpen)
```

The third routine draws a dot at every point of a pair array `z`. If the special variable `Label` is given as the `Label` argument to the fourth routine, the `format` argument will be used to format a string based on the dot location (here `defaultformat` is `"$%.4g$"`). One can also draw a dot at every node of a path:

```
void dot(picture pic=currentpicture, guide g, pen p=currentpen);
```

See [\[markers\]](#), page 84 for a more general way of marking path nodes.

To draw a fixed-sized object (in PostScript coordinates) about the user coordinate origin, use the routine

```
void draw(pair origin, picture pic=currentpicture, Label L="", path g,
         align align=NoAlign, pen p=currentpen, arrowbar arrow=None,
         arrowbar bar=None, margin margin=NoMargin, Label legend="",
         marker marker=nomarker);
```

4.2 fill

```
void fill(picture pic=currentpicture, path g, pen p=currentpen);
```

Fill the interior region bounded by the cyclic path `g` on the picture `pic`, using the pen `p`.

There is also a convenient `filldraw` command, which fills the path and then draws in the boundary. One can specify separate pens for each operation:

```
void filldraw(picture pic=currentpicture, path g, pen fillpen=currentpen,
             pen drawpen=currentpen);
```

This fixed-size version of `fill` allows one to fill an object described in PostScript coordinates about the user coordinate `origin`:

```
void fill(pair origin, picture pic=currentpicture, path g, pen p=currentpen);
```

This is just a convenient abbreviation for the commands:

```
picture opic;
fill(opic,g,p);
add(pic,opic,origin);
```

The routine

```
void filloutside(picture pic=currentpicture, path g, pen p=currentpen);
```

fills the region exterior to the path `g`.

Lattice gradient shading varying smoothly over a two-dimensional array of pens `p`, using `fillrule fillrule`, can be produced with

```
void latticeshade(picture pic=currentpicture, path g,
                 pen fillrule=currentpen, pen[][] p)
```

The pens in `p` must belong to the same color space. One can use the functions `rgb(pen)` or `cmk(pen)` to promote pens to a higher color space, as illustrated in the example file `latticeshading.asy`.

Axial gradient shading varying smoothly from `pena` to `penb` in the direction of the line segment `a--b` can be achieved with

```
void axialshade(picture pic=currentpicture, path g,
               pen pena, pair a,
               pen penb, pair b);
```

Radial gradient shading varying smoothly from `pena` on the circle with center `a` and radius `ra` to `penb` on the circle with center `b` and radius `rb` is similar:

```
void radialshade(picture pic=currentpicture, path g,
                pen pena, pair a, real ra,
                pen penb, pair b, real rb);
```

Illustrations of radial shading are provided in the example files `shade.asy` and `ring.asy`.

Gouraud shading using `fillrule fillrule` and the vertex colors in the pen array `p` on a triangular lattice defined by the vertices `z` and edge flags `edges` is implemented with

```
void gouraudshade(picture pic=currentpicture, path g,
                 pen fillrule=currentpen, pen[] p, pair[] z,
                 int[] edges);
```

The pens in `p` must belong to the same color space. Illustrations of Gouraud shading are provided in the example file `Gouraud.asy` and in the solid geometry module `solid.asy`.

The following routine uses `evenodd` clipping together with the `^^` operator to unfill a region:

```
void unfill(picture pic=currentpicture, path g);
```

4.3 clip

```
void clip(picture pic=currentpicture, path g, pen p=currentpen);
```

Clip the current contents of picture `pic` to the region bounded by the path `g`, using fillrule `p` (see [\[fillrule\]](#), [page 33](#)). For an illustration of picture clipping, see the first example in [Chapter 6 \[LaTeX usage\]](#), [page 65](#).

4.4 label

```
void label(picture pic=currentpicture, Label L, pair position,
          align align=NoAlign, pen p=nullpen, filltype filltype=NoFill)
```

Draw Label `L` on picture `pic` using pen `p`. If `align` is `NoAlign`, the label will be centered at user coordinate `position`; otherwise it will be aligned in the direction of `align` and displaced from `position` by the PostScript offset `align*labelmargin(p)`. If `p` is `nullpen`, the pen specified within the Label, which defaults to `currentpen`, will be used. The Label `L` can either be a string or the structure obtained by calling one of the functions

```
Label Label(string s="", pair position, align align=NoAlign,
            pen p=nullpen, embed embed=Rotate, filltype filltype=NoFill);
Label Label(string s="", align align=NoAlign,
            pen p=nullpen, embed embed=Rotate, filltype filltype=NoFill);
Label Label(Label L, pair position, align align=NoAlign,
            pen p=nullpen, embed embed=L.embed, filltype filltype=NoFill);
Label Label(Label L, align align=NoAlign,
            pen p=nullpen, embed embed=L.embed, filltype filltype=NoFill);
```

The text of a Label can be scaled, slanted, rotated, or shifted by multiplying it on the left by an affine transform (see [Section 5.4 \[Transforms\]](#), [page 37](#)). For example, `rotate(45)*xscale(2)*L` first scales `L` in the x direction and then rotates it counter-clockwise by 45 degrees. The final position of a Label can also be shifted by a PostScript coordinate translation: `shift(10,0)*L`. The `embed` argument determines how the Label should transform with the embedding picture:

Shift	only shift with embedding picture;
Rotate	only shift and rotate with embedding picture (default);
Slant	only shift, rotate, slant, and reflect with embedding picture;
Scale	shift, rotate, slant, reflect, and scale with embedding picture.

To add a label to a path, use

```
void label(picture pic=currentpicture, Label L, path g, align align=NoAlign,
          pen p=nullpen, filltype filltype=NoFill);
```

By default the label will be positioned at the midpoint of the path. An alternative label location (an `arctime` value between 0 and `length(g)` see [\[arctime\]](#), page 28) may be specified as real value for `position` in constructing the Label. The position `Relative(real)` specifies a location relative to the total arclength of the path. These convenient abbreviations are predefined:

```
position BeginPoint=Relative(0);
position MidPoint=Relative(0.5);
position EndPoint=Relative(1);
```

Path labels are aligned in the direction `align`, which may be specified as an absolute compass direction (pair) or a direction `Relative(pair)` measured relative to a north axis in the local direction of the path. For convenience `LeftSide`, `Center`, and `RightSide` are defined as `Relative(W)`, `Relative((0,0))`, and `Relative(E)`, respectively. Multiplying `LeftSide`, `Center`, `RightSide` on the left by a real scaling factor will move the label further away from or closer to the path.

A label with a fixed-size arrow of length `arrowlength` pointing to `b` from direction `dir` can be produced with the routine

```
void arrow(picture pic=currentpicture, Label L="", pair b, pair dir,
          real length=arrowlength, align align=NoAlign,
          pen p=currentpen, arrowbar arrow=Arrow, margin margin=EndMargin);
```

If no alignment is specified (either in the Label or as an explicit argument), the optional Label will be aligned in the direction `dir`, using margin `margin`.

The function `string graphic(string name, string options="")` returns a string that can be used to include an encapsulated PostScript (EPS) file. Here, `name` is the name of the file to include and `options` is a string containing a comma-separated list of optional bounding box (`bb=llx lly urx ury`), width (`width=value`), height (`height=value`), rotation (`angle=value`), scaling (`scale=factor`), clipping (`clip=bool`), and draft mode (`draft=bool`) parameters. The `layer()` function can be used to force future objects to be drawn on top of the included image:

```
label(graphic("file.eps","width=1cm"),(0,0),NE);
layer();
```


The `string baseline(string s, align align=S, string template="M")` function can be used to enlarge the bounding box of letters aligned below a horizontal line to match a given template, so that their baselines lie on a horizontal line. See `Pythagoras.asy` for an example.

One can prevent labels from overwriting one another with the `overwrite` pen attribute (see [\[overwrite\]](#), page 36).

The structure `object` defined in `Label.asy` allows Labels and frames to be treated in a uniform manner. A group of objects may be packed together into single frame with the routine

```
frame pack(pair align=2S ... object inset[]);
```

The `string minipage(string s, width=100pt)` function can be used to format string `s` into a paragraph of width `width`, as illustrated in the example `advection.asy`:



This example uses `clip`, `minipage`, and `graphic` to produce a CD label:

```
usepackage("fontenc", "T1");
//defaultpen(font("T1", "cmr"));
```

```


size(11.7cm,11.7cm);

asy(nativeformat(),"logo");
fill(unitcircle^(scale(2/11.7)*unitcircle),
evenodd+rgb(124/255,205/255,124/255));

label(scale(1.1)*minipage(
"\centering\scriptsize \textbf{Nonlinear Modelling, Tutorial and Manual}\\
\textsc{Gntner H. Mehring}\\
(edited by \textsc{Peter Schopf} and \textsc{Jens Schwaiger})\\
with an \textbf{Appendix} written by\\
\textsc{Wolfgang Prager} and \textsc{Jens Schwaiger}",6cm),(0,0.6));
label(scale(1.1)*minipage("\centering\scriptsize Bericht Nr. 349(2005)\\
{\bfseries Grazer Mathematische Berichte}\\
ISSN 1016--7692",4cm),(0.55,0.2));

label(graphic("logo."+nativeformat(),"height=6cm"),(0,-0.5));
clip(unitcircle^(scale(2/11.7)*unitcircle),evenodd);

```



5 Programming

Here is a short introductory example to the **Asymptote** programming language that highlights the similarity of its control structures with those of C and C++.

```
// This is a comment.

// Declaration: Declare x to be a real variable;
real x;

// Assignment: Assign the real variable x the value 1.
x=1.0;

// Conditional: Test if x equals 1 or not.
if(x == 1.0) {
    write("x equals 1.0");
} else {
    write("x is not equal to 1.0");
}

// Loop: iterate 10 times
for(int i=0; i < 10; ++i) {
    write(i);
}
```

Loops, together with user-defined functions, are illustrated in the files **wheel.asy** and **cube.asy** in the **animations** subdirectory of the examples directory. These examples use the **gifmerge** command to **merge** multiple images into a gif animation, using the **ImageMagick** **convert** program.

Asymptote also supports **while**, **do**, **break**, and **continue** statements just as in C/C++. In addition, it supports many features beyond the ones found in those languages.

5.1 Data types

Asymptote supports the following data types (in addition to user-defined types):

void	The void type is used only by functions that take or return no arguments.
bool	a boolean type that can only take on the values true and false . For example: <pre>bool b=true;</pre> defines a boolean variable b and initializes it to the value true . If no initializer is given: <pre>bool b;</pre> the value false is assumed.
int	an integer type; if no initializer is given, the implicit value 0 is assumed. The maximum integer is intMax .
real	a real number; this should be set to the highest-precision native floating-point type on the architecture. The implicit initializer for reals is 0.0. Real numbers

have precision `realEpsilon`, with `realDigits` significant digits. The smallest positive real number is `realMin` and the largest positive real number is `realMax`.

pair complex number, that is, an ordered pair of real components (`x`,`y`). The real and imaginary parts of a pair `z` can read as `z.x` and `z.y`. We say that `x` and `y` are virtual members of the data element pair; they cannot be directly modified, however. The implicit initializer for pairs is `(0.0,0.0)`.

There are a number of ways to take the complex conjugate of a pair:

```
pair z=(3,4);
z=(z.x,-z.y);
z=z.x-I*z.y;
z=conj(z);
```

A number of built-in functions are defined for pairs:

```
pair conj(pair z)
    returns the conjugate of z;

real length(pair z)
    returns the complex modulus |z| of its argument z. For example,
        pair z=(3,4);
        write(length(z));
    produces the result 5. A synonym for length(pair) is abs(pair);

real angle(pair z)
    returns the angle of z in radians in the interval [-pi,pi];

real degrees(pair z, bool warn=true)
    returns the angle of z in degrees in the interval [0,360) or 0 if warn
    is false and z.x=z.y=0 (rather than producing an error);

pair unit(pair z)
    returns a unit vector in the direction of the pair z;

pair expi(real angle)
    returns a unit vector in the direction angle measured in radians;

pair dir(real angle)
    returns a unit vector in the direction angle measured in degrees;

real xpart(pair z)
    returns z.x;

real ypart(pair z)
    returns z.y;

pair realmult(pair z, pair w)
    returns the element-by-element product (z.x*w.x,z.y*w.y);

real dot(pair z, pair w)
    returns the dot product z.x*w.x+z.y*w.y;

pair minbound(pair z, pair w)
    returns (min(z.x,w.x),min(z.y,w.y));
```

```
pair maxbound(pair z, pair w)
    returns (max(z.x,w.x),max(z.y,w.y)).
```

triple an ordered triple of real components (x,y,z) used for three-dimensional drawings. The respective components of a triple *v* can read as *v.x*, *v.y*, and *v.z*. The implicit initializer for triples is (0.0,0.0,0.0).

Here are the built-in functions for triples:

```
real length(triple v)
    returns the length |v| of the vector v. A synonym for
    length(triple) is abs(triple);
```

```
real polar(triple v)
    returns the colatitude of v measured from the z axis in radians;
```

```
real azimuth(triple v)
    returns the longitude of v measured from the x axis in radians;
```

```
real colatitude(triple v)
    returns the colatitude of v measured from the z axis in degrees;
```

```
real latitude(triple v)
    returns the latitude of v measured from the xy plane in degrees;
```

```
real longitude(triple v, bool warn=true)
    returns the longitude of v measured from the x axis in degrees; or
    0 if warn is false and v.x=v.y=0 (rather than producing an error);
```

```
triple unit(triple v)
    returns a unit triple in the direction of the triple v;
```

```
triple expi(real colatitude, real longitude)
    returns a unit triple in the direction (colatitude,longitude)
    measured in radians;
```

```
triple dir(real colatitude, real longitude)
    returns a unit triple in the direction (colatitude,longitude)
    measured in degrees;
```

```
real xpart(triple v)
    returns v.x;
```

```
real ypart(triple v)
    returns v.y;
```

```
real zpart(triple v)
    returns v.z;
```

```
real dot(triple u, triple v)
    returns the dot product u.x*v.x+u.y*v.y+u.z*v.z;
```

```
triple cross(triple u, triple v)
    returns the cross product
    (u.y*v.z-u.z*v.y,u.z*v.x-u.x*v.z,u.x*v.y-v.x*u.y);
```

```
triple minbound(triple u, triple v)
    returns (min(u.x,v.x),min(u.y,v.y),min(u.z,v.z));

triple maxbound(triple u, triple v)
    returns (max(u.x,v.x),max(u.y,v.y),max(u.z,v.z)).
```

string a character string, implemented using the STL `string` class.

Strings delimited by double quotes (") are subject to the following mapping to allow the use of double quotes in \TeX (e.g. for using the `babel` package, see [Section 7.15 \[babel\]](#), page 73):

- `\"` maps to `"`

Strings delimited by single quotes (') have the same mappings as character strings in ANSI C:

- `\'` maps to `'`
- `\"` maps to `"`
- `\?` maps to `?`
- `\\` maps to backslash
- `\a` maps to alert
- `\b` maps to backspace
- `\f` maps to form feed
- `\n` maps to newline
- `\r` maps to carriage return
- `\t` maps to tab
- `\v` maps to vertical tab
- `\0-\377` map to corresponding octal byte
- `\x0-\xFF` map to corresponding hexadecimal byte

The implicit initializer for strings is the empty string `""`. In the following string functions, position 0 denotes the start of the string.

```
int length(string s)
    returns the length of the string s;

int find(string s, string t, int pos=0)
    returns the position of the first occurrence of string t in string s at
    or after position pos, or -1 if t is not a substring of s;

int rfind(string s, string t, int pos=-1)
    returns the position of the last occurrence of string t in string s at
    or before position pos (if pos=-1, at the end of the string s), or -1
    if t is not a substring of s;

string insert(string s, int pos, string t)
    return the string formed by inserting string t at position pos in s;

string erase(string s, int pos, int n)
    returns the string formed by erasing the string of length n (if n=-1,
    to the end of the string s) at position pos in s;
```

```

string substr(string s, int pos, int n=-1)
    returns the substring of s starting at position pos and of length n
    (if n=-1, until the end of the string s);

string reverse(string s)
    return the string formed by reversing string s;

string replace(string s, string before, string after)
    returns a string with all occurrences of the string before in the
    string s changed to the string after;

string replace(string s, string[] [] table)
    returns a string constructed by translating in string s all
    occurrences of the string before in an array table of string pairs
    {before,after} to the corresponding string after;

string format(string s, int n)
    returns a string containing n formatted according to the C-style
    format string s using the current locale;

string format(string s, real x)
    returns a string containing x formatted according to the C-style
    format string s using the current locale (see the documentation for
    the C-function fprintf), except that only one data field is allowed,
    trailing zeros are removed by default (unless # is specified) and TEX
    is used to typeset scientific notation;

string time(string *format="%a %b %d %T %Z %Y")
    returns the current time formatted by the ANSI C routine strftime
    according to the string format using the current locale. Thus
    write(time());
    write(time("%a %b %d %H:%M:%S %Z %Y"));
    are equivalent ways of outputting the current time in the default
    format used by the UNIX date command.

int seconds(string t="", string *format="")
    returns the time measured in seconds after the Epoch (Thu Jan
    01 00:00:00 UTC 1970) corresponding to the string t formatted ac-
    cording to the string format using the current locale, or the current
    time if t is the empty string. For example,
    write(1+(seconds()-seconds("Jan 1","%b %d"))/(24*60*60))
    outputs today's ordinal date, measured from the beginning of the
    year.

string time(int seconds, string *format="%a %b %d %T %Z %Y")
    returns the time corresponding to seconds seconds after the Epoch
    (Thu Jan 01 00:00:00 UTC 1970) formatted by the ANSI C routine
    strftime according to the string s using the current locale. For
    example, to print the date corresponding to 24 hours ago:
    write(time(seconds()-24*60*60));

```

```
string string(real x, int digits)
    casts x using precision digits (rather than realDigits) to a string,
    using the C locale.
```

```
abort(string *s);
    Aborts execution. If string s is nonempty, a diagnostic message
    constructed from the source file, line number, and s is printed.
```

As in C/C++, complicated types may be abbreviated with **typedef** (see the example in [Section 5.10 \[Functions\]](#), page 49).

5.2 Guides and paths

guide an unresolved cubic spline (list of cubic-spline nodes and control points).

This is like a path except that the computation of the cubic spline is deferred until drawing time (when it is resolved into a path); this allows two guides with free endpoint conditions to be joined together smoothly. The solid curve in the following example is built up incrementally as a guide, but only resolved at drawing time:

```
size(200);


real mexican(real x) {return (1-8x^2)*exp(-(4x^2));}

int n=30;
real a=1.5;
real width=2a/n;

guide hat;
path solved;

for(int i=0; i < n; ++i) {
    real t=-a+i*width;
    pair z=(t,mexican(t));
    hat=hat..z;
    solved=solved..z;
}

draw(hat);
dot(hat,red);
draw(solved,dashed);
```



path a cubic spline resolved into a fixed path. In the previous example, the dashed curve is incrementally resolved at each iteration, before the entire set of nodes (shown in red) is known.

The implicit initializer for paths and guides is `nullpath`, which is useful for building up a path within a loop.

The routine `circle(pair c, real r)`, which returns a Bezier curve approximating a circle of radius `r` centered on `c`, is based on `unitcircle`:

```
path circle(pair c, real r)
{
    return shift(c)*scale(r)*unitcircle;
}
```

If high accuracy is needed, a true circle may be produced with this routine, defined in the module `graph.asy`:

```
path Circle(pair c, real r, int n=400);
```

A circular arc consistent with the above approximation centered on `c` with radius `r` from `angle1` to `angle2` degrees, drawing counterclockwise if `angle2 >= angle1`, can be constructed with

```
guide arc(pair c, real r, real angle1, real angle2);
```

If `r < 0`, the complementary arc of radius `|r|` is constructed. For convenience, an arc centered at `c` from pair `z1` to `z2` (assuming `|z2-c|=|z1-c|`) in the direction CCW (counter-clockwise) or CW (clockwise) may also be constructed with

```
guide arc(pair c, explicit pair z1, explicit pair z2,
          bool direction=CCW)
```

If high accuracy is needed, a true arc may be produced with this routine, defined in the module `graph.asy`:

```
guide Arc(pair c, real r, real angle1, real angle2,
          int n=400);
```

An ellipse can be drawn with the routine

```
guide ellipse(pair c, real a, real b)
{
    return shift(c)*xscale(a)*yscale(b)*unitcircle;
}
```

```
size(300,0);
pair[] z=new pair[10];


z[0]=(0,100); z[1]=(50,0); z[2]=(180,0);

for(int n=3; n <= 9; ++n)
    z[n]=z[n-3]+(200,0);

path p=z[0]..z[1]---z[2]::{up}z[3]
    &z[3]..z[4]--z[5]::{up}z[6]
    &z[6]::z[7]---z[8]..{up}z[9];

draw(p, grey+linewidth(4mm));

dot(z);
```



```
int length(path);
    This is the number of (linear or cubic) segments in the path. If the
    path is cyclic, this is the same as the number of nodes in the path.

int size(path);
    This is the number of nodes in the path. If the path is cyclic, this
    is the same as the path length.

pair point(path p, int n);
    If p is cyclic, return the coordinates of node n mod length(p).
    Otherwise, return the coordinates of node n, unless n < 0 (in
    which case point(0) is returned) or n > length(p) (in which case
    point(length(p)) is returned).

pair point(path p, real t);
    This returns the coordinates of the point between node floor(t)
    and floor(t)+1 corresponding to the cubic spline parameter
     $t = t - \text{floor}(t)$  (see \[Bezier\], page 9). If t lies outside the range
    [0,length(p)], it is first reduced modulo length(p) in the case
    where p is cyclic or else converted to the corresponding endpoint
    of p.

pair dir(path, int n);
    This returns the direction (as a pair) of the tangent to the path at
    node n. If the path contains only one point, (0,0) is returned.
```


`pair dir(path, real t);`
 This returns the direction of the tangent to the path at the point between node `floor(t)` and `floor(t)+1` corresponding to the cubic spline parameter $t = t - \text{floor}(t)$ (see [Bezier], page 9). If the path contains only one point, (0,0) is returned.

`pair precontrol(path, int n);`
 This returns the precontrol point of node `n`.

`pair precontrol(path, real t);`
 This returns the effective precontrol point at parameter `t`.

`pair postcontrol(path, int n);`
 This returns the postcontrol point of node `n`.

`pair postcontrol(path, real t);`
 This returns the effective postcontrol point at parameter `t`.

`real arclength(path);`
 This returns the length (in user coordinates) of the piecewise linear or cubic curve that the path represents.

`real arctime(path, real L);`
 This returns the path "time", a real number between 0 and the length of the path in the sense of `point(path, real)`, at which the cumulative arclength (measured from the beginning of the path) equals `L`.

`real dirstime(path, pair z);`
 This returns the first "time", a real number between 0 and the length of the path in the sense of `point(path, real)`, at which the tangent to the path has the direction of pair `z`, or -1 if this never happens.

`path reverse(path p);`
 returns a path running backwards along `p`.

`path subpath(path p, int n, int m);`
 returns the subpath running from node `n` to node `m`. If `n < m`, the direction of the subpath is reversed.

`path subpath(path p, real a, real b);`
 returns the subpath running from path time `a` to path time `b`, in the sense of `point(path, real)`. If `a < b`, the direction of the subpath is reversed.

`pair intersect(path p, path q, real fuzz=0);`
 If `p` and `q` have at least one intersection point, return a pair of times representing the respective path times along `p` and `q`, in the sense of `point(path, real)`, for one such intersection point (as chosen by the algorithm described on page 137 of *The MetaFontbook*). Perform the computations to the absolute error specified by `fuzz`, or, if `fuzz` is 0, to machine precision. If the paths do not intersect, return the pair (-1,-1).

```
pair intersectionpoint(path p, path q, real fuzz=0);
```

This returns `point(p,intersect(p,q,fuzz).x)`, the actual point of intersection.

```
slice firstcut(path p, path q);
```

Return the portions of path `p` before and after the first intersection of `p` with path `q` as a structure `slice` (if no such intersection exists, the entire path is considered to be ‘before’ the intersection):

```
struct slice {
    public path before,after;
}
```

Note that `firstcut.after` plays the role of the MetaPost `cutbefore` command.

```
slice lastcut(path p, path q);
```

Return the portions of path `p` before and after the last intersection of `p` with path `q` as a `slice` (if no such intersection exists, the entire path is considered to be ‘after’ the intersection).

Note that `lastcut.before` plays the role of the MetaPost `cutafter` command.

```
path buildcycle(... path[] g);
```

This returns the path surrounding a region bounded by a list of consecutively intersecting paths, following the behaviour of the MetaPost `buildcycle` command.

```
pair min(path);
```

returns the `pair(left,bottom)` for the path bounding box.

```
pair max(path);
```

returns the `pair(right,top)` for the path bounding box.

```
bool cyclic(path);
```

returns `true` iff path is cyclic

```
bool straight(path, int i);
```

returns `true` iff the segment between node `i` and node `i+1` is straight.

```
bool inside(path g, pair z, pen p=currentpen);
```

returns `true` iff the point `z` is inside the region bounded by the cyclic path `g` according to the fillrule of pen `p` (see [\[fillrule\]](#), page 33).

We point out an efficiency distinction in the use of guides and paths:

```
guide g;
for(int i=0; i < 10; ++i)
    g=g--(i,i);
path p=g;
runs in linear time, whereas

path p;
for(int i=0; i < 10; ++i)
```

```
p=p--(i,i);
```

runs in quadratic time, as the entire path up to that point is copied at each step of the iteration.

As a technical detail we note that a direction specifier given to `nullpath` modifies the node on the other side: the paths

```
a..{up>nullpath..b;
c..nullpath{up}..d;
e..{up>nullpath{down}..f;
```

are respectively equivalent to

```
a..nullpath..{up}b;
c{up}..nullpath..d;
e{down}..nullpath..{up}f;
```

5.3 Pens

In *Asymptote*, pens provide a context for the four basic drawing commands (see [Chapter 4 \[Drawing commands\]](#), page 13). They are used to specify the following drawing attributes: color, line type, line width, line cap, line join, fill rule, text alignment, font, font size, pattern, overwrite mode, and calligraphic transforms on the pen nib. The default pen used by the drawing routines is called `currentpen`. This provides the same functionality as the *MetaPost* command `pickup`.

Pens may be added together with the binary operator `+`. This will mix the colors of the two pens. All other non-default attributes of the rightmost pen will override those of the leftmost pen. Thus, one can obtain a yellow dashed pen by saying `dashed+red+green` or `red+green+dashed` or `red+dashed+green`. The binary operator `*` can be used to scale the color of a pen by a real number, until it saturates with one or more color components equal to 1.

- Colors are specified using one of the following colorspaces:

```
pen gray(real g);
```

This produces a grayscale color, where the intensity `g` lies in the interval `[0,1]`, with 0.0 denoting black and 1.0 denoting white.

```
pen rgb(real r, real g, real b);
```

This produces an RGB color, where each of the red, green, and blue intensities `r`, `g`, `b`, lies in the interval `[0,1]`.

```
pen cmyk(real c, real m, real y, real k);
```

This produces a CMYK color, where each of the cyan, magenta, yellow, and black intensities `c`, `m`, `y`, `k`, lies in the interval `[0,1]`.

```
pen invisible;
```

This special pen writes in invisible ink, but adjusts the bounding box as if something had been drawn (like the `\phantom` command in *T_EX*).

```
pen colorless(pen);
```

The function `colorless(pen)`; returns a copy of `pen p` with its color attributes stripped (to avoid color mixing).

The default color is `black`; this may be changed with the routine `defaultpen(pen)`. The function `real[] colors(pen)` returns the color components of a pen. The functions `pen gray(pen)`, `pen rgb(pen)`, and `pen cmyk(pen)` return new pens obtained by converting their arguments to the respective color spaces.

Various shades and mixtures of the grayscale primary colors `black` and `white`, RGB primary colors `red`, `green`, and `blue`, and RGB secondary colors `cyan`, `magenta`, and `yellow` are defined as named colors, along with the CMYK primary colors `Cyan`, `Magenta`, `Yellow`, and `Black`, in the module `plain`:



The standard 140 RGB X11 colors can be imported with the command

```
import x11colors;
```

and the standard 68 CMYK T_EX colors can be imported with the command

```
import texcolors;
```


Note that there is some overlap between these two standards and the definitions of some colors (e.g. `Green`) actually disagree.

Asymptote also comes with a `asycolors.sty` LaTeX package that defines to LaTeX CMYK versions of Asymptote's predefined colors, so that they can be used directly within LaTeX strings. Normally, such colors are passed to LaTeX via a pen argument; however, to change the color of only a portion of a string, say for a slide presentation, (see [Section 7.11 \[slide\]](#), [page 71](#)) it may be desirable to specify the color directly to LaTeX. This file can be passed to LaTeX with the `Asymptote` command

```
usepackage("asycolors");
```

- Line types are specified with the function `pen linetype(string s, real offset=0, bool scale=true, bool adjust=true)`, where `s` is a string of integer or real numbers separated by spaces. The optional parameter `offset` specifies where in the pattern to begin. The first number specifies how far (if `scale` is `true`, in units of the pen linewidth; otherwise in `PostScript` units) to draw with the pen on, the second number specifies how far to draw with the pen off, and so on. If `adjust` is `true`, these spacings are automatically adjusted by `Asymptote` to fit the arclength of the path. Here are the predefined line types:

```
pen solid=linetype("");
pen dotted=linetype("0 4");
pen dashed=linetype("8 8");
pen longdashed=linetype("24 8");
pen dashdotted=linetype("8 8 0 8");
pen longdashdotted=linetype("24 8 0 8");
pen Dotted=dotted+1.0;
pen Dotted(pen p=currentpen) {return dotted+2*linewidth(p);}
```



The default `linetype` is `solid`; this may be changed with `defaultpen(pen)`.

- The pen line width is specified in `PostScript` units with `pen linewidth(real)`. The default line width is 0.5 bp; this value may be changed with `defaultpen(pen)`. The line width of a pen is returned by `real linewidth(pen p=currentpen)`. For convenience, in the module `plain` we define

```
static void defaultpen(real w) {defaultpen(linewidth(w));}
static pen operator +(pen p, real w) {return p+linewidth(w);}
static pen operator +(real w, pen p) {return linewidth(w)+p;}
```

so that one may set the linewidth like this:

```
defaultpen(2);
pen p=red+0.5;
```

- A pen with a specific `PostScript` line cap is returned on calling `linecap` with an integer argument:

```
pen squarecap=linecap(0);
pen roundcap=linecap(1);
pen extendcap=linecap(2);
```

The default line cap, `roundcap`, may be changed with `defaultpen(pen)`.

- A pen with a specific `PostScript` join style is returned on calling `linejoin` with an integer argument:

```
pen miterjoin=linejoin(0);
```

```
pen roundjoin=linejoin(1);
pen beveljoin=linejoin(2);
```

The default join style, `roundjoin`, may be changed with `defaultpen(pen)`.

- A pen with a specific PostScript fill rule is returned on calling `fillrule` with an integer argument:

```
pen zerowinding=fillrule(0);
pen evenodd=fillrule(1);
```

The fill rule, which identifies the algorithm used to determine the insideness of a path or array of paths, only affects the `clip`, `fill`, and `inside` functions. For the `zerowinding` fill rule, a point `z` is outside the region bounded by a path if the number of upward intersections of the path with the horizontal line `z--z+infinity` minus the number of downward intersections is zero. For the `evenodd` fill rule, `z` is considered to be outside the region if the total number of such intersections is even. The default fill rule, `zerowinding`, may be changed with `defaultpen(pen)`.

- A pen with a specific text alignment setting is returned on calling `basealign` with an integer argument:

```
pen nobasealign=basealign(0);
pen basealign=basealign(1);
```

The default setting, `nobasealign`, which may be changed with `defaultpen(pen)`, causes the label alignment routines to use the full label bounding box for alignment. In contrast, `basealign` requests that the TeX baseline be respected.

- The font size is specified in TeX points (1 pt = 1/72.27 inches) with the function `pen fontsize(real size, real lineskip=1.2*size)`. The default font size, 12pt, may be changed with `defaultpen(pen)`. Nonstandard font sizes may require inserting

```
import fontsize;
```

at the beginning of the file. The font size and line skip of a pen can be examined with the routines `real fontsize(pen p=currentpen)` and `real lineskip(pen p=currentpen)`, respectively.

- A pen using a specific LaTeX NFSS font is returned by calling the function `pen font(string encoding, string family, string series="m", string shape="n")`. The default setting, `font("OT1","cmr","m","n")`, corresponds to 12pt Computer Modern Roman; this may be changed with `defaultpen(pen)`. Support for standardized international characters is provided by the `unicode` package (see [Section 7.13 \[unicode\]](#), page 73).

Alternatively, one may select a fixed-size TeX font (on which `fontsize` has no effect) like `"cmr12"` (12pt Computer Modern Roman) or `"pcrr"` (Courier) using the function `pen font(string name)`. An optional size argument can also be given to scale the font to the requested size: `pen font(string name, real size)`.

A nonstandard font command can be generated with `pen fontcommand(string)`.

A convenient interface to the following standard PostScript fonts is also provided:

```
pen AvantGarde(string series="m", string shape="n");
pen Bookman(string series="m", string shape="n");
pen Courier(string series="m", string shape="n");
pen Helvetica(string series="m", string shape="n");
```

```

pen NewCenturySchoolBook(string series="m", string shape="n");
pen Palatino(string series="m", string shape="n");
pen TimesRoman(string series="m", string shape="n");
pen ZapfChancery(string series="m", string shape="n");
pen Symbol(string series="m", string shape="n");
pen ZapfDingbats(string series="m", string shape="n");

```

- The transparency of a pen can be changed with the command:

```
pen opacity(real opacity=1, string *blend="Compatible");
```

The opacity can be varied from 0 (fully transparent) to the default value of 1 (opaque), and `blend` specifies one of the following foreground–background blending operations:

```

"Compatible", "Normal", "Multiply", "Screen", "Overlay", "SoftLight",
"HardLight", "ColorDodge", "ColorBurn", "Darken", "Lighten", "Difference",
"Exclusion", "Hue", "Saturation", "Color", "Luminosity",

```

as described in

<http://partners.adobe.com/public/developer/en/pdf/PDFReference16.pdf>.

Since PostScript does not support transparency, this feature is only effective with the `-f pdf` output format option; other formats can be produced from the resulting PDF file with the ImageMagick `convert` program. Labels are always drawn with an opacity of 1. A simple example of transparent filling is provided in the example file `transparency.asy`.

- PostScript commands within a `picture` may be used to create a tiling pattern, identified by the string `name`, for fill and draw operations by adding it to the default PostScript preamble frame `patterns`, with optional left-bottom margin `lb` and right-top margin `rt`.

```

void add(frame preamble=patterns, string name, picture pic, pair lb=0,
        pair rt=0)

```

To fill or draw using pattern `name`, use the pen `pattern("name")`. For example, rectangular tilings can be constructed using the routines `picture tile(real Hx=5mm, real Hy=0, pen p=currentpen, filltype filltype=NoFill)`, `picture checker(real Hx=5mm, real Hy=0, pen p=currentpen)`, and `picture brick(real Hx=5mm, real Hy=0, pen p=currentpen)` defined in `patterns.asy`:

```


size(0,90);
import patterns;

add("tile",tile());
add("filledtilewithmargin",tile(6mm,4mm,red,Fill),(1mm,1mm),(1mm,1mm));
add("checker",checker());
add("brick",brick());

real s=2.5;
filldraw(unitcircle,pattern("tile"));
filldraw(shift(s,0)*unitcircle,pattern("filledtilewithmargin"));
filldraw(shift(2s,0)*unitcircle,pattern("checker"));

```

```
filldraw(shift(3s,0)*unitcircle,pattern("brick"));
```




Hatch patterns can be generated with the routines `picture hatch(real H=5mm, pair dir=NE, pen p=currentpen),` `picture crosshatch(real H=5mm, pen p=currentpen):`

```
size(0,100);
import patterns;
```

```
add("hatch",hatch());
add("hatchback",hatch(NW));
add("crosshatch",crosshatch(3mm));
```

```
real s=1.25;
filldraw(unitsquare,pattern("hatch"));
filldraw(shift(s,0)*unitsquare,pattern("hatchback"));
filldraw(shift(2s,0)*unitsquare,pattern("crosshatch"));
```



You may need to turn off aliasing in your PostScript viewer for patterns to appear correctly. Custom patterns can easily be constructed, following the examples in `pattern.asy`. The tiled pattern can even incorporate shading (see [\[gradient shading\]](#), page 15), as illustrated in this example (not included in the manual because not all printers support PostScript 3):

```
size(0,100);
import patterns;

real d=4mm;
picture tiling;
guide square=scale(d)*unitsquare;
axialshade(tiling,square,white,(0,0),black,(d,d));
```




```
fill(tiling,shift(d,d)*square,blue);
add("shadedtiling",tiling);

filldraw(unitcircle,pattern("shadedtiling"));
```

- One can specify a custom pen nib as an arbitrary polygonal path with `pen makepen(path)`; this path represents the mark to be drawn for paths containing a single point. This pen nib path can be recovered from a pen with `path nib(pen)`. Unlike in MetaPost, the path need not be convex:

```
size(200);
pen convex=makepen(scale(10)*polygon(8))+grey;
draw((1,0.4),convex);
draw((0,0)---(1,1)..(2,0)--cycle,convex);

pen nonconvex=scale(10)*
  makepen((0,0)--(0.25,-1)--(0.5,0.25)--(1,0)--(0.5,1.25)--cycle)+red;
draw((0.5,-1.5),nonconvex);
draw((0,-1.5)..(1,-0.5)..(2,-1.5),nonconvex);
```



The value `nullpath` represents a circular pen nib (the default); an elliptical pen can be achieved simply by multiplying the pen by a transform: `yscale(2)*currentpen`.

- One can prevent labels from overwriting one another by using the pen attribute `overwrite`, which takes a single argument:

Allow Allow labels to overwrite one another. This is the default behaviour (unless overridden with `defaultpen(pen)`).

Suppress Suppress, with a warning, each label that would overwrite another label.

SuppressQuiet Suppress, without warning, each label that would overwrite another label.

Move Move a label that would overwrite another out of the way and issue a warning. As this adjustment is during the final output phase (in `PostScript` coordinates) it could result in a larger figure than requested.

MoveQuiet Move a label that would overwrite another out of the way, without warning. As this adjustment is during the final output phase (in `PostScript` coordinates) it could result in a larger figure than requested.

The routine `defaultpen()` returns the current default pen attributes. Calling the routine `resetdefaultpen()` resets all pen default attributes to their initial values.

5.4 Transforms

`Asymptote` makes extensive use of affine transforms. A pair (x,y) is transformed by the transform $t=(t.x,t.y,t.xx,t.xy,t.yx,t.yy)$ to (x',y') , where

$$\begin{aligned}x' &= t.x + t.xx * x + t.xy * y \\y' &= t.y + t.yx * x + t.yy * y\end{aligned}$$

This is equivalent to the `PostScript` transformation `[t.xx t.yx t.xy t.yy t.x t.y]`.

Transforms can be applied to pairs, guides, paths, pens, transforms, frames, and pictures by multiplication (via the binary operator `*`) on the left (see [\[circle\]](#), [page 26](#) for an example). Transforms can be composed with one another and inverted with the function `transform inverse(transform t)`; they can also be raised to any integer power with the `^` operator.

The built-in transforms are:

```
transform identity();
    the identity transform;

transform shift(pair z);
    translates by the pair z;

transform shift(real x, real y);
    translates by the pair (x,y);

transform xscale(real x);
    scales by x in the x direction;

transform yscale(real y);
    scales by y in the y direction;

transform scale(real s);
    scale by s in both x and y directions;

transform slant(real s);
    maps (x,y) -> (x+s*y,y);

transform rotate(real angle, pair z=(0,0));
    rotates by angle in degrees about z;

transform reflect(pair a, pair b);
    reflects about the line a--b.
```

The implicit initializer for transforms is `identity()`. The routines `shift(transform t)` and `shiftless(transform t)` return the transforms $(t.x,t.y,0,0,0,0)$ and $(0,0,t.xx,t.xy,t.yx,t.yy)$ respectively.

5.5 Frames and pictures

frame Frames are canvases for drawing in **PostScript** coordinates. While working with frames directly is occasionally necessary for constructing deferred drawing routines, pictures are usually more convenient to work with. The implicit initializer for frames is **newframe**. The function **bool empty(frame f)** returns **true** only if the frame **f** is empty. The functions **pair min(frame f)** and **pair max(frame f)** return the (left,bottom) and (right,top) coordinates of the frame bounding box, respectively. The contents of frame **src** may be appended to frame **dest** with the command

```
void add(frame dest, frame src);
```

or prepended with

```
void prepend(frame dest, frame src);
```

A frame obtained by aligning frame **f** in the direction **align**, in a manner analogous to the **align** argument of **label** (see [Section 4.4 \[label\]](#), page 16), is returned by

```
frame align(frame f, pair align);
```

picture Pictures are high-level structures (see [Section 5.7 \[Structures\]](#), page 45) defined in the module **plain** that provide canvases for drawing in user coordinates. The default picture is called **currentpicture**. A new picture can be created like this:

```
picture pic;
```

Anonymous pictures can be made by the expression **new picture**.

The **size** routine specifies the dimensions of the desired picture:

```
void size(picture pic=currentpicture, real x, real y,
          bool keepAspect=Aspect);
```

If the **x** and **y** sizes are both 0, user coordinates will be interpreted as **PostScript** coordinates. In this case, the transform mapping **pic** to the final output frame is **identity()**.

If exactly one of **x** or **y** is 0, no size restriction is imposed in that direction; it will be scaled the same as the other direction.

If **keepAspect** is set to **Aspect** or **true**, the picture will be scaled with its aspect ratio preserved such that the final width is no more than **x** and the final height is no more than **y**.

If **keepAspect** is set to **IgnoreAspect** or **false**, the picture will be scaled in both directions so that the final width is **x** and the height is **y**.

To ensure that each dimension is no more than **size**, use the routine

```
void size(picture pic=currentpicture, real size,
          bool keepAspect=pic.keepAspect);
```

Alternatively, calling the routine

```
transform fixedscaling(picture pic=currentpicture, pair min,
                       pair max, pen p=currentpen);
```

will cause picture `pic` to use a fixed scaling to map user coordinates in `box(min,max)` to the (already specified) picture size.

A picture can be fit to a frame and converted into a PostScript image by calling the function `shipout`:

```
void shipout(string prefix=defaultfilename, picture pic,
             real unitsize=0,
             real xunitsize=unitsize != 0 ? unitsize : 0,
             real yunitsize=unitsize != 0 ? unitsize : 0,
             frame preamble=patterns,
             orientation orientation=orientation,
             string format="", bool wait=NoWait, bool view=true);
void shipout(string prefix=defaultfilename, real unitsize=0,
             real xunitsize=unitsize != 0 ? unitsize : 0,
             real yunitsize=unitsize != 0 ? unitsize : 0,
             orientation orientation=orientation,
             string format="", bool wait=NoWait, bool view=true);
```

A `shipout()` command is added implicitly at file exit if no previous `shipout` commands have been executed.

A picture `pic` can be explicitly fit to a frame by calling

```
frame pic.fit(real xsize=pic.xsize, real ysize=pic.ysize,
              bool keepAspect=pic.keepAspect);
```

The default size and aspect ratio settings are those given to the `size` command (which default to 0, 0, and `true`, respectively).

In certain cases (e.g. 2D graphs) where only an approximate size estimate for `pic` is available, the picture fitting routine

```
frame pic.scale(real xsize=this.xsize, real ysize=this.ysize,
                bool keepAspect=this.keepAspect);
```

(which scales the resulting frame, including labels and fixed-size objects) will enforce perfect compliance with the requested size specification, but should not normally be required.

The default page orientation is `Portrait`; this may be modified by changing the variable `orientation`. To output in landscape mode, simply set `orientation=Landscape` or issue the command

```
shipout(Landscape);
```

To rotate the page by -90 degrees, use the orientation `Seascape`. To rotate the page by 180 degrees, use the orientation `UpsideDown`.

To draw a bounding box with margins around a picture, fit the picture to a frame using the function

```
frame bbox(picture pic=currentpicture, real xmargin=0,
           real ymargin=xmargin, pen p=currentpen,
           filltype filltype=NoFill);
```

Here `filltype` specifies one of the following fill types:

FillDraw Fill with the pen used to draw the boundary.

FillDraw(real *xmargin*=0, real *ymargin*=*xmargin*, pen *p*=nullpen)
 If *p* is nullpen, fill with the pen used to draw the boundary; otherwise fill with pen *p*. An optional margin of *xmargin* and *ymargin* can be specified.

Fill Fill with the drawing pen.

Fill(real *xmargin*=0, real *ymargin*=*xmargin*, pen *p*=nullpen)
 If *p* is nullpen, fill with the drawing pen; otherwise fill with pen *p*. An optional margin of *xmargin* and *ymargin* can be specified.

NoFill Do not fill.

Draw Draw only the boundary.

Draw(real *xmargin*=0, real *ymargin*=*xmargin*, pen *p*=nullpen)
 If *p* is nullpen, draw the boundary with the drawing pen; otherwise draw with pen *p*. An optional margin of *xmargin* and *ymargin* can be specified.

UnFill Clip the region.

UnFill(real *xmargin*=0, real *ymargin*=*xmargin*
 Clip the region and surrounding margins *xmargin* and *ymargin*.

RadialShade(pen *penc*, pen *penr*)
 Fill varying radially from *penc* at the center of the bounding box to *penr* at the edge.

For example, to draw a bounding box around a picture with a 0.25 cm margin and output the resulting frame, use the command:

```
shipout(bbox(0.25cm));
```

A picture may be fit to a frame with the background color of pen *p* with the function `bbox(p,Fill)`.

The functions

```
pair min(picture pic);
pair max(picture pic);
```

calculate the `PostScript` bounds that picture *pic* would have if it were currently fit to a frame using its default size specification.

The function

```
pair point(picture pic=currentpicture, pair dir);
```

is a convenient way of determining the point on the boundary of the user-coordinate bounding box of *pic*, ignoring the contributions from fixed-size objects (such as labels and arrowheads), in the direction *dir* relative to its center.

The function

```
pair truepoint(picture pic=currentpicture, pair dir);
```

is identical to `point`, except that it also accounts for fixed-size objects, using the scaling transform that picture *pic* would have if currently fit to a frame using its default size specification.

The function

```
pair framepoint(picture pic=currentpicture, pair dir);
```

works similarly, except the result is a point in PostScript coordinates on the bounding box that picture `pic` would have if currently fit to a frame using its default size specification.

Sometimes it is useful to draw objects on separate pictures and add one picture to another using the `add` function:

```
void add(picture src, bool group=true,
        filltype filltype=NoFill, bool put=Above);
void add(picture dest, picture src, bool group=true,
        filltype filltype=NoFill, bool put=Above);
```

The first example adds `src` to `currentpicture`; the second one adds `src` to `dest`. The `group` option specifies whether or not the graphical user interface `xasy` should treat all of the elements of `src` as a single entity (see [Chapter 10 \[GUI\]](#), [page 126](#)), `filltype` requests optional background filling or clipping, and `put` specifies whether to add `src` above or below existing objects.

There are also routines to add a picture or frame `src` specified in postscript coordinates to another picture about the user coordinate `position`:

```
void add(picture dest, picture src, pair position, bool group=true,
        filltype filltype=NoFill, bool put=Above);
void add(picture src, pair position, bool group=true,
        filltype filltype=NoFill, bool put=Above);
void add(picture dest=currentpicture, frame src, pair position=0,
        bool group=true, filltype filltype=NoFill,
        bool put=Above);
void add(picture dest=currentpicture, frame src, pair position,
        pair align, bool group=true, filltype filltype=NoFill,
        bool put=Above);
```

The optional `align` argument in the last form specifies a direction to use for aligning the frame, in a manner analogous to the `align` argument of `label` (see [Section 4.4 \[label\]](#), [page 16](#)). However, one key difference is that when `align` is not specified, labels are centered, whereas frames and pictures are aligned so that their origin is at `position`. Illustrations of frame alignment can be found in the examples [\[errorbars\]](#), [page 84](#) and [\[image\]](#), [page 99](#). If you want to align 3 or more subpictures, group them two at a time:

```
picture pic1;
real size=50;
size(pic1,size);
fill(pic1,(0,0)--(50,100)--(100,0)--cycle,red);
```

```
picture pic2;
size(pic2,size);
fill(pic2,unitcircle,green);
```

```
picture pic3;
size(pic3,size);
```


```

fill(pic3,unitsquare,blue);

picture pic;
add(pic,pic1.fit() ,(0,0),N);
add(pic,pic2.fit() ,(0,0),10S);

add(pic.fit() ,(0,0),N);
add(pic3.fit() ,(0,0),10S);

```



Alternatively, one can use `attach` to automatically increase the size of picture `dest` to accommodate adding a frame `src` about the user coordinate `position`:

```

void attach(picture dest=currentpicture, frame src,
            pair position=0, bool group=true,
            filltype filltype=NoFill, bool put=Above);
void attach(picture dest=currentpicture, frame src,
            pair position=0, pair align, bool group=true,
            filltype filltype=NoFill, bool put=Above);

```

To draw or fill a box or ellipse around a label, frame, or picture, use one of the routines (the first two routines for convenience also return the boundary as a guide):

```

guide box(frame f, Label L="", real xmargin=0,
          real ymargin=xmargin, pen p=currentpen,
          filltype filltype=NoFill, bool put=Above);
guide ellipse(frame f, Label L="", real xmargin=0,
              real ymargin=xmargin, pen p=currentpen,
              filltype filltype=NoFill, bool put=Above);
void box(picture pic=currentpicture, Label L,
         real xmargin=0, real ymargin=xmargin, pen p=currentpen,
         filltype filltype=NoFill, bool put=Above);

```

To erase the contents of a picture (but not the size specification), use the function

```
void erase(picture pic=currentpicture);
```

To save a snapshot of `currentpicture`, `currentpen`, and `currentprojection`, use the function `save()`.

To restore a snapshot of `currentpicture`, `currentpen`, and `currentprojection`, use the function `restore()`.

Many further examples of picture and frame operations are provided in the base module `plain`.

It is possible to insert verbatim PostScript commands in a picture with the routine

```
void postscript(picture pic=currentpicture, string s);
```

Verbatim \TeX commands can be inserted in the intermediate LaTeX output file with the function

```
void tex(picture pic=currentpicture, string s);
```

To issue a global \TeX command (such as a \TeX macro definition) in the \TeX preamble (valid for the remainder of the top-level module) use:

```
void texpreamble(string s);
```

The \TeX environment can be reset to its initial state, clearing all macro definitions, with the function

```
void texreset();
```

The routine

```
void usepackage(string s);
```

is a convenient abbreviation for `texpreamble("\usepackage{"+s+"}");` that can be used for importing LaTeX packages.

5.6 Files

`Asymptote` can read and write text files (including comma-separated value) files and portable XDR (External Data Representation) binary files.

An input file must first be opened with `input(string name, bool check=true, string commentchar="#");` reading is then done by assignment:

```
file fin=input("test.txt");
real a=fin;
```

If the optional boolean argument `check` is `false`, no check will be made that the file exists. If the file does not exist or is not readable, the function `bool error(file)` will return `true`. The first character of the string `commentchar` specifies a comment character. If this character is encountered in a data file, the remainder of the line is ignored. When reading strings, the comment character must be in the first column (otherwise it will be treated as an ordinary character).

One can change the current working directory to the contents of the string `d` with the function `string cd(string d)`, which returns the new working directory.

When reading pairs, the enclosing parenthesis are optional. Strings are also read by assignment, by reading characters up to but not including a newline. In addition, `Asymptote`

provides the function `string getc(file)` to read the next character only, returning it as a string.

A file named `name` can be open for output with

```
file output(string name, bool append=false);
```

data will be appended to an existing file only if the file is opened with `append=true`. Data of a built-in type `T` can be written to an output file by calling one of the functions

```
write(string s="", T x, suffix suffix=endl ... T[]);
write(file fout, string s="", T x, suffix suffix=None ... T[]);
write(file fout=stdout, string s="", explicit T[] x ... T[] []);
write(file fout=stdout, T[] []);
write(file fout=stdout, T[] [] []);
write(suffix suffix=endl);
write(file fout, suffix suffix=None);
```

If `fout` is not specified, `stdout` is used and terminated by default with a newline. If specified, the optional identifying string `s` is written before the data `x`. An arbitrary number of data values may be listed when writing scalars or one-dimensional arrays. The `suffix` may be one of the following: `None` (do nothing), `endl` (terminate with a newline), or `tab` (terminate with a tab). Here are some simple examples of data output:

```
file fout=output("test.txt");
write(fout,1);           // Writes "1"
write(fout);             // Writes a new line
write(fout,"List: ",1,2,3); // Writes "List: 1      2      3"
```

There are two special files: `stdin`, which reads from the keyboard, and `stdout`, which writes to the terminal.

A file may also be opened with `xinput` or `xoutput` instead of `input` or `output`, in which case it will read or write double precision values written in Sun Microsystem's XDR (External Data Representation) portable binary format (available on all UNIX platforms). The function `file single(file)` sets the file to read single precision XDR values; calling `file single(file,false)` sets it back to read doubles again. The default initializer for file is `stdout`.

One can test a file for end-of-file with the boolean function `eof(file)`, end-of-line with `eol(file)`, and for I/O errors with `error(file)`. One can flush the output buffers with `flush(file)`, clear a previous I/O error with `clear(file)`, and close the file with `close(file)`. To set the number of digits of output precision, use `precision(file,int)`. The function `int tell(file)` returns the current position in an input file relative to the beginning. The function `seek(file, int)` can be used to change this position; for example, to rewind a file, use the command `seek(file,0)`.

Assigning `settings.scroll=n` for a positive integer `n` requests a pause after every `n` output lines to `stdout`. One may then press `Enter` to continue or `q` followed by `Enter` to quit. If `n` is negative, the output scrolls a page at a time (i.e. by one less than the current number of display lines). The default value, `settings.scroll=0`, specifies continuous scrolling.

The routines

```

string getstring(string name="", string default="", string prompt="",
                bool save=true);
int getint(string name="", int default=0, string prompt="",
           bool save=true);
real getreal(string name="", real default=0, string prompt="",
             bool save=true);
pair getpair(string name="", pair default=0, string prompt="",
             bool save=true)

```

defined in the module `plain` may be used to prompt for a value from `stdin` using the GNU `readline` library. If `save=true`, the history of values for `name` is saved to the file `".asy_"+name` (see [\[history\]](#), page 125). The most recent value in the history will be used to provide a default value for subsequent runs. The default value (initially `default`) is displayed after `prompt`. These routines are based on the following interface to `readline`, which prompts the user with the default value formatted according to `prompt` and saves the local history under the name `".asy_"+history`, unless the string `history` begins with a linefeed (`'\n'`):

```

string readline(string prompt="", string history="", string initial="",
               bool tabcompletion=false);

```

5.7 Structures

Users may also define their own data types as structures, along with user-defined operators, much as in C++. By default, structure members are `public` (may be read and modified anywhere in the code), but may be optionally declared `restricted` (readable anywhere but writeable only inside the structure where they are defined) or `private` (readable and writable only inside the structure). The virtual structure `this` refers to the enclosing structure. Any code at the top-level scope within the structure is executed on initialization.

A default initializer for a structure `S` can be defined by creating a function `S operator init()`. This can be used to initialize each instance of `S` with `new S` (which creates a new anonymous instance of `S`).

```

struct S {
    public real a=1;
    real f(real a) {return a+this.a;}
}

S operator init() {return new S;}

S s;                                // Initializes s with S operator init();

write(s.f(2));                       // Outputs 3

S operator + (S s1, S s2)
{
    S result;
    result.a=s1.a+s2.a;
    return result;
}

```

```

}

write((s+s).f(0));           // Outputs 2

```

In the following example, the static function `T.T(real x)` is a constructor that initializes and returns a new instance of `T`:

```

struct T {
    real x;
    static T T(real x) {T t=new T; t.x=x; return t;}
}

```

```

T operator init() {return new T;}

```

```

T a;
T b=T.T(1);

```

```

write(a.x);           // Outputs 0
write(b.x);           // Outputs 1

```

The name of the constructor need not be identical to the name of the structure; for example, see `triangle.SAS` in `geometry.asy`.

Structure assignment does a shallow copy; a deep copy requires writing an explicit `copy()` member. The function `bool alias(T,T)` checks to see if two instances of the structure `T` are identical. The boolean operators `==` and `!=` are by default equivalent to `alias` and `!alias` respectively, but may be overwritten for a particular type do a deep comparison.

When `a` is defined both as a variable and a type, the qualified name `a.b` refers to the variable instead of the type.

Much like in C++, casting (see [Section 5.12 \[Casts\]](#), page 60) provides for an elegant implementation of structure inheritance, including virtual functions:

```

struct parent {
    real x=1;
    public void virtual(int) {write (0);}
    void f() {virtual(1);}
}

```

```

parent operator init() {return new parent;}

```

```

void write(parent p) {write(p.x);}

```

```

struct child {
    parent parent;
    real y=2;
    void virtual(int x) {write (x);}
    parent.virtual=virtual;
    void f()=parent.f;
}

```

```

parent operator cast(child child) {return child.parent;}

child operator init() {return new child;}

parent p;
child c;

write(c);                      // Outputs 1;

p.f();                         // Outputs 0;
c.f();                         // Outputs 1;

write(c.parent.x);             // Outputs 1;
write(c.y);                    // Outputs 2;

```

Further examples of structures are `Legend` and `picture` in the `Asymptote` base module `plain`.

5.8 Operators

5.8.1 Arithmetic & logical operators

`Asymptote` uses the standard binary arithmetic operators. However, when one integer is divided by another, both arguments are converted to real values before dividing and a real quotient is returned (since this is usually what is intended). The function `int quotient(int x, int y)` returns the greatest integer less than or equal to x/y . In all other cases both operands are promoted to the same type, which will also be the type of the result:

+	addition
-	subtraction
*	multiplication
/	division
%	modulo; the result always has the same sign as the divisor. In particular, this makes <code>q*quotient(p,q)+p%q == p</code> for all integers <code>p</code> and nonzero integers <code>q</code> .
^	power; if the exponent (second argument) is an int, recursive multiplication is used; otherwise, logarithms and exponentials are used. <code>**</code> is a synonym for <code>^</code> .

The usual boolean operators are also defined:

==	equals
!=	not equals
<	less than
<=	less than or equals
>=	greater than or equals
>	greater than

<code>&&</code>	and
<code> </code>	or
<code>^</code>	xor
<code>!</code>	not

Asymptote also supports the C-like conditional syntax:

```
bool positive=(pi >= 0) ? true : false;
```

The function `T interp(T a, T b, real t)` returns `a+t*(b-a)` for all built-in arithmetic types `T`.

5.8.2 Self & prefix operators

As in C, each of the arithmetic operators `+`, `-`, `*`, `/`, `%`, and `^` can be used as a self operator. The prefix operators `++` (increment by one) and `--` (decrement by one) are also defined. For example,

```
int i=1;
i += 2;
int j=++i;
```

is equivalent to the code

```
int i=1;
i=i+2;
int j=i=i+1;
```

However, postfix operators like `i++` and `i--` are not defined (because of the inherent ambiguities that would arise with the `--` path-joining operator). In the rare instances where `i++` and `i--` are really needed, one can substitute the expressions `(++i-1)` and `(--i+1)`, respectively.

5.8.3 User-defined operators

The following symbols may be used with `operator` to define or redefine operators on structures and built-in types:

```
- + * / % ^ ! < > == != <= >= & && || ^^ .. :: -- --- ++
<< >> $ $$ @ @@
```

The operators on the second line have precedence one higher than the boolean operators `<`, `>`, `<=`, and `>=`.

Guide operators like `..` may be overloaded, say, to write a user function that produces a new guide from a given guide:

```
guide dots(... guide[] g)=operator ..;
```

```
guide operator ..(... guide[] g) {
  guide G;
  if(g.length > 0) {
    write(g[0]);
    G=g[0];
  }
  for(int i=1; i < g.length; ++i) {
```

```

        write(g[i]);
        write();
        G=dots(G,g[i]);
    }
    return G;
}

guide g=(0,0){up}..{SW}(100,100){NE}..{curl 3}(50,50)..(10,10);
write("g=",none);
write(g);

```

5.9 Implicit scaling

If a numeric literal is in front of certain types of expressions, then the two are multiplied:

```

int x=2;
real y=2.0;
real cm=72/2.540005;

write(3x);
write(2.5x);
write(3y);
write(-1.602e-19 y);
write(0.5(x,y));
write(2x^2);
write(3x+2y);
write(3(x+2y));
write(3sin(x));
write(3(sin(x))^2);
write(10cm);

```

This produces the output

```

6
5
6
-3.204e-19
(1,1)
8
10
18
2.72789228047704
2.48046543129542
283.464008929116

```

5.10 Functions

`Asymptote` functions are treated as variables with a signature (non-function variables have null signatures). Variables with the same name are allowed, so long as they have distinct signatures.

Functions arguments are passed by value. To pass an argument by reference, simply enclose it in a structure (see [Section 5.7 \[Structures\]](#), page 45).

Here are some significant features of `Asymptote` functions:

1. Variables with signatures (functions) and without signatures (nonfunction variables) are distinct:

```
int x, x();
x=5;
x=new int() {return 17;};
x=x();           // calls x() and puts the result, 17, in the scalar x
```

2. Traditional function definitions are allowed:

```
int sqr(int x)
{
    return x*x;
}
sqr=null;           // but the function is still just a variable.
```

3. Casting can be used to resolve ambiguities:

```
int a, a(), b, b(); // Valid: creates four variables.
a=b;                // Invalid: assignment is ambiguous.
a=(int) b;          // Valid: resolves ambiguity.
(int) (a=b);        // Valid: resolves ambiguity.
(int) a=b;          // Invalid: cast expressions cannot be L-values.
```

```
int c();
c=a;                // Valid: only one possible assignment.
```

4. Anonymous (so-called "high-order") functions are also allowed:

```
typedef int intop(int);
intop adder(int m)
{
    return new int(int n) {return m+n;};
}
intop addby7=adder(7);
write(addby7(1));   // Writes 8.
```

5. One may redefine a function `f`, even for calls to `f` in previously declared functions, by assigning another (anonymous or named) function to it. However, if `f` is overloaded by a new function definition, previous calls will still access the original version of `f`, as illustrated in this example:

```
void f() {
    write("hi");
}

void g() {
    f();
}
```

```

g(); // writes "hi"

f=new void() {write("bye");};

g(); // writes "bye"

void f() {write("overloaded");};

f(); // writes "overloaded"
g(); // writes "bye"

```

6. Anonymous functions can be used to redefine a function variable that has been declared (and implicitly initialized to the null function) but not yet explicitly defined:

```

void f(bool b);

void g(bool b) {
    if(b) f(b);
    else write(b);
}

f=new void(bool b) {
    write(b);
    g(false);
};

g(true); // Writes true, then writes false.

```

Asymptote is the only language we know of that treats functions as variables, but allows overloading by distinguishing variables based on their signatures.

Functions are allowed to call themselves recursively. As in C++, infinite nested recursion will generate a stack overflow (reported as a segmentation fault, unless a fully working version of the GNU library `libsigsegv` (e.g. 2.4 or later) is installed at configuration time).

5.10.1 Default arguments

Asymptote supports a more flexible mechanism for default function arguments than C++: they may appear anywhere in the function prototype. Because certain data types are implicitly cast to more sophisticated types (see [Section 5.12 \[Casts\]](#), page 60) one can often avoid ambiguities by ordering function arguments from the simplest to the most complicated. For example, given

```
real f(int a=1, real b=0) {return a+b;}
```

then `f(1)` returns 1.0, but `f(1.0)` returns 2.0.

The value of a default argument is determined by evaluating the given **Asymptote** expression in the scope where the called function is defined.

5.10.2 Named arguments

It is sometimes difficult to remember the order in which arguments appear in a function declaration. Named (keyword) arguments make calling functions with multiple arguments

easier. Unlike in the C and C++ languages, an assignment in a function argument is interpreted as an assignment to a parameter of the same name in the function signature, *not within the local scope*. The command-line option `-d` may be used to check `Asymptote` code for cases where a named argument may be mistaken for a local assignment.

When matching arguments to signatures, first all of the keywords are matched, then the arguments without names are matched against the unmatched formals as usual. For example,

```
int f(int x, int y) {
    return 10x+y;
}
write(f(4,x=3));
```

output 34, as `x` is already matched when we try to match the unnamed argument 4, so it gets matched to the next item, `y`.

For the rare occasions where it is desirable to assign a value to local variable within a function argument (generally *not* a good programming practice), simply enclose the assignment in parentheses. For example, given the definition of `f` in the previous example,

```
int x;
write(f(4,(x=3)));
```

is equivalent to the statements

```
int x;
x=3;
write(f(4,3));
```

and outputs 43.

As a technical detail, we point out that, since variables of the same name but different signatures are allowed in the same scope, the code

```
int f(int x, int x()) {
    return x+x();
}
int seven() {return 7;}
```

is legal in `Asymptote`, with `f(2,seven)` returning 9. A named argument matches the first unmatched formal of the same name, so `f(x=2,x=seven)` is an equivalent call, but `f(x=seven,2)` is not, as the first argument is matched to the first formal, and `int ()` cannot be implicitly cast to `int`. Default arguments do not affect which formal a named argument is matched to, so if `f` were defined as

```
int f(int x=3, int x()) {
    return x+x();
}
```

then `f(x=seven)` would be illegal, even though `f(seven)` obviously would be allowed.

5.10.3 Rest arguments

Rest arguments allow one to write functions that take a variable number of arguments:

```
// This function sums its arguments.
int sum(... int[] nums) {
    int total=0;
```

```

    for (int i=0; i < nums.length; ++i)
        total += nums[i];
    return total;
}

sum(1,2,3,4);           // returns 10
sum();                  // returns 0

// This function subtracts subsequent arguments from the first.
int subtract(int start ... int[] subs) {
    for (int i=0; i < subs.length; ++i)
        start -= subs[i];
    return start;
}

subtract(10,1,2);       // returns 7
subtract(10);           // returns 10
subtract();             // illegal

```

Putting an argument into a rest array is called *packing*. One can give an explicit list of arguments for the rest argument, so `subtract` could alternatively be implemented as

```

int subtract(int start ... int[] subs) {
    return start - sum(... subs);
}

```

One can even combine normal arguments with rest arguments:

```
sum(1,2,3 ... new int[] {4,5,6}); // returns 21
```

This builds a new six-element array that is passed to `sum` as `nums`. The opposite operation, *unpacking*, is not allowed:

```
subtract(... new int[] {10, 1, 2});
```

is illegal, as the start formal is not matched.

If no arguments are packed, then a zero-length array (as opposed to `null`) is bound to the rest parameter. Note that default arguments are ignored for rest formals and the rest argument is not bound to a keyword.

The overloading resolution in *Asymptote* is similar to the function matching rules used in C++. Every argument match is given a score. Exact matches score better than matches with casting, and matches with formals (regardless of casting) score better than packing an argument into the rest array. A candidate is maximal if all of the arguments score as well in it as with any other candidate. If there is one unique maximal candidate, it is chosen; otherwise, there is an ambiguity error.

```

int f(path g);
int f(guide g);
f((0,0)--(100,100)); // matches the second; the argument is a guide

int g(int x, real y);
int g(real x, int x);

```

```

g(3,4); // ambiguous; the first candidate is better for the first argument,
        // but the second candidate is better for the second argument

int h(... int[] rest);
int h(real x ... int[] rest);

h(1,2); // the second definition matches, even though there is a cast,
        // because casting is preferred over packing

int i(int x ... int[] rest);
int i(real x, real y ... int[] rest);

i(3,4); // ambiguous; the first candidate is better for the first argument,
        // but the second candidate is better for the second one

```

5.10.4 Mathematical functions

Asymptote has built-in versions of the standard `libm` mathematical `real(real)` functions `sin`, `cos`, `tan`, `asin`, `acos`, `atan`, `exp`, `log`, `pow10`, `log10`, `sinh`, `cosh`, `tanh`, `asinh`, `acosh`, `atanh`, `sqrt`, `cbrt`, `fabs`, as well as the identity function `identity`. Asymptote also defines the order `n` Bessel functions of the first kind `J(int n, real)` and second kind `Y(int n, real)`, as well as the gamma function `gamma`, the error function `erf`, and the complementary error function `erfc`. The standard `real(real, real)` functions `atan2`, `hypot`, `fmod`, `remainder` are also included.

The functions `degrees(real radians)` and `radians(real degrees)` can be used to convert between radians and degrees. The function `Degrees(real radians)` returns the angle in degrees in the interval `[0,360)`. For convenience, Asymptote defines variants `Sin`, `Cos`, `Tan`, `aSin`, `aCos`, and `aTan` of the standard trigonometric functions using degrees rather than radians.

The functions `floor`, `ceil`, and `round` differ from their usual definitions in that they all return an `int` value rather than a `real` (since that is normally what one wants). The functions `Floor`, `Ceil`, and `Round` are respectively similar, except that if the result cannot be converted to a valid `int`, they return `intMax` for positive arguments and `-intMax` for negative arguments, rather than generating an integer overflow. We also define a function `sgn`, which returns the sign of its `real` argument as an integer (`-1`, `0`, or `1`).

There is an `abs(int)` function, as well as an `abs(real)` function (equivalent to `fabs(real)`) and an `abs(pair)` function (equivalent to `length(pair)`).

Random numbers can be seeded with `srand(int)` and generated with the `int rand()` function, which returns a random integer between `0` and the integer `randMax`. A Gaussian random number generator `Gaussrand` and a collection of statistics routines, including `histogram`, are provided in the base file `stats.asy`.

When configured with the GNU Scientific Library (GSL), available from <http://www.gnu.org/software/gsl/>, Asymptote contains an internal module `gsl` that defines the airy functions `Ai(real)`, `Bi(real)`, `Ai_deriv(real)`, `Bi_deriv(real)`, `zero_Ai(int)`, `zero_Bi(int)`, `zero_Ai_deriv(int)`, `zero_Bi_deriv(int)`, the Bessel functions `I(int, real)`, `K(int, real)`, `j(int, real)`, `y(int, real)`, `i_scaled(int, real)`, `k_scaled(int, real)`, `J(real, real)`, `Y(real, real)`, `I(real, real)`, `K(real,`

`real`), `zero_J(real, int)`, the elliptic functions `F(real, real)`, `E(real, real)`, and `P(real, real)`, the exponential/trigonometric integrals `Ei`, `Si`, and `Ci`, the Legendre polynomials `Pl(int, real)`, and the Riemann zeta function `zeta(real)`. For example, to compute the sine integral `Si` of 1.0:

```
import gsl;
write(Si(1.0));
```

5.11 Arrays

Appending `[]` to a built-in or user-defined type yields an array. The array element `i` of an array `A` can be accessed as `A[i]`. By default, attempts to access or assign to an array element using a negative index generates an error. Reading an array element with an index beyond the length of the array also generates an error; however, assignment to an element beyond the length of the array causes the array to be resized to accommodate the new element. One can also index an array `A` with an integer array `B`: the array `A[B]` is formed by indexing array `A` with successive elements of array `B`.

The declaration

```
real[] A;
```

initializes `A` to be an empty (zero-length) array. Empty arrays should be distinguished from null arrays. If we say

```
real[] A=null;
```

then `A` cannot be dereferenced at all (null arrays have no length and cannot be read from or assigned to).

Arrays can be explicitly initialized like this:

```
real[] A={0,1,2};
```

Array assignment in `Asymptote` does a shallow copy: only the pointer is copied (if one copy is modified, the other will be too). The `copy` function listed below provides a deep copy of an array.

Every array `A` of type `T[]` has the virtual members `int length`, `void cyclic(bool b)`, `bool cyclicflag`, `T push(T x)`, `void append(T[] a)`, `T pop()`, `T insert(int i, T x)`, and `T delete(int i)`. The member `A.length` evaluates to the length of the array. Setting `A.cyclic(true)` signifies that array indices should be reduced modulo the current array length. Reading from or writing to a nonempty cyclic array never leads to out-of-bounds errors or array resizing. The member `A.cyclicflag` returns the current setting of the cyclic flag. The functions `A.push` and `A.append` append their arguments onto the end of the array, while `A.insert(int i, T x)` inserts `x` into the array so that it will be the element with index `i`. For convenience `A.push` and `A.insert` return the inserted item. The function `A.pop()` pops and returns the last element, while `A.delete(int i)` pops and returns the element with index `i`, shifting the position of all higher-indexed elements down by one. Like all `Asymptote` functions, `cyclic`, `push`, `append`, `pop`, `insert`, and `delete` can be "pulled off" of the array and used on their own. For example,

```
int[] A={1};
A.push(2);           // A now contains {1,2}.
A.append(A);         // A now contains {1,2,1,2}.
int f(int)=A.push;
```

```

f(3);           // A now contains {1,2,1,2,3}.
int g()=A.pop;
write(g());     // Outputs 3.
A.delete(0);    // A now contains {2,1,2}.
A.insert(1,3);  // A now contains {2,3,1,2}.

```

The `[]` suffix can also appear after the variable name; this is sometimes convenient for declaring a list of variables and arrays of the same type:

```
real a,A[];
```

This declares `a` to be `real` and implicitly declares `A` to be of type `real[]`.

In the following list of built-in array functions, `T` represents a generic type. Note that the internal functions `alias`, `copy`, `concat`, `sequence`, `map`, and `transpose`, which depend on type `T[]`, are defined only after the first declaration of a variable of type `T[]`.

`new T[]` returns a new empty array of type `T[]`;

`new T[] {list}`
returns a new array of type `T[]` initialized with `list` (a comma delimited list of elements).

`new T[n]` returns a new array of `n` elements of type `T[]`. Unless they are arrays themselves, these `n` array elements are not initialized.

`int[] sequence(int n)`
if `n >= 1` returns the array `{0,1,...,n-1}` (otherwise returns a null array);

`int[] sequence(int n, int m)`
if `m >= n` returns an array `{n,n+1,...,m}` (otherwise returns a null array);

`T[] sequence(T f(int), int n)`
if `n >= 1` returns the sequence `{f_i : i=0,1,...,n-1}` given a function `T f(int)` and integer `int n` (otherwise returns a null array);

`int[] reverse(int n)`
if `n >= 1` returns the array `{n-1,n-2,...,0}` (otherwise returns a null array);

`int[] complement(int[] a, int n)`
returns the complement of the integer array `a` in `{1,2,...,n}`, so that `b[complement(a,b.length)]` yields the complement of `b[a]`.

`int find(bool[], int n=1)`
returns the index of the `n`th `true` value or `-1` if not found. If `n` is negative, search backwards from the end of the array for the `-n`th value;

`int search(T[] a, T key)`
For built-in ordered types `T`, searches a sorted ordered array `a` of `n` elements to find an interval containing `key`, returning `-1` if `key` is less than the first element, `n-1` if `key` is greater than or equal to the last element, and otherwise the index corresponding to the left-hand (smaller) endpoint.

`T[] copy(T[] a)`
returns a deep copy of the array `a`;

```

T[] [] copy(T[] [] a)
    returns a deep copy of the array a;

T[] [] [] copy(T[] [] [] a)
    returns a deep copy of the array a;

T[] concat(T[] a, T[] b)
    returns a new array formed by concatenating arrays a and b;

bool alias(T[] a, T[] b)
    returns true if the arrays a and b are identical;

T[] sort(T[] a)
    For built-in ordered types T, returns a copy of a sorted in ascending order;

T[] [] sort(T[] [] a)
    For built-in ordered types T, returns a copy of a with the rows sorted by the
    first column, breaking ties with successively higher columns. For example:
    string[] [] a={{ "bob", "9"}, {"alice", "5"}, {"pete", "7"},
                   {"alice", "4"} };
    // Row sort (by column 0, using column 1 to break ties):
    write(stdout, sort(a));

    produces
    alice    4
    alice    5
    bob      9
    pete     7

T[] [] transpose(T[] [] a)
    returns the transpose of a.

T[] [] [] transpose(T[] [] [] a, int[] perm)
    returns the 3D transpose of a obtained by applying the permutation perm of
    new int[] {0,1,2} to the indices of each entry.

T sum(T[] a)
    For arithmetic types T, returns the sum of a.

T min(T[] a)
T min(T[] [] a)
T min(T[] [] [] a)
    For built-in ordered types T, returns the minimum element of a.

T max(T[] a)
T max(T[] [] a)
T max(T[] [] [] a)
    For built-in ordered types T, returns the maximum element of a.

map(f(T), T[] a)
    returns the array obtained by applying the function f to each element of the
    array a.

```

`T[] min(T[] a, T[] b)`

For built-in ordered types `T`, and arrays `a` and `b` of the same length, returns an array composed of the minimum of the corresponding elements of `a` and `b`.

`T[] max(T[] a, T[] b)`

For built-in ordered types `T`, and arrays `a` and `b` of the same length, returns an array composed of the maximum of the corresponding elements of `a` and `b`.

`pair[] fft(pair[] a, int sign=1)`

returns the Fast Fourier Transform of `a` (if the optional FFTW package is installed), using the given `sign`. Here is a simple example:

```
int n=4;
pair[] f=sequence(n);
write(f);
pair[] g=fft(f,-1);
write();
write(g);
f=fft(g,1);
write();
write(f/n);
```

`real[] tridiagonal(real[] a, real[] b, real[] c, real[] f);`

Solve the periodic tridiagonal problem $L^{-1} \mathbf{f}$, where \mathbf{f} is an n vector and L is the $n \times n$ matrix

```
[ b[0] c[0]          a[0]    ]
[ a[1] b[1] c[1]          ]
[      a[2] b[2] c[2]      ]
[              ...        ]
[ c[n-1]          a[n-1] b[n-1] ]
```

For Dirichlet boundary conditions (denoted here by `u[-1]` and `u[n]`), replace `f[0]` by `f[0]-a[0]u[-1]` and `f[n-1]-c[n-1]u[n]`; then set `a[0]=c[n-1]=0`.

`real[] solve(real[][] a, real[] b)`

Solve the linear equation $\mathbf{ax} = \mathbf{b}$ by LU decomposition, returning the solution \mathbf{x} , where `a` is an $n \times n$ matrix and `b` is an array of length n . For example:

```
import math;
real[][] a={{1,-2,3,0},{4,-5,6,2},{-7,-8,10,5},{1,50,1,-2}};
real[] b={7,19,33,3};
real[] x=solve(a,b);
write(a); write();
write(b); write();
write(x); write();
write(a*x);
```

If the matrix `a` is tridiagonal, the routine `tridiagonal` provides a more efficient algorithm (see [\[tridiagonal\]](#), page 58).

`real[][] solve(real[][] a, real[][] b)`

Solve the linear equation $\mathbf{ax} = \mathbf{b}$ returning the solution \mathbf{x} , where `a` is an $n \times n$ matrix and `b` is an $n \times m$ matrix.

```
real[] [] inverse(real[] [] a)
    Return the inverse of an  $n \times n$  matrix a.

real[] quadraticroots(real a, real b, real c);
    This numerically robust solver returns the real roots of the quadratic equation
     $ax^2 + bx + c = 0$ .

real[] cubicroots(real a, real b, real c, real d);
    This numerically robust solver returns the real roots of the cubic equation
     $ax^3 + bx^2 + cx + d = 0$ .

cputime cputime();
    Returns a structure cputime with cumulative CPU times broken down into the
    fields parent.user, parent.system, child.user, and child.system.
```

Asymptote includes a full set of vectorized array instructions for arithmetic (including self) and logical operations. These element-by-element instructions are implemented in C++ code for speed. Given

```
real[] a={1,2};
real[] b={3,2};
```

then **a == b** and **a >= 2** both evaluate to the vector **{false, true}**. To test whether all components of **a** and **b** agree, use the boolean function **all(a == b)**. One can also use conditionals like **(a >= 2) ? a : b**, which returns the array **{3,2}**, or **write((a >= 2) ? a : null)**, which returns the array **{2}**.

All of the standard built-in **libm** functions of signature **real(real)** also take a real array as an argument, effectively like an implicit call to **map**.

As with other built-in types, arrays of the basic data types can be read in by assignment. In this example, the code

```
file fin=input("test.txt");
real[] A=fin;
```

reads real values into **A** until the end of file is reached (or an I/O error occurs). If line mode is set with **line(file)**, then reading will stop once the end of the line is reached instead (line mode may be cleared with **line(file,false)**):

```
file fin=input("test.txt");
real[] A=line(fin);
```

Another useful mode is comma-separated-value mode, set with **csv(file)** and cleared with **csv(file,false)**, which skips over any comma delimiters:

```
file fin=input("test.txt");
real[] A=csv(fin);
```

To restrict the number of values read, use the **dimension(file,int)** function:

```
file fin=input("test.txt");
real[] A=dimension(fin,10);
```

This reads 10 values into **A**, unless end-of-file (or end-of-line in line mode) occurs first. Attempting to read beyond the end of the file will produce a runtime error message. Specifying a value of 0 for the integer limit is equivalent to the previous example of reading until end-of-file (or end-of-line in line mode) is encountered.

Two- and three-dimensional arrays of the basic data types can be read in like this:


```
file fin=input("test.txt");
real[] [] A=dimension(fin,2,3);
real[] [] [] B=dimension(fin,2,3,4);
```

Again, an integer limit of zero means no restriction.

Sometimes the array dimensions are stored with the data as integer fields at the beginning of an array. Such arrays can be read in with the functions `read1`, `read2`, and `read3`, respectively:

```
file fin=input("test.txt");
real[] A=read1(fin);
real[] [] B=read2(fin);
real[] [] [] C=read3(fin);
```

One, two, and three-dimensional arrays of the basic data types can be output with the functions `write(file,T[])`, `write(file,T[] [])`, `write(file,T[] [] [])`, respectively.

5.12 Casts

Asymptote implicitly casts `int` to `real`, `int` to `pair`, `real` to `pair`, `pair` to `path`, `pair` to `guide`, `path` to `guide`, `guide` to `path`, `real` to `pen`, `pair[]` to `guide[]`, `pair[]` to `path[]`, `path` to `path[]`, and `guide` to `path[]`, along with various three-dimensional casts defined in `three.asy`. Implicit casts are automatically attempted on assignment and when trying to match function calls with possible function signatures. Implicit casting can be inhibited by declaring individual arguments `explicit` in the function signature, say to avoid an ambiguous function call in the following example, which outputs 0:

```
int f(pair a) {return 0;}
int f(explicit real x) {return 1;}

write(f(0));
```

Other conversions, say `real` to `int` or `real` to `string`, require an explicit cast:

```
int i=(int) 2.5;
string s=(string) 2.5;

real[] a={2.5,-3.5};
int[] b=(int []) a;
write(stdout,b);      // Outputs 2,-3
```

Casting to user-defined types is also possible using `operator cast`:

```
struct rpair {
    public real radius;
    public real angle;
}

rpair operator init() {return new rpair;}

pair operator cast(rpair x) {
    return (x.radius*cos(x.angle),x.radius*sin(x.angle));
}
```

```

rpair x;
x.radius=1;
x.angle=pi/6;

write(x);           // Outputs (0.866025403784439,0.5)

```

One must use care when defining new cast operators. Suppose that in some code one wants all integers to represent multiples of 100. To convert them to reals, one would first want to multiply them by 100. However, the straightforward implementation

```
real operator cast(int x) {return x*100;}
```

is equivalent to an infinite recursion, since the result `x*100` needs itself to be cast from an integer to a real. Instead, we want to use the standard conversion of int to real:

```
real convert(int x) {return x*100;}
real operator cast(int x)=convert;
```

Explicit casts are implemented similarly, with `operator ecast`.

5.13 Import

While `Asymptote` provides many features by default, some applications require specialized features contained in external `Asymptote` modules. For instance, the lines

```
access graph;
graph.axes();
```

draw x and y axes on a two-dimensional graph. Here, the command looks up the module under the name `graph` in a global dictionary of modules and puts it in a new variable named `graph`. The module is a structure, and we can refer to its fields as we usually would with a structure.

Often, one wants to use module functions without having to specify the module name. The code

```
from graph access axes;
```

adds the `axes` field of `graph` into the local name space, so that subsequently, one can just write `axes()`. If the given name is overloaded, all types and variables of that name are added. To add more than one name, just use a comma-separated list:

```
from graph access axes, xaxis, yaxis;
```

Wild card notation can be used to add all non-private fields and types of a module to the local name space:

```
from graph access *;
```

Similarly, one can add the non-private fields and types of a structure to the local environment with the `unravel` keyword:

```
struct matrix {
    real a,b,c,d;
}
```

```
real det(matrix m) {
    unravel m;
```

```
    return a*d-b*c;
}
```

Alternatively, one can unravel selective fields:

```
real det(matrix m) {
    from m unravel a,b,c as C,d;
    return a*d-b*C;
}
```

The command

```
import graph;
```

is a convenient abbreviation for the commands

```
access graph;
unravel graph;
```

That is, `import graph` first loads a module into a structure called `graph` and then adds its non-private fields and types to the local environment. This way, if a member variable (or function) is overwritten with a local variable (or function of the same signature), the original one can still be accessed by qualifying it with the module name.

Wild card importing will work fine in most cases, but one does not usually know all of the internal types and variables of a module, which can also change as the module writer adds or changes features of the module. As such, it is prudent to add `import` commands at the start of an `Asymptote` file, so that imported names won't shadow locally defined functions. Still, imported names may shadow other imported names, depending on the order in which they were imported, and imported functions may cause overloading resolution problems if they have the same name as local functions defined later.

To rename modules or fields when adding them to the local environment, use `as`:

```
access graph as graph2d;
from graph access xaxis as xline, yaxis as yline;
```

The command

```
import graph as graph2d;
```

is a convenient abbreviation for the commands

```
access graph as graph2d;
unravel graph2d;
```

Except for a few built-in modules, such as `settings`, all modules are implemented as `Asymptote` files. When looking up a module that has not yet been loaded, `Asymptote` searches the standard search paths (see [Section 2.5 \[Search paths\]](#), page 4) for the matching file. The file corresponding to that name is read and the code within it is interpreted as the body of a structure defining the module.

If the file name contains nonalphanumeric characters, enclose it with quotation marks:

```
access "/usr/share/asymptote/graph.asy" as graph;
from "/usr/share/asymptote/graph.asy" access axes;
import "/usr/share/asymptote/graph.asy" as graph;
```

It is an error if modules import themselves (or each other in a cycle). The module name to be imported must be known at compile time.

However, you can import an `Asymptote` module determined by the string `s` at runtime like this:

```
eval("import "+s,true);
```

To conditionally execute an array of asy files, use

```
void asy(string format, bool overwrite ... string[] s);
```

The file will only be processed, using output format `format`, if `overwrite` is `true` or the output file is missing.

One can evaluate an `Asymptote` expression (without any return value, however) contained in the string `s` with:

```
void eval(string s, bool embedded=false);
```

It is not necessary to terminate the string `s` with a semicolon. If `embedded` is `true`, the string will be evaluated at the top level of the current environment. If `embedded` is `false` (the default), the string will be evaluated in an independent environment, sharing the same `settings` module (see [\[settings\]](#), page 123).

One can evaluate arbitrary `Asymptote` code (which may contain unescaped quotation marks) with the command

```
void eval(code s, bool embedded=false);
```

Here `code` is a special type used with quote `{}` to enclose `Asymptote` code like this:

```
real a=1;
code s=quote {
  write(a);
};
eval(s,true);          // Outputs 1
```

To include the contents of a file `graph` verbatim (as if the contents of the file were inserted at that point), use one of the forms:

```
include graph;
include "/usr/share/asymptote/graph.asy";
```

To list all global functions and variables defined in a module named by the contents of the string `s`, use the function

```
void list(string *s, bool imports=false);
```

Imported global functions and variables are also listed if `imports` is `true`.

5.14 Static

Static qualifiers allocate the memory address of a variable in a higher enclosing level.

For a function body, the variable is allocated in the block where the function is defined; so in the code

```
struct s {
  int count() {
    static int c=0;
    ++c;
    return c;
  }
}
```

```
}
```

there is one instance of the variable `c` for each object `s` (as opposed to each call of `count`).

Similarly, in

```
int factorial(int n) {
    int helper(int k) {
        static int x=1;
        x *= k;
        return k == 1 ? x : helper(k-1);
    }
    return helper(n);
}
```

there is one instance of `x` for every call to `factorial` (and not for every call to `helper`), so this is a correct, but ugly, implementation of factorial.

Similarly, a static variable declared within a structure is allocated in the block where the structure is defined. Thus,

```
struct A {
    struct B {
        static pair z;
    }
}
```

creates one object `z` for each object of type `A` created.

In this example,

```
int pow(int n, int k) {
    struct A {
        static int x=1;
        void helper() {
            x *= n;
        }
    }
    A operator init() {return new A;}
    for (int i=0; i < k; ++i) {
        A a;
        a.helper();
    }
    return A.x;
}
```

there is one instance of `x` for each call to `pow`, so this is an ugly implementation of exponentiation.

Statements may also be declared static, in which case they are run at the place where the enclosing function or structure is defined.

Declarations or statements not enclosed in a function or structure definition are already at the top level, so static modifiers are meaningless. A warning is given in such a case.

6 LaTeX usage

Asymptote comes with a convenient LaTeX style file `asymptote.sty` that makes LaTeX Asymptote-aware. Entering Asymptote code directly into the LaTeX source file, at the point where it is needed, keeps figures organized and avoids the need to invent new file names for each figure. Simply add the line `\usepackage{asymptote}` at the beginning of your file and enclose your Asymptote code within a `\begin{asy}...\end{asy}` environment. As with the LaTeX `comment` environment, the `\end{asy}` command must appear on a line by itself, with no leading spaces or trailing commands/comments.

The sample LaTeX file below, named `latexusage.tex`, can be run as follows:

```
latex latexusage
asy latexusage
latex latexusage

or

pdflatex latexusage
asy latexusage
pdflatex latexusage
```

If the `inline` option is given to the `asymptote.sty` package, inline LaTeX code is generated instead of EPS files. This makes LaTeX symbols visible to the `\begin{asy}...\end{asy}` environment. In this mode, Asymptote correctly aligns LaTeX symbols defined outside of `\begin{asy}...\end{asy}`, but treats their size as zero; an optional second string can be given to `Label` to provide an estimate of the unknown label size.

Note that if `latex` is used with the `inline` option, the labels might not show up in DVI viewers that cannot handle raw PostScript code. One can use `dvips/dvipdf` to produce PostScript/PDF output (we recommend using the modified version of `dvipdf` in the Asymptote patches directory, which accepts the `dvips -z hyperdvi` option).

An excellent tutorial by Dario Teixeira on integrating Asymptote and LaTeX is available at <http://dario.dse.nl/projects/asylatex/>.

Here now is `latexusage.tex`:

```
\documentclass[12pt]{article}

% Use this form with to include eps (latex) or pdf (pdflatex) files:
\usepackage{asymptote}

% Use this form with latex or pdflatex to include inline LaTeX code:
%\usepackage[inline]{asymptote}

% Enable this line to produce pdf hyperlinks with latex:
%\usepackage[hypertex]{hyperref}

% Enable this line to produce pdf hyperlinks with pdftex:
%\usepackage[pdftex]{hyperref}

\begin{document}
```

```

\begin{asydef}
// Global definitions can be put here.
\end{asydef}

Here is a venn diagram
%(Figure~\ref{venn})
produced with Asymptote, drawn to width 5cm:

\def\A{A}
\def\B{B}

%\begin{figure}
\begin{center}
\begin{asy}
size(5cm,0);
pen colour1=red;
pen colour2=green;

pair z0=(0,0);
pair z1=(-1,0);
pair z2=(1,0);
real r=1.5;
guide c1=circle(z1,r);
guide c2=circle(z2,r);
fill(c1,colour1);
fill(c2,colour2);

picture intersection=new picture;
fill(intersection,c1,colour1+colour2);
clip(intersection,c2);

add(intersection);

draw(c1);
draw(c2);

//box(Label("\A",z1));           // Requires [inline] package option.
//box(Label("\B",z2));           // Requires [inline] package option.
box(Label("\A",z1));
box(Label("\B",z2));

pair z=(0,-2);
real m=3;
margin BigMargin=Margin(0,m*dot(unit(z1-z),unit(z0-z)));

draw(Label("\A\cap B",0),conj(z)--z0,Arrow,BigMargin);
draw(Label("\A\cup B",0),z--z0,Arrow,BigMargin);

```

```

draw(z--z1,Arrow,Margin(0,m));
draw(z--z2,Arrow,Margin(0,m));

shipout(bbox(0.25cm));
\end{asy}
%\caption{Venn diagram}\label{venn}
\end{center}
%\end{figure}

```

Each graph is drawn in its own environment. One can specify the width and height to `\LaTeX` explicitly:

```

\begin{center}
\begin{asy}[3cm,0]
guide center = (0,1){W}..tension 0.8..(0,0){(1,-.5)}..tension 0.8..{W}(0,-1);

draw((0,1)..(-1,0)..(0,-1));
filldraw(center{E}..{N}(1,0)..{W}cycle);
fill(circle((0,0.5),0.125),white);
fill(circle((0,-0.5),0.125));
\end{asy}
\end{center}

```

The default width is the full line width:

```

\begin{center}
\begin{asy}
import graph;

real f(real x) {return sqrt(x);}
pair F(real x) {return (x,f(x));}

real g(real x) {return -sqrt(x);}
pair G(real x) {return (x,g(x));}

guide p=(0,0)--graph(f,0,1,operator ..)--(1,0);
fill(p--cycle,lightgray);
draw(p);
draw((0,0)--graph(g,0,1,operator ..)--(1,0),dotted);

real x=0.5;
pair c=(4,0);

transform T=xscale(0.5);
draw((2.695,0),T*arc(0,0.30cm,20,340),ArcArrow);
fill(shift(c)*T*circle(0,-f(x)),red+white);
draw(F(x)--c+(0,f(x)),dashed+red);

```



```

draw(G(x)--c+(0,g(x)),dashed+red);

dot(Label,(1,1));
arrow("$y=\sqrt{x}$",F(0.7),N);

arrow((3,0.5*f(x)),W,1cm,red);
arrow((3,-0.5*f(x)),W,1cm,red);


xaxis("$x$",0,c.x,dashed);
yaxis("$y$");

draw("$r$",(x,0)--F(x),E,red,Arrows,BeginBar,PenMargins);
draw("$r$",(x,0)--G(x),E,red,Arrows,PenMargins);
draw("$r$",c--c+(0,f(x)),Arrow,PenMargin);
dot(c);
\end{asy}
\end{center}


\end{document}

```


Here is a venn diagram produced with Asymptote, drawn to width 5cm:



Each graph is drawn in its own environment. One can specify the width and height to L^AT_EX explicitly:



The default width is the full line width:



7 Base modules

Asymptote currently ships with the following base modules:

7.1 plain

This is the default **Asymptote** base file, which defines key parts of the drawing language (such as the **picture** structure).

By default, an implicit **private import plain;** occurs before translating a file and before the first command given in interactive mode. This also applies when translating files for module definitions (except when translating **plain**, of course). This means that the types and functions defined in **plain** are accessible in almost all **Asymptote** code. Use the **-noautoplain** command-line option to disable this feature.

7.2 simplex

This package solves the two-variable linear programming problem using the simplex method. It is used by the module **plain** for automatic sizing of pictures.

7.3 math

This package extends **Asymptote**'s mathematical capabilities with point-in-polygon and intersection algorithms and matrix arithmetic.

Unlike **MetaPost**, **Asymptote** does not implicitly solve linear equations and therefore does not have the notion of a **whatever** unknown. Use the built-in explicit linear equation solver **solve** instead. The following routine provides a useful replacement for a common use of **whatever**: finding the intersection point of the lines through **P**, **Q** and **p**, **q**, respectively:

```
@item pair extension(pair P, pair Q, pair p, pair q);
```

Return the intersection point of the extensions of the line segments **PQ** and **pq**.

Here are some additional routines provided in the **math** package:

```
void drawline(picture pic=currentpicture, pair P, pair Q, pen p=currentpen);
    draw the visible portion of the (infinite) line going through P and Q, without
    altering the size of picture pic, using pen p.
```

```
real intersect(triple P, triple Q, triple n, triple Z);
    Return the intersection time of the extension of the line segment PQ with the
    plane perpendicular to n and passing through Z.
```

```
triple intersectionpoint(triple n0, triple P0, triple n1, triple P1);
    Return any point on the intersection of the two planes with normals n0 and
    n1 passing through points P0 and P1, respectively. If the planes are parallel,
    return (infinity,infinity,infinity).
```

```
bool straight(path p)
    returns true iff the path p is straight.
```

7.4 geometry

This module provides the beginnings of a geometry package. It currently includes a triangle structure and functions to draw interior arcs of triangles and perpendicular symbols.

7.5 stats

This package implements a Gaussian random number generator and a collection of statistics routines, including `histogram` and `leastsquares`.

7.6 patterns

This package implements `Postscript` tiling patterns and includes several convenient pattern generation routines.

7.7 tree

This package implements an example of a dynamic binary search tree.

7.8 drawtree

This is a simple tree drawing module used by the example `treetest.asy`.

7.9 feynman

This package, contributed by Martin Wiebusch, is useful for drawing Feynman diagrams, as illustrated by the examples `eetomumu.asy` and `fermi.asy`.

7.10 roundedpath

This package, contributed by Stefan Knorr, is useful for rounding the sharp corners of paths, as illustrated in the example file `roundpath.asy`.

7.11 slide

This package provides a simple yet high-quality facility for making presentation slides. An example is provided in the file `slidedemo.asy`:

```
// Slide demo.
// Command-line options to enable stepping and/or reverse video:
// asy [-u stepping=true] [-u reverse=true] slidedemo

orientation=Landscape;

// Generated needed files if they don't already exist.
asy(nativeformat(),"Pythagoras","log","near_earth");

import slide;

usersetting();

// Optional background color:
// import x11colors;
// fill(background,box((-1,-1),(1,1)),Azure);
```

```

titlepage("Slides with {\tt Asymptote}: A Demo","John C. Bowman",
          "University of Alberta","\today","http://asymptote.sf.net");

outline();
item("item");
subitem("subitem");
remark("remark");
item("draw");
item("figure");

title("Items");
item("First item.");
subitem("First subitem.");
subitem("Second subitem.");
item("Second item.");
equation("a^2+b^2=c^2.");
equations("\frac{\sin^2\theta+\cos^2\theta}{\cos^2\theta}
&=&\frac{1}{\cos^2\theta}\nonumber\\
&=&\sec^2\theta.");
remark("A remark.");
item("To enable pausing between bullets:");
remark("{\tt asy -u stepping=true}");
item("To enable reverse video:");
remark("{\tt asy -u reverse=true}");

title("Can draw on a slide, preserving the aspect ratio:");
picture pic,pic2;
draw(pic,unitcircle);
add(pic.fit(15cm));
step();
fill(pic2,unitcircle,paleblue);
add(pic2.fit(15cm));

newslide();
item("The slide \Red{title} \Green{can} \Blue{be} omitted.");
figure("Pythagoras","height=12cm",
      "A simple proof of Pythagoras' Theorem.");

newslide();
item("Single skip:");
skip();
item("Double skip:");
skip(2);
figure(new string[] {"log."+nativeformat(),"near_earth."+nativeformat()},
      "width=10cm","Examples of {\tt Asymptote} graphs.");

```

7.12 MetaPost

This package provides some useful routines to help **MetaPost** users migrate old **MetaPost** code to **Asymptote**.

7.13 unicode

Import this package at the beginning of the file to instruct LaTeX to accept **unicode** (UTF-8) standardized international characters. You will also need to set up LaTeX support for **unicode** by unpacking in your LaTeX source directory (e.g. `/usr/share/texmf/tex/latex`) the file

<http://www.unruh.de/DniQ/latex/unicode/unicode.tgz>

and then running the command

texhash

To use Cyrillic fonts, you will need to change the font encoding:

```
import unicode;
texpreamble("\usepackage{mathtext}\usepackage[russian]{babel}");
defaultpen(font("T2A","cmr"));
```

Support for Chinese, Japanese, and Korean fonts is provided by the CJK package:

<http://www.tug.org/tex-archive/languages/chinese/CJK/>

The following commands enable the CJK song family (within a label, you can also temporarily switch to another family, say **kai**, by prepending `"\CJKfamily{kai}"` to the label string):

```
texpreamble("\usepackage{CJK}
\AtBeginDocument{\begin{CJK*}{GBK}{song}}
\AtEndDocument{\clearpage\end{CJK*}}");
```

7.14 latin1

If you don't have LaTeX support for **unicode** installed, you can enable support for Western European languages (ISO 8859-1) by importing the module **latin1**. This module can be used as a template for providing support for other ISO 8859 alphabets.

7.15 babel

This module implements the LaTeX **babel** package in **Asymptote**. For example:

```
import babel;
babel("german");
```

7.16 embed

This module provides an interface to the LaTeX package (included with **MikTeX**)

<http://www.tug.org/tex-archive/macros/latex/contrib/movie15>

for embedding movies, sounds, and 3D objects into a PDF document. Note that Adobe unfortunately has not yet made available the required plugins for the Linux version of Adobe Reader.

An example of this interface is provided in the file `embeddedmovie.asy` in the `animations` subdirectory of the `examples` directory.

7.17 `annotate`

This module supports PDF annotations for viewing with Adobe Reader, via the function

```
void annotate(picture pic=currentpicture, string title, string text,
             pair position);
```

Annotations are illustrated in the example file `annotation.asy`. Currently, annotations are only implemented for the `latex` (default) and `tex` \TeX engines.

7.18 `CAD`

This package, contributed by Mark Henning, provides basic pen definitions and measurement functions for simple 2D CAD drawings according to DIN 15. It is documented separately, in the file `CAD.pdf`.

7.19 `graph`

This package implements two-dimensional linear and logarithmic graphs, including automatic scale and tick selection (with the ability to override manually). A graph is a `guide` (that can be drawn with the `draw` command, with an optional legend) constructed with one of the following routines:

-

```
guide graph(picture pic=currentpicture, real f(real), real a, real b,
            int n=ngraph, interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` (see [\[automatic scaling\]](#), page 86) of the function `f` on the interval `[a,b]`, sampling at `n` evenly spaced points, with one of these interpolation types:

- `operator --` (linear interpolation; the abbreviation `Straight` is also accepted)
- `operator ..` (piecewise Bezier cubic spline interpolation; the abbreviation `Spline` is also accepted)

-

```
guide graph(picture pic=currentpicture, real x(real), real y(real),
            real a, real b, int n=ngraph,
            interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` of the parametrized function $(x(t), y(t))$ for t in `[a,b]`, sampling at `n` evenly spaced points, with the given interpolation type.

-

```
guide graph(picture pic=currentpicture, pair z(real), real a, real b,
            int n=ngraph, interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` of the parametrized function $z(t)$ for t in `[a,b]`, sampling at `n` evenly spaced points, with the given interpolation type.

- ```
guide graph(picture pic=currentpicture, pair[] z, bool[] cond={},
 interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` of those elements of the array `z` for which the corresponding elements of the boolean array `cond` are `true`, with the given interpolation type.
- ```
guide graph(picture pic=currentpicture, real[] x, real[] y,
             bool[] cond={}, interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` of those elements of the arrays `(x,y)` for which the corresponding elements of the boolean array `cond` are `true`, with the given interpolation type.
- ```
guide graph(picture pic=currentpicture, real f(real), real a, real b,
 int n=ngraph, real T(real), interpolate join=operator --);
```

Returns a graph using the scaling information for picture `pic` of the function `f` on the interval `[T(a),T(b)]`, sampling at `n` points evenly spaced in `[a,b]`, with the given interpolation type. These parametric versions also accept an arbitrary sampling function:

```
guide graph(picture pic=currentpicture, real x(real), real y(real),
 real a, real b, int n=ngraph, real T(real),
 interpolate join=operator --);
```

```
guide graph(picture pic=currentpicture, pair z(real), real a, real b,
 int n=ngraph, real T(real), interpolate join=operator --)
```
- ```
guide polargraph(picture pic=currentpicture, real f(real), real a,
                  real b, int n=ngraph, interpolate join=operator --);
```

Returns a polar-coordinate graph using the scaling information for picture `pic` of the function `f` on the interval `[a,b]`, sampling at `n` evenly spaced points, with the given interpolation type.

An axis can be drawn on a picture with one of the following commands:

- ```
void xaxis(picture pic=currentpicture, Label L="", axis axis=YZero,
 real xmin=-infinity, real xmax=infinity, pen p=currentpen,
 ticks ticks=NoTicks, arrowbar arrow=None, bool put=Below);
```

Draw an  $x$  axis on picture `pic` from  $x=xmin$  to  $x=xmax$  using pen `p`, optionally labelling it with Label `L`. The relative label location along the axis (a real number from `[0,1]`) defaults to 1 (see [\[Label\]](#), page 16), so that the label is drawn at the end of the axis. An infinite value of `xmin` or `xmax` specifies that the corresponding axis limit will be automatically determined from the picture limits. The optional `arrow` argument takes the same values as in the `draw` command (see [\[arrows\]](#), page 13). If `put=Below` and the `extend` flag for `axis` is `false`, the axis is drawn before any existing objects in the current picture. The axis placement is determined by one of the following axis types:



```

YZZero(bool extend=true)
 Request an x axis at $y=0$ (or $y=1$ on a logarithmic axis) extending to the
 full dimensions of the picture, unless extend=false.

YEquals(real Y, bool extend=true)
 Request an x axis at $y=Y$ extending to the full dimensions of the picture,
 unless extend=false.

Bottom(bool extend=false)
 Request a bottom axis.

Top(bool extend=false)
 Request a top axis.

BottomTop(bool extend=false)
 Request a bottom and top axis.

```

Custom axis types can be created by following the examples in `graph.asy`. One can easily override the default values for the standard axis types:

```

import graph;

YZZero=new axis(bool extend=true) {
 return new void(picture pic, axisT axis) {
 real y=pic.scale.x.scale.logarithmic ? 1 : 0;
 axis.value=I*pic.scale.y.T(y);
 axis.position=1;
 axis.side=right;
 axis.align=2.5E;
 axis.value2=Infinity;
 axis.extend=extend;
 };
};
YZZero=YZZero();

```

The default tick option is `NoTicks`. The options `LeftTicks`, `RightTicks`, or `Ticks` can be used to draw ticks on the left, right, or both sides of the path, relative to the direction in which the path is drawn. These tick routines accept a number of optional arguments:

```

ticks LeftTicks(Label format="", ticklabel ticklabel=null,
 bool beginlabel=true, bool endlabel=true,
 int N=0, int n=0, real Step=0, real step=0,
 bool begin=true, bool end=true,
 real Size=0, real size=0, bool extend=false,
 pen pTick=nullpen, pen ptick=nullpen);

```

If any of these parameters are omitted, reasonable defaults will be chosen:

**Label format**  
 override the default tick label format (`defaultformat`, initially `"$%.4g$"`), rotation, pen, and alignment (for example, `LeftSide`, `Center`,

or `RightSide`) relative to the axis. To enable LaTeX math mode fonts, the format string should begin and end with `$` see [\[format\]](#), page 24; if the format string is `"%`, the tick label will be suppressed;

```

ticklabel
 is a function string(real x) returning the label (by default,
 format(format.s,x)) for each tick value x;

bool beginlabel
 include the first label;

bool endlabel
 include the last label;

int N
 when automatic scaling is enabled (the default; see \[automatic scaling\],
 page 86), divide the values evenly into this many intervals, separated by
 big ticks;

int n
 divide each value interval into this many subintervals, separated by small
 ticks;

real Step
 the tick value spacing between big ticks (if N=0);

real step
 the tick value spacing between small ticks (if n=0);

bool begin
 include the first big tick;

bool end
 include the last big tick;

real Size
 the size of the big ticks (in PostScript coordinates);

real size
 the size of the small ticks (in PostScript coordinates);

bool extend;
 extend the ticks between two axes (useful for drawing a grid on the graph);

pen pTick
 an optional pen used to draw the big ticks;

pen ptick
 an optional pen used to draw the small ticks.

```

It is also possible to specify custom tick locations with `LeftTicks`, `RightTicks`, and `Ticks` by passing explicit real arrays `Ticks` and (optionally) `ticks` containing the locations of the big and small ticks, respectively:

```

ticks LeftTicks(Label format="", ticklabel ticklabel=null,
 bool beginlabel=true, bool endlabel=true,
 real[] Ticks, real[] ticks=new real[],
 real Size=0, real size=0, bool extend=false,
 pen pTick=nullpen, pen ptick=nullpen)

```

•

```

void yaxis(picture pic=currentpicture, Label L="", axis axis=XZero,
 real ymin=-infinity, real ymax=infinity, pen p=currentpen,
 ticks ticks=NoTicks, arrowbar arrow=None, bool put=Below);

```

Draw a  $y$  axis on picture `pic` from  $y=ymin$  to  $y=ymax$  using pen `p`, optionally labelling it with Label `L`. The relative location of the label (a real number from `[0,1]`) defaults to 1

(see [Label], page 16). An infinite value of `ymin` or `ymax` specifies that the corresponding axis limit will be automatically determined from the picture limits. The optional `arrow` argument takes the same values as in the `draw` command (see [arrows], page 13). If `put=Below` and the `extend` flag for `axis` is `false`, the axis is drawn before any existing objects in the current picture. The tick type is specified by `ticks` and the axis placement is determined by one of the following `axis` types:

`XZero(bool extend=true)`

Request a  $y$  axis at  $x=0$  (or  $x=1$  on a logarithmic axis) extending to the full dimensions of the picture, unless `extend=false`.

`XEquals(real X, bool extend=true)`

Request a  $y$  axis at  $x=X$  extending to the full dimensions of the picture, unless `extend=false`.

`Left(bool extend=false)`

Request a left axis.

`Right(bool extend=false)`

Request a right axis.

`LeftRight(bool extend=false)`

Request a left and right axis.

- For convenience, the functions

```
void xequals(picture pic=currentpicture, Label L="", real x,
 bool extend=false, real ymin=-infinity, real ymax=infinity,
 pen p=currentpen, ticks ticks=NoTicks, bool put=Above,
 arrowbar arrow=None);
```

and

```
void yequals(picture pic=currentpicture, Label L="", real y,
 bool extend=false, real xmin=-infinity, real xmax=infinity,
 pen p=currentpen, ticks ticks=NoTicks, bool put=Above,
 arrowbar arrow=None);
```

can be respectively used to call `yaxis` and `xaxis` with the appropriate axis types `XEquals(x,extend)` and `YEquals(y,extend)`. This is the recommended way of drawing vertical or horizontal lines and axes at arbitrary locations.

- 

```
void axis(picture pic=currentpicture, Label L="", guide g,
 pen p=currentpen, ticks ticks, ticklocate locate,
 arrowbar arrow=None, int[] divisor=new int[],
 bool put=Above, bool opposite=false);
```

This routine can be used to draw on picture `pic` a general axis based on an arbitrary path `g`, using pen `p`. One can optionally label the axis with Label `L` and add an arrow `arrow`. The tick type is given by `ticks`. The optional integer array `divisor` specifies what tick divisors to try in the attempt to produce uncrowded tick labels. A `true` value for the flag `opposite` identifies an unlabelled secondary axis (typically drawn opposite a primary axis). The axis is drawn on top of any existing objects in the current picture only if `put` is `Above`. The tick locator `ticklocate` is constructed by the routine

```

ticklocate ticklocate(real a, real b, autoscaleT S=defaultS,
 real tickmin=-infinity, real tickmax=infinity,
 real time(real)=null, pair dir(real)=zero);

```

where  $a$  and  $b$  specify the respective tick values at  $\text{point}(g,0)$  and  $\text{point}(g,\text{length}(g))$ ,  $S$  specifies the autoscaling transformation, the function  $\text{real time}(\text{real } v)$  returns the time corresponding to the value  $v$ , and  $\text{pair dir}(\text{real } t)$  returns the absolute tick direction as a function of  $t$  (zero means draw the tick perpendicular to the axis).

- These routines are useful for manually putting ticks and labels on axes (if the special variable `Label` is given as the `Label` argument, the `format` argument will be used to format a string based on the tick location):

```

void xtick(picture pic=currentpicture, Label L="", explicit pair z,
 pair dir=N, string format="",
 real size=Ticksize, pen p=currentpen);
void xtick(picture pic=currentpicture, Label L="", real x,
 pair dir=N, string format="",
 real size=Ticksize, pen p=currentpen);
void ytick(picture pic=currentpicture, Label L="", explicit pair z,
 pair dir=E, string format="",
 real size=Ticksize, pen p=currentpen);
void ytick(picture pic=currentpicture, Label L="", real y,
 pair dir=E, string format="",
 real size=Ticksize, pen p=currentpen);
void tick(picture pic=currentpicture, pair z,
 pair dir, real size=Ticksize, pen p=currentpen);
void labelx(picture pic=currentpicture, Label L="", explicit pair z,
 align align=S, string format="", pen p=nullpen);
void labelx(picture pic=currentpicture, Label L="", real x,
 align align=S, string format="", pen p=nullpen);
void labelx(picture pic=currentpicture, Label L,
 string format="", explicit pen p=currentpen);
void labely(picture pic=currentpicture, Label L="", explicit pair z,
 align align=W, string format="", pen p=nullpen);
void labely(picture pic=currentpicture, Label L="", real y,
 align align=W, string format="", pen p=nullpen);
void labely(picture pic=currentpicture, Label L,
 string format="", explicit pen p=nullpen);

```

Here are some simple examples of two-dimensional graphs:

1. This example draws a textbook-style graph of  $y = \exp(x)$ , with the  $y$  axis starting at  $y = 0$ :

```

import graph;
size(150,0);

real f(real x) {return exp(x);}
pair F(real x) {return (x,f(x));}

```


```

xaxis("x");
yaxis("y",0);

draw(graph(f,-4,2,operator ..),red);

labely(1,E);
label("e^x",F(1),SE);

```



2. The next example draws a scientific-style graph with a legend. The position of the legend can be adjusted either explicitly or by using the graphical user interface `xasy` (see [Chapter 10 \[GUI\]](#), page 126). If an `UnFill(real xmargin=0, real ymargin=xmargin)` or `Fill(pen)` option is specified to `add`, the legend will obscure any underlying objects. Here we illustrate how to clip the portion of the picture covered by a label:

```

import graph;

size(400,200,IgnoreAspect);


real Sin(real t) {return sin(2pi*t);}
real Cos(real t) {return cos(2pi*t);}

draw(graph(Sin,0,1),red,"$\sin(2\pi x)$");
draw(graph(Cos,0,1),blue,"$\cos(2\pi x)$");

xaxis("x",BottomTop,LeftTicks);
yaxis("y",LeftRight,RightTicks);

label("LABEL",point(0),UnFill(1mm));

```



To specify a fixed size for the graph proper, use `attach`:

```
import graph;

size(250,200,IgnoreAspect);

real Sin(real t) {return sin(2pi*t);}
real Cos(real t) {return cos(2pi*t);}

draw(graph(Sin,0,1),red,"$\sin(2\pi x)$");
draw(graph(Cos,0,1),blue,"$\cos(2\pi x)$");

xaxis("x",BottomTop,LeftTicks);
yaxis("y",LeftRight,RightTicks);

label("LABEL",point(0),UnFill(1mm));

attach(legend(),point(E),20E,UnFill);
```



A legend can have multiple entries per line:

```
import graph;
size(8cm,6cm,IgnoreAspect);

typedef real realfcn(real);
realfcn F(real p) {
 return new real(real x) {return sin(p*x);};
};

for(int i=1; i < 5; ++i)
 draw(graph(F(i*pi),0,1),Pen(i),
 "$\sin("+i == 1 ? "" : (string) i)+"\pi x$");
xaxis("x",BottomTop,LeftTicks);
yaxis("y",LeftRight,RightTicks);

attach(legend(2),(point(S).x,truepoint(S).y),10S,UnFill);
```



3. This example draws a graph of one array versus another (both of the same size) using custom tick locations and a smaller font size for the tick labels on the  $y$  axis.


```
import graph;

size(200,150,IgnoreAspect);

real[] x={0,1,2,3};
real[] y=x^2;

draw(graph(x,y),red,MarkFill[0]);
```

```
xaxis("x",BottomTop,LeftTicks);
yaxis("y",LeftRight,RightTicks(Label(fontsize(8)),new real[] {0,4,9}));
```



4. This example shows how to label an axis with arbitrary strings.

```
import graph;

size(400,150,IgnoreAspect);


real[] x=sequence(12);
real[] y=sin(2pi*x/12);

scale(false);

string[] month={"Jan","Feb","Mar","Apr","May","Jun",
 "Jul","Aug","Sep","Oct","Nov","Dec"};

draw(graph(x,y),red,MarkFill[0]);

xaxis(BottomTop,LeftTicks(new string(real x) {
 return month[round(x % 12)];}));
yaxis("y",LeftRight,RightTicks(4));
```





5. The next example draws two graphs of an array of coordinate pairs, using frame alignment and data markers. In the left-hand graph, the markers, constructed with

```
marker marker(path g, markroutine markroutine=marknodes,
 pen p=currentpen, filltype filltype=NoFill,
 bool put=Above);
```

using the path `unitcircle` (see [\[filltype\]](#), page 39), are drawn below each node. Any frame can be converted to a marker, using

```
marker marker(frame f, markroutine markroutine=marknodes,
 bool put=Above);
```

In the right-hand graph, the unit  $n$ -sided regular polygon `polygon(int n)` and the unit  $n$ -point cross `cross(int n)` are used to build a custom marker frame. Here `markuniform(int n)` adds this frame at  $n$  uniformly spaced points along the arclength of the path. This example also illustrates the `errorbar` routines:

```
void errorbars(picture pic=currentpicture, pair[] z, pair[] dp,
 pair[] dm={}, bool[] cond={}, pen p=currentpen,
 real size=0);
```

```
void errorbars(picture pic=currentpicture, real[] x, real[] y,
 real[] dpx, real[] dpy, real[] dmxd={}, real[] dmyd={},
 bool[] cond={}, pen p=currentpen, real size=0);
```

Here, the positive and negative extents of the error are given by the absolute values of the elements of the pair array `dp` and the optional pair array `dm`. If `dm` is not specified, the positive and negative extents of the error are assumed to be equal.

```
import graph;
```

```
picture pic;
real xsize=200, ysize=140;
size(pic,xsize,ysize,IgnoreAspect);
```

```
pair[] f={(5,5),(50,20),(90,90)};
pair[] df={(0,0),(5,7),(0,5)};
```

```
errorbars(pic,f,df,red);
draw(pic,graph(pic,f),"legend",
 marker(scale(0.8mm)*unitcircle,red,FillDraw(blue),Below));
```

```
xaxis(pic,"x",BottomTop,LeftTicks);
yaxis(pic,"y",LeftRight,RightTicks);
add(pic,legend(pic),point(pic,NW),20SE,UnFill);
```

```
picture pic2;
size(pic2,xsize,ysize,IgnoreAspect);
```

```
frame mark;
filldraw(mark,scale(0.8mm)*polygon(6),green,green);
```

```

draw(mark,scale(0.8mm)*cross(6),blue);

draw(pic2,graph(pic2,f),marker(mark,markuniform(5)));



axis(pic2,"x",BottomTop,LeftTicks);
axis(pic2,"y",LeftRight,RightTicks);

yequals(pic2,55.0,red+Dotted);
xequals(pic2,70.0,red+Dotted);

// Fit pic to W of origin:
add(pic.fit()),(0,0),W);

// Fit pic2 to E of (5mm,0):
add(pic2.fit()),(5mm,0),E);

```



6. This example draws a graph of a parametrized curve.

The calls to

```

xlimits(picture pic=currentpicture, real min=-infinity,
 real max=infinity, bool crop=NoCrop);

```

and the analogous function `ylimits` can be uncommented to set the respective axes limits for picture `pic` to the specified `min` and `max` values (alternatively, the function `limits(pair, pair)` can be used to limit the axes to the box having opposite vertices at the given pairs). Existing objects in picture `pic` will be cropped to lie within the given limits if `crop=Crop`. The function `crop(picture pic)` is equivalent to calling both `xlimits(Crop)` and `ylimits(Crop)`.

```
import graph;
```

```
size(0,200);
```


```
real x(real t) {return cos(2pi*t);}
```

```
real y(real t) {return sin(2pi*t);}
```

```
draw(graph(x,y,0,1));
```

```
//xlimits(0,1,Crop);
//ylimits(-1,0,Crop);

axis("x",BottomTop,LeftTicks("$%#.1f$"));
yaxis("y",LeftRight,RightTicks("$%#.1f$"));
```



Axis scaling can be requested and/or automatic selection of the axis limits can be inhibited with one of these `scale` routines:

```
void scale(picture pic=currentpicture, scaleT x, scaleT y);
```

```
void scale(picture pic=currentpicture, bool xautoscale=true,
 bool yautoscale=xautoscale, bool zautoscale=yautoscale);
```

This sets the scalings for picture `pic`. The `graph` routines accept an optional `picture` argument for determining the appropriate scalings to use; if none is given, it uses those set for `currentpicture`.

Two frequently used scaling routines `Linear` and `Log` are predefined in `graph`.

All picture coordinates (including those in paths and those given to the `label` and `limits` functions) are always treated as linear (post-scaled) coordinates. Use

```
pair Scale(picture pic=currentpicture, pair z);
```

to convert a graph coordinate into a scaled picture coordinate.

The predefined scaling routines can be given two optional boolean arguments: `automin` and `automax`. These default to `true`, but can be respectively set to `false` to disable automatic selection of "nice" axis minimum and maximum values. `Linear` can also take as an optional final argument a multiplicative scaling factor (e.g. for a depth axis, `Linear(-1)` requests axis reversal).

For example, to draw a log/log graph of a function, use `scale(Log,Log)`:

```

import graph;

size(200,200,IgnoreAspect);

real f(real t) {return 1/t;}

scale(Log,Log);


draw(graph(f,0.1,10));

//xlimits(1,10);
//ylimits(0.1,1);

dot(Scale((3,5)));
label("(3,5)",Scale((3,5)),S);

axis("x",BottomTop,LeftTicks);
axis("y",LeftRight,RightTicks);

```



By extending the ticks, one can easily produce a logarithmic grid:

```


import graph;
size(200,200,IgnoreAspect);

real f(real t) {return 1/t;}

scale(Log,Log);
draw(graph(f,0.1,10),red);
pen thin=linewidth(0.5*linewidth());
axis("x",BottomTop,LeftTicks(begin=false,end=false,extend=true,
 ptick=thin));

```


```
yaxis("y",LeftRight,RightTicks(begin=false,end=false,extend=true,
 ptick=thin));
```



One can also specify custom tick locations and formats for logarithmic axes:

```
import graph;

size(300,175,IgnoreAspect);
scale(Log,Log);
draw(graph(identity,5,20));
xlimits(5,20);
ylimits(1,100);
xaxis("M/M_{\odot}",BottomTop,LeftTicks(new real[] {6,10,12,14,16,18}));
yaxis("$\nu_{\rm up}$ [Hz]",LeftRight,
 RightTicks(new string(real x){return format(pow10(x));}));
```



It is easy to draw logarithmic graphs with respect to other bases:

```
import graph;
size(200,IgnoreAspect);

// Base-2 logarithmic scale on y-axis:
scaleT yscale;


real log2(real x) {static real log2=log(2); return log(x)/log2;}
real pow2(real x) {return 2^x;}

yscale.init(log2,pow2,logarithmic=true);
scale(Linear,yscale);

real f(real x) {return 1+x^2;}

draw(graph(f,-4,4));

xaxis("x",xmin=-5,xmax=5,LeftTicks,EndArrow);
yaxis("y",ymin=1,ymax=f(5),RightTicks,EndArrow);
```



Here is an example of a "broken" linear  $x$  axis that omits the segment  $[3,8]$ :

```
import graph;

size(200,150,IgnoreAspect);

// Break the axis at 3; restart at 8.
real a=3, b=8;

scale(Broken(a,b),Linear);
```

```


real[] x={1,2,10};
real[] y=x^2;

draw(graph(x,y),red,MarkFill[0]);

xaxis("x",BottomTop,LeftTicks(new real[]{0,1,2,9,10}));
yaxis("y",LeftRight,RightTicks);

label(rotate(90)*Break,(a,currentpicture.userMin.y));
label(rotate(90)*Break,(a,currentpicture.userMax.y));

```



#### 7. Asymptote can draw secondary axes with the routines

```

picture secondaryX(picture primary=currentpicture, void f(picture));
picture secondaryY(picture primary=currentpicture, void f(picture));
In this example, secondaryY is used to draw a secondary linear y axis against a primary
logarithmic y axis:
import graph;
tex preamble("\def\Arg{\mathop {\rm Arg}\nolimits}");

size(10cm,5cm,IgnoreAspect);

real ampl(real x) {return 2.5/(1+x^2);}
real phas(real x) {return -atan(x)/pi;}

scale(Log,Log);
draw(graph(ampl,0.01,10));
ylimits(0.001,100);

xaxis("$\omega\tau_0$",BottomTop,LeftTicks);
yaxis("$|G(\omega\tau_0)|$",Left,RightTicks);


picture q=secondaryY(new void(picture pic) {
 scale(pic,Log,Linear);

```

```

draw(pic,graph(pic,phas,0.01,10),red);
ylimits(pic,-1.0,1.5);
yaxis(pic,"$\Arg G/\pi$",Right,red,
 LeftTicks("% #.1f$",
 begin=false,end=false));
yequals(pic,1,Dotted);
});
label(q,"(1,0)",Scale(q,(1,0)),red);
add(q);

```



A secondary logarithmic  $y$  axis can be drawn like this:

```

import graph;

size(9cm,6cm,IgnoreAspect);
string data="secondaryaxis.csv";

file in=line(csv(input(data)));

string[] titleref=in;
string[] columnlabel=in;

real[][] a=dimension(in,0,0);
a=transpose(a);
real[] t=a[0], susceptible=a[1], infectious=a[2], dead=a[3], larvae=a[4];
real[] susceptibleM=a[5], exposed=a[6],infectiousM=a[7];

draw(graph(t,susceptible,t >= 10 && t <= 15));
draw(graph(t,dead,t >= 10 && t <= 15),dashed);

xaxis("Time (τ)",BottomTop,LeftTicks);
yaxis(Left,RightTicks);

picture secondary=secondaryY(new void(picture pic) {
 scale(pic,Linear,Log);

```




```

 draw(pic,graph(pic,t,infectious,t >= 10 && t <= 15),red);
 yaxis(pic,Right,red,LeftTicks(begin=false,end=false));
});

add(secondary);
label(shift(5mm*N)*"Proportion of crows",point(NW),E);

```



8. Here is a histogram example, which uses the `stats` module.

```

import graph;
import stats;

size(400,200,IgnoreAspect);


int n=10000;
real[] a=new real[n];
for(int i=0; i < n; ++i) a[i]=Gaussrand();

histogram(a,min(a),max(a),n=100,normalize=true,low=0);

draw(graph(Gaussian,min(a),max(a)),red);

xaxis("x",BottomTop,LeftTicks);
yaxis("dP/dx",LeftRight,RightTicks);

```



9. Here is an example of reading column data in from a file and a least-squares fit, using the `stats` module.

```
size(400,200,IgnoreAspect);

import graph;
import stats;

file fin=line(input("leastquares.dat"));

real[] [] a=dimension(fin,0,0);
a=transpose(a);

real[] t=a[0], rho=a[1];

// Read in parameters from the keyboard:
//real first=getreal("first");
//real step=getreal("step");
//real last=getreal("last");

real first=100;
real step=50;
real last=700;

// Remove negative or zero values of rho:
t=rho > 0 ? t : null;
rho=rho > 0 ? rho : null;

scale(Log,Linear);
```

```

int n=step > 0 ? ceil((last-first)/step) : 0;

real[] T,xi,dxi;

for(int i=0; i <= n; ++i) {
 real first=first+i*step;
 real[] logrho=(t >= first && t <= last) ? log(rho) : null;
 real[] logt=(t >= first && t <= last) ? -log(t) : null;

 if(logt.length < 2) break;

 // Fit to the line logt=L.m*logrho+L.b:
 linefit L=leastquares(logt,logrho);


 T.push(first);
 xi.push(L.m);
 dxi.push(L.dm);
}

draw(graph(T,xi),blue);
errorbars(T,xi,dxi,red);

ylimits(0);

xaxis("T",BottomTop,LeftTicks);
yaxis("ξ",LeftRight,RightTicks);

```



10. Here is an example that illustrates the general `axis` routine.


```
import graph;
```

```

size(0,100);

guide g=ellipse((0,0),1,2);
axis(Label("C",align=10W),g,LeftTicks(endlabel=false,8,end=false),
 ticklocate(0,360,new real(real v) {
 path h=(0,0)--max(abs(max(g)),abs(min(g)))*dir(v);
 return intersect(g,h).x;});));

```



11. To draw a vector field along a path, first define a routine that returns a path as a function of a relative position parameter from  $[0,1]$  and use

```

typedef path vector(real);

void vectorfield(picture pic=currentpicture, path g, int n,
 vector vector, real arrowsize=0, real arrowlength=0,
 pen p=currentpen);

```

Here is a simple example of a flow field:

```

import graph;
defaultpen(1.0);

size(0,150,IgnoreAspect);

real arrowsize=4mm;
real arrowlength=2arrowsize;

// Return a vector interpolated linearly between a and b.
vector vector(pair a, pair b) {
 return new path(real x) {
 return (0,0)--arrowlength*interp(a,b,x);
 };
}

real alpha=1;
real f(real x) {return alpha/x;}

real epsilon=0.5;
path p=graph(f,epsilon,1/epsilon);


```

```



int n=2;
draw(p);
xaxis("x");
yaxis("y");

vectorfield(p,n,vector(W,W),arrowsize);
vectorfield((0,0)--(currentpicture.userMax.x,0),n,vector(NE,NW),
 arrowsize);
vectorfield((0,0)--(0,currentpicture.userMax.y),n,vector(NE,NE),
 arrowsize);

```



12. The following scientific graphs, which illustrate many features of **Asymptote's** graphics routines, were generated from the examples `diatom.asy` and `westnile.asy`, using the comma-separated data in `diatom.csv` and `westnile.csv`.



## 7.20 palette

`Asymptote` can also generate color density images and palettes. The following palettes are predefined in `palette.asy`:

```

pen[] Grayscale(int NColors=256)
 a grayscale palette;

pen[] Rainbow(int NColors=65501)
 a rainbow spectrum;

pen[] BWRainbow(int NColors=65485)
 a rainbow spectrum tapering off to black/white at the ends;

pen[] BWRainbow2(int NColors=65485)
 a double rainbow palette tapering off to black/white at the ends, with a linearly
 scaled intensity.

```

The function `cmk(pen[] Palette)` may be used to convert any of these palettes to the CMYK colorspace.

A color density plot using palette `palette` can be generated from a function  $f(x,y)$  and added to a picture `pic`:

```

bounds image(picture pic=currentpicture, real f(real,real),
 range range=Full, pair initial, pair final,
 int nx=ngraph, int ny=nx, pen[] palette);

```

The function `f` will be sampled at `nx` and `ny` evenly spaced points over a rectangle defined by the points `initial` and `final`, respecting the current graphical scaling of `pic`. The color space is scaled according to the  $z$  axis scaling (see [\[automatic scaling\]](#), page 86). A bounds structure for the function values is returned:

```

struct bounds {
 public real min;
 public real max;
 // Possible tick intervals:
 public int[] divisor;
}

```

This information can be used for generating an optional palette bar. The palette color space corresponds to a range of values specified by the argument `range`, which can be `Full`, `Automatic`, or an explicit range `Range(real min, real max)`. Here `Full` specifies a range varying from the minimum to maximum values of the function over the sampling interval, while `Automatic` selects "nice" limits. The example `imagecontour.asy` illustrates how level sets (contour lines) can be drawn on a color density plot (see [Section 7.27 \[contour\]](#), page 115).

A color density plot can also be generated from an explicit `real[][]` array `data`:

```

bounds image(picture pic=currentpicture, real[][] f, range range=Full,
 pair initial, pair final, pen[] palette,
 bool transpose=(initial.x < final.x && initial.y < final.y));

```

If the initial point is to the left and below the final point, by default the array indices are interpreted according to the Cartesian convention (first index:  $x$ , second index:  $y$ ) rather than the usual matrix convention (first index:  $-y$ , second index:  $x$ ).

To construct an image from an array of irregularly spaced points and an array of values `f` at these points, use one of the routines

```

bounds image(picture pic=currentpicture, pair[] z, real[] f,
 range range=Full, pen[] palette)
bounds image(picture pic=currentpicture, real[] x, real[] y, real[] f,
 range range=Full, pen[] palette)

```

An optionally labelled palette bar may be generated with the routine

```

void palette(picture pic=currentpicture, Label L="", bounds range,
 pair initial, pair final, axis axis=Right, pen[] palette,
 pen p=currentpen, paletteticks ticks=PaletteTicks);

```

The color space of `palette` is taken to be over bounds `range` with scaling given by the `z` scaling of `pic`. The palette orientation is specified by `axis`, which may be one of `Right`, `Left`, `Top`, or `Bottom`. The bar is drawn over the rectangle from `initial` to `final`. The argument `paletteticks` is a special tick type (see [\[ticks\]](#), page 76) that takes the following arguments:

```

paletteticks PaletteTicks(Label format="", ticklabel ticklabel=null,
 bool beginlabel=true, bool endlabel=true,
 int N=0, int n=0, real Step=0, real step=0,
 pen pTick=nullpen, pen ptick=nullpen);

```

The image and palette bar can be fit to a frame and added and optionally aligned to a picture at the desired location:

```

size(12cm,12cm);

import graph;
import palette;

int n=256;
real ninv=2pi/n;
real[][] v=new real[n][n];

for(int i=0; i < n; ++i)
 for(int j=0; j < n; ++j)
 v[i][j]=sin(i*ninv)*cos(j*ninv);


pen[] Palette=BWRainbow();

picture bar;

bounds range=image(v,(0,0),(1,1),Palette);
palette(bar,"A",range,(0,0),(0.5cm,8cm),Right,Palette,
 PaletteTicks("$%+#.1f$"));
add(bar.fit(),point(E),30E);

```





Here is an example that uses logarithmic scaling of the function values:

```
import graph;
import palette;

size(10cm,10cm,IgnoreAspect);

real f(real x, real y) {
 return 0.9*pow10(2*sin(x/5+2*y^0.25)) + 0.1*(1+cos(10*log(y)));
}


scale(Linear,Log,Log);

pen[] Palette=BWRainbow();

bounds range=image(f,Automatic,(0,1),(100,100),nx=200,Palette);

xaxis("x",BottomTop,LeftTicks,Above);
yaxis("y",LeftRight,RightTicks,Above);

palette("$f(x,y)$",range,(0,200),(100,250),Top,Palette,
 PaletteTicks(ptick=linewidth(0.5*linewidth())));
```



One can also draw an image directly from a two-dimensional pen array:

```
void image(picture pic=currentpicture, pen[] [] data,
 pair initial, pair final,
 bool transpose=(initial.x < final.x && initial.y < final.y));
```

as illustrated in the following example:

```
size(200);
```

```
import palette;
```

```
int n=256;
```

```
real ninv=2pi/n;
```

```
pen[] [] v=new pen[n][n];
```


```
for(int i=0; i < n; ++i)
```

```
 for(int j=0; j < n; ++j)
```

```
 v[i][j]=rgb(0.5*(1+sin(i*ninv)),0.5*(1+cos(j*ninv)),0);
```

```
picture bar;
```

```
image(v,(0,0),(1,1));
```




## 7.21 three

This module fully extends the notion of guides and paths in **Asymptote** to three dimensions, introducing the new types **guide3** and **path3**, along with a three-dimensional cycle specifier **cycle3**, tension operator **tension3**, and curl operator **curl3**. Just as in two dimensions, the nodes within a **guide3** can be qualified with these operators and also with explicit directions and control points (using braces and **controls**, respectively). This generalization of John Hobby's spline algorithm is shape-invariant under three-dimensional rotation, scaling, and shifting, and reduces in the planar case to the two-dimensional algorithm used in **Asymptote**, **MetaPost**, and **MetaFont**.

For example, a unit circle in the  $XY$  plane may be filled and drawn like this:


```
import three;
size(100,0);
guide3 g=(1,0,0)..(0,1,0)..(-1,0,0)..(0,-1,0)..cycle3;
filldraw(g,lightgrey);
draw(0--Z,red+dashed,BeginBar,Arrow);
draw(((-1,-1,0)--(1,-1,0)--(1,1,0)--(-1,1,0)--cycle3));
dot(g,red);
```



and then distorted into a saddle:

```
import three;
size(100,0);
guide3 g=(1,0,0)..(0,1,1)..(-1,0,0)..(0,-1,1)..cycle3;
```

```
filldraw(g,lightgrey);
dot(g,red);
draw((-1,-1,0)--(1,-1,0)--(1,1,0)--(-1,1,0)--cycle3));
```



Here  $\mathbf{0}$  is the triple  $(0,0,0)$  and  $\mathbf{X}$ ,  $\mathbf{Y}$ , and  $\mathbf{Z}$  are the triples  $(1,0,0)$ ,  $(0,1,0)$ , and  $(0,0,1)$ , respectively. A general circle can be drawn perpendicular to the direction `normal` with the routine

```
path3 circle(triple c, real r, triple normal=Z);
```

A circular arc centered at  $\mathbf{c}$  with radius  $r$  from  $\mathbf{c}+r*\text{dir}(\text{theta1},\text{phi1})$  to  $\mathbf{c}+r*\text{dir}(\text{theta2},\text{phi2})$ , drawing counterclockwise relative to the normal vector  $\text{cross}(\text{dir}(\text{theta1},\text{phi1}),\text{dir}(\text{theta2},\text{phi2}))$  if  $\text{theta2} > \text{theta1}$  or if  $\text{theta2} == \text{theta1}$  and  $\text{phi2} \geq \text{phi1}$ , can be constructed with

```
path3 arc(triple c, real r, real theta1, real phi1, real theta2, real phi2,
 triple normal=0);
```

The normal must be explicitly specified if  $\mathbf{c}$  and the endpoints are colinear. If  $r < 0$ , the complementary arc of radius  $|r|$  is constructed. For convenience, an arc centered at  $\mathbf{c}$  from triple  $\mathbf{v1}$  to  $\mathbf{v2}$  (assuming  $|\mathbf{v2}-\mathbf{c}|=|\mathbf{v1}-\mathbf{c}|$ ) in the direction CCW (counter-clockwise) or CW (clockwise) may also be constructed with

```
path3 arc(triple c, triple v1, triple v2, triple normal=0,
 bool direction=CCW);
```

When high accuracy is needed, the routines `Circle` and `Arc` defined in `graph3` may be used instead. See [\[surface\]](#), page 111 for an example of a three-dimensional circular arc.

A representation of the plane passing through point  $\mathbf{O}$  with normal  $\text{cross}(\mathbf{u},\mathbf{v})$  is given by

```
path3 plane(triple u, triple v, triple O=0);
```

A three-dimensional box with opposite vertices at triples  $\mathbf{v1}$  and  $\mathbf{v2}$  may be drawn with the function

```
guide3[] box(triple v1, triple v2);
```

For example, a unit cube is predefined as

```
guide3[] unitcube=box((0,0,0),(1,1,1));
```

These projections to two dimensions are predefined:

```
oblique
```

```
oblique(real angle)
```

The point  $(x,y,z)$  is projected to  $(x-0.5z,y-0.5z)$ . If an optional real argument is given, the negative  $z$  axis is drawn at this angle in degrees. The projection `obliqueZ` is a synonym for `oblique`.

```
obliqueX
```

`obliqueX(real angle)`

The point  $(x,y,z)$  is projected to  $(y-0.5x, z-0.5x)$ . If an optional real argument is given, the negative  $x$  axis is drawn at this angle in degrees.

`obliqueY`

`obliqueY(real angle)`

The point  $(x,y,z)$  is projected to  $(x+0.5y, z+0.5y)$ . If an optional real argument is given, the positive  $y$  axis is drawn at this angle in degrees.

`orthographic(triple camera, triple up=Z)`

`orthographic(real x, real y, real z, triple up=Z)`

This projects three dimensions onto two using the view seen at the location `camera` or  $(x,y,z)$ , respectively, orienting the camera so that, if possible, the vector `up` points upwards. Parallel lines are projected to parallel lines.

`perspective(triple camera, triple up=Z)`

`perspective(real x, real y, real z, triple up=Z)`

These project three dimensions onto two taking account of perspective, as seen from the location `camera` or  $(x,y,z)$ , respectively, orienting the camera so that, if possible, the vector `up` points upwards.

The default projection, `currentprojection`, is initially set to `perspective(5,4,2)`.

It is occasionally useful to be able to invert a projection, sending a pair `z` onto the plane perpendicular to `normal` and passing through `point`:

`triple invert(pair z, triple normal, triple point,  
                  projection P=currentprojection);`

Three-dimensional objects may be transformed with one of the following built-in `transform3` types:

`shift(triple v)`

translates by the triple `v`;

`xscale3(real x)`

scales by `x` in the  $x$  direction;

`yscale3(real y)`

scales by `y` in the  $y$  direction;

`zscale3(real z)`

scales by `z` in the  $z$  direction;

`scale3(real s)`

scales by `s` in the  $x$ ,  $y$ , and  $z$  directions;

`rotate(real angle, triple v)`

rotates by `angle` in degrees about an axis `v` through the origin;

`rotate(real angle, triple u, triple v)`

rotates by `angle` in degrees about the axis `u--v`;

`reflect(triple u, triple v, triple w)`

reflects about the plane through `u`, `v`, and `w`.

Three-dimensional versions of the path functions `length`, `size`, `point`, `dir`, `precontrol`, `postcontrol`, `arclength`, `arctime`, `reverse`, `subpath`, `intersect`, `intersectionpoint`, `min`, `max`, `cyclic`, and `straight` are also defined in the module `three`.

Planar hidden surface removal is implemented with a binary space partition and picture clipping. A planar path is first converted to a struct `face` derived from `picture`. A `face` may be given to a drawing routine in place of any `picture` argument. An array of such faces may then be drawn, removing hidden surfaces:

```
void add(picture pic=currentpicture, face[] faces,
 projection P=currentprojection);
```

Here is an example showing three orthogonal intersecting planes:

```
size(6cm,0);
import math;
import three;


real u=2.5;
real v=1;

currentprojection=oblique;

path3 y=plane((2u,0,0),(0,2v,0),(-u,-v,0));
path3 l=rotate(90,Z)*rotate(90,Y)*y;
path3 g=rotate(90,X)*rotate(90,Y)*y;

face[] faces;
filldraw(faces.push(y),y,yellow);
filldraw(faces.push(l),l,lightgrey);
filldraw(faces.push(g),g,green);

add(faces);
```



Here is an example showing all five 3D path connectors:

```
import graph3;

size(0,175);

currentprojection=orthographic(500,-500,500);

triple[] z=new triple[10];

z[0]=(0,100,0); z[1]=(50,0,0); z[2]=(180,0,0);

for(int n=3; n <= 9; ++n)
 z[n]=z[n-3]+(200,0,0);


path3 p=z[0]..z[1]---z[2]::{Y}z[3]
 &z[3]..z[4]--z[5]::{Y}z[6]
 &z[6]::z[7]---z[8]..{Y}z[9];

draw(p,greyscale(0.5)+linewidth(4mm));

bbox3 b=limits(0,(700,200,100));

xaxis(Label("x",1),b,red,Arrow);
yaxis(Label("y",1),b,red,Arrow);
zaxis(Label("z",1),b,red,Arrow);

dot(z);
```



A three-dimensional bounding box structure is returned by calling `bbox3(triple min, triple max)` with the opposite corners `min` and `max`. This can be used to adjust the aspect ratio (see the example `helix.asy`):

```
void aspect(picture pic=currentpicture, bbox3 b,
 real x=0, real y=0, real z=0);
```

Further three-dimensional examples are provided in the files `near_earth.asy`, `conicurv.asy`, and (in the `animations` subdirectory) `cube.asy`.

## 7.22 light

This module provides a simple implementation of three-dimensional lighting effects. An illustration is provided in the solid geometry package `solids.asy`.

## 7.23 graph3

This module implements three-dimensional versions of the functions in `graph.asy`. They work much like their two-dimensional counterparts, except that the user has to keep track of the three-dimensional axes limits (which in two dimension are stored in the picture) in a `bbox3` bounding box. The function

```
bbox3 autolimits(picture pic=currentpicture, triple min, triple max);
```

can be used to determine “nice” values for the bounding box corners. A user-space bounding box that takes into account of the axes scalings for picture `pic` is returned by

```
bbox3 limits(picture pic=currentpicture, triple min, triple max);
```

To crop a bounding box to a given interval use:

```
void xlimits(bbox3 b, real min, real max);
void ylimits(bbox3 b, real min, real max);
void zlimits(bbox3 b, real min, real max);
void limits(bbox3 b, triple min, triple max);
```

To draw an  $x$  axis in three dimensions from `triple min` to `triple max` with ticks in the direction `dir`, use the routine

```
void xaxis(picture pic=currentpicture, Label L="", triple min, triple max,
 pen p=currentpen, ticks ticks=NoTicks, triple dir=Y,
 arrowbar arrow=None, bool put=Above,
 projection P=currentprojection, bool opposite=false);
```

To draw an  $x$  axis in three dimensions from `triple min` to `triple (max,min.y,min.z)` with ticks in the direction `dir`, use the routine

```
void xaxis(picture pic=currentpicture, Label L="", triple min, real max,
 pen p=currentpen, ticks ticks=NoTicks, triple dir=Y,
 arrowbar arrow=None, bool put=Above,
 projection P=currentprojection, bool opposite=false);
```

To draw an  $x$  axis in three dimensions using `bbox3 b` with ticks in the direction `dir`, use the routine

```
void xaxis(picture pic=currentpicture, Label L="", bool all=false,
 bbox3 b, pen p=currentpen, ticks ticks=NoTicks,
 triple dir=Y, arrowbar arrow=None, bool put=Above,
 projection P=currentprojection);
```

If `all=true`, also draw opposing edges of the three-dimensional bounding box. Analogous routines `yaxis` and `zaxis` can be used to draw  $y$  and  $z$  axes in three dimensions. There are also some convenient routines for drawing all three axis:



```

void axes(Label xlabel="x", Label ylabel="y", Label zlabel="z",
 triple min, triple max, pen p=currentpen, arrowbar arrow=None,
 bool put=Below, projection P=currentprojection);

void axes(Label xlabel="x", Label ylabel="y", Label zlabel="z",
 bbox3 b, pen p=currentpen, arrowbar arrow=None,
 bool put=Below, projection P=currentprojection);

```

Here is an example of a helix and bounding box axes with rotated tick and axis labels, using orthographic projection:

```

import graph3;

size(0,200);

currentprojection=orthographic(4,6,3);

real x(real t) {return cos(2pi*t);}
real y(real t) {return sin(2pi*t);}
real z(real t) {return t;}


defaultpen(overwrite(SuppressQuiet));

path3 p=graph(x,y,z,0,2.7,Spline);
bbox3 b=autolimits(min(p),max(p));
aspect(b,1,1,1);

xaxis(rotate(X)*"x",all=true,b,red,RightTicks(rotate(X)*Label,2,2));
yaxis(rotate(Y)*"y",all=true,b,red,RightTicks(rotate(Y)*Label,2,2));
zaxis("z",all=true,b,red,RightTicks);

draw(p,Arrow);

```



The next example illustrates three-dimensional  $x$ ,  $y$ , and  $z$  axes, with autoscaling of the upper  $z$  limit disabled:

```
import graph3;

size(0,200,IgnoreAspect);


currentprojection=perspective(5,2,2);

defaultpen(overwrite(SuppressQuiet));

scale(Linear,Linear,Log(automax=false));

bbox3 b=autolimits(Z,X+Y+30Z);

xaxis("x",b,red,RightTicks(2,2));
yaxis("y",b,red,RightTicks(2,2));
zaxis("z",b,red,RightTicks);
```




One can also place ticks along a general three-dimensional axis:

```
import graph3;

size(0,100);

path3 G=xscale3(1)*(yscale3(2)*unitcircle3);

axis(Label("C",align=Relative(5E)),G,
 LeftTicks(endlabel=false,8,end=false),
 ticklocate(0,360,new real(real v) {
 path g=G;
 path h=0--max(abs(max(G)),abs(min(G)))*dir(90,v);
 return intersect(g,h).x;
 },perpendicular(G,Z)));
```



Surface plots of matrices and functions over the region `box(a,b)` in the  $XY$  plane are also implemented:

```
picture surface(real[][] f, pair a, pair b,
```

```

 pen surfacepen=lightgray, pen meshpen=nullpen,
 light light=currentlight, projection P=currentprojection);
picture surface(triple[][] f, pen surfacepen=lightgray, pen meshpen=nullpen,
 light light=currentlight, projection P=currentprojection);
picture surface(real f(pair z), int nsub=1, pair a, pair b, int nx=nmesh,
 int ny=nx, pen surfacepen=lightgray, pen meshpen=nullpen,
 light light=currentlight, projection P=currentprojection);

```

A parametric surface for a function  $f(u,v)$  over the parameter space  $\text{box}(a,b)$  may be drawn with

```

picture surface(triple f(pair z), int nsub=1, pair a, pair b, int nu=nmesh,
 int nv=nu, pen surfacepen=lightgray, pen meshpen=nullpen,
 light light=currentlight, projection P=currentprojection);

```

as illustrated in the example `parametricsurface.asy`. The optional argument `nsub` allows one to subsample the function `nsub` times along the cell edges. However, smoother Gouraud shading is used for lighting effects if `nsub=1`. Lighting can be disabled by setting `light=0`, as in this example of a Gaussian surface subsampled 4 times:

```

import graph3;

size(200,0);

currentprojection=perspective(5,4,2);

real f(pair z) {return 0.5+exp(-abs(z)^2);}

draw((-1,-1,0)--(1,-1,0)--(1,1,0)--(-1,1,0)--cycle3);

draw(arc(0.12Z,0.2,90,60,90,15),ArcArrow);

picture surface=surface(f,nsub=4,(-1,-1),(1,1),nx=10,light=0);


bbox3 b=limits(0,1.75(1,1,1));

xaxis(Label("x",1),b,red,Arrow);
yaxis(Label("y",1),b,red,Arrow);
zaxis(Label("z",1),b,red,Arrow);

label("0", (0,0,0),S,red);

add(surface);

```



Surface lighting is illustrated in the example files `parametricsurface.asy` and `sinc.asy`. A mesh can be drawn without solid fills by specifying `nullpen` for the `surfacepen`. A mesh is always drawn in the absence of lighting, unless `meshpen=invisible`.


## 7.24 solids

This solid geometry package defines a struct `revolution` that can be used to fill and draw surfaces of revolution. The following example uses it to display the outline of a circular cylinder of radius 1 and height 1.5 aligned in the direction  $(0,1,1)$  with perspective projection:

```
import solids;

size(0,100);

revolution r=cylinder(0,1,1.5,Y+Z);
r.draw(heavygreen);
```



Further illustrations are provided in the example files `cylinder.asy`, `sphere.asy`, `cones.asy`, `hyperboloid.asy`, and `torus.asy`.

## 7.25 featpost3D

To facilitate the conversion of existing MetaPost programs, this module contains a partial port of the MetaPost 3D package `featpost3D` of L. Nobre G., C. Barbarosie, and J. Schwaiger to `Asymptote`. However, much (but not all) of the functionality of this port is now obsoleted by the package `three`, which fully extends the notion of a path to three dimensions, together with the packages `solids` and `graph3`. The original `featpost3D` package is documented at

<http://matagalatlante.org/nobre/featpost/doc/featpost.html>

## 7.26 flowchart

This package provides routines to assist in drawing flowcharts. The primary `struct` is a `block`, which represents a single block on the flowchart.

`pair block.bound`

The upper right corner of the block relative to its lower left corner.

`pair block.llcorner`

The absolute lower left corner of the block.

`pair block.center()`

`void block.center(pair loc)`

Returns the current absolute center of the block, or sets the absolute center of the block.

`pair block.position(real x)`

Returns an absolute position along the border path of the block.

`pair block.top()`

`pair block.left()`

`pair block.right()`

`pair block.bottom()`

`pair block.topleft()`

`pair block.topright()`

`pair block.bottomleft()`

`pair block.bottomright()`

These eight methods return an absolute position on the appropriate edge of the block.

`picture block.draw(pen p=currentpen)` Returns a frame representing the block.

The following block generation routines accept a `Label`, `string`, or `frame` for their object argument.

```
block rectangle(object header, object body,
 pen headercolor=mediumgray,
 pen bodycolor=currentpen,
 pair center=(0,0), real dx=3)
```

```
block rectangle(object body, pen headercolor=mediumgray,
 pen bodycolor=currentpen,
 pair center=(0,0), real dx=3)
```

Create a rectangular block with the given header (if specified) and body. Here `dx` is the amount of padding around the header and body.

```
block diamond(object body, pair center=(0,0), real ds=5,
 real dw=1, real height=20, real dh=0)
```

Create a diamond-shaped chart block.

```
block circle(object body, pair center=(0,0), real dr=3)
```

Create a circular chart block.

```
block roundrectangle(object body, pair center=(0,0),
 real ds=5, real dw=0)
```

Create a rectangular chart block with rounded corners.

```
block bevel(object body, pair center=(0,0), real dh=5, real dw=5)
```

Create a rectangular chart block with beveled edges.

```
path path(pair point[] ... bool horizontal[])
```

Draw paths joining the pairs in `point` with right-angled lines. The entries in `horizontal` identify whether successive segments between the pairs specified by `point` should be drawn horizontal (`true`) or vertical (`false`).

Here is a simple flowchart example:

```
import flowchart;


block block1=rectangle("Start",(-50,300));
block block1=rectangle("Example",pack("Start:", "", "$A:=0$", "$B:=1$"),
 (-50,300));

block block2=diamond("Choice?",(0,200));
block block3=roundrectangle("Do something",(-100,100));
block block4=bevel("Don't do something",(100,100));
block block5=circle("End",(0,0));

draw(block1);
draw(block2);
draw(block3);
draw(block4);
draw(block5);

draw(path(new pair[]{block1.right(),block2.top()},Horizontal),
 Arrow, PenMargin);
draw(Label("Yes",0.5),path(new pair[]{block2.left(),block3.top()},
 Horizontal),Arrow, PenMargin);
draw(Label("Yes",0.5),path(new pair[]{block2.left(),block3.top()},
 Horizontal),Arrow, PenMargin);
draw(Label("No",0.5,N),path(new pair[]{block2.right(),block4.top()},
 Horizontal),Arrow, PenMargin);
draw(path(new pair[]{block3.bottom(),block5.left()},Vertical),
 Arrow, PenMargin);
draw(path(new pair[]{block4.bottom(),block5.right()},Vertical),
```

```
Arrow, PenMargin);
```



## 7.27 contour

This package draws contour lines. To construct contours for a function `f`, use

```
guide[] [] contour(real f(real, real), pair a, pair b,
 real[] c, int nx=ngraph, int ny=nx,
 interpolate join=operator --);
```

The contour lines `c` for the function `f` are drawn on the rectangle defined by the bottom-left and top-right points `a` and `b`. The integers `nx` and `ny` define the resolution. The default resolution, `ngraph x ngraph` (here `ngraph` defaults to 100), can be increased for greater accuracy. The default interpolation operator is `operator --` (linear). Spline interpolation (`operator ..`) generally produces more accurate pictures, but as usual, can overshoot in certain cases.

To construct contours for an array of data values on a uniform two-dimensional lattice, use

```
guide[] [] contour(real[] [] f, real[] [] midpoint=new real[] [],
 pair a, pair b, real[] c,
 interpolate join=operator --);
```



To construct contours for an array of irregularly spaced points and an array of values **f** at these points, use one of the routines

```
guide[] [] contour(pair[] z, real[] f, real[] c,
 interpolate join=operator --);
guide[] [] contour(real[] x, real[] y, real[] f, real[] c,
 interpolate join=operator --);
```

The contours themselves can be drawn with one of the routines


```
void draw(picture pic=currentpicture, Label[] L=new Label[],
 guide[] [] g, pen p=currentpen)

void draw(picture pic=currentpicture, Label[] L=new Label[],
 guide[] [] g, pen[] p)
```

The following simple example draws the contour at value 1 for the function  $z = x^2 + y^2$ , which is a unit circle:

```
import contour;
size(75);

real f(real a, real b) {return a^2+b^2;}
draw(contour(f,(-1,-1),(1,1),new real[] {1}));
```



The next example draws and labels multiple contours for the function  $z = x^2 - y^2$  with the resolution 100 x 100, using a dashed pen for negative contours and a solid pen for positive (and zero) contours:

```
import contour;
import stats;
size(200);

real f(real x, real y) {return x^2-y^2;}
int n=10;
real[] c = new real[n];
for(int i=0; i < n; ++i) c[i]=(i-n/2)/n;

pen[] p=sequence(new pen(int i) {
 return (c[i] >= 0 ? solid : dashed)+fontsize(6);
},n);


Label[] Labels=sequence(new Label(int i) {
```

```

 return Label(c[i] != 0 ? (string) c[i] : "", Relative(unitrand()), (0,0),
 UnFill(1bp));
},n);

draw(Labels, contour(f, (-1,-1), (1,1), c), p);

```



The next example illustrates how contour lines can be drawn on color density images:

```

import graph;
import palette;
import contour;

size(10cm,10cm,IgnoreAspect);

pair a=(0,0);
pair b=(2pi,2pi);

real f(real x, real y) {return cos(x)*sin(y);}

int N=200;
int Divs=10;
int divs=2;

defaultpen(1bp);
pen Tickpen=black;
pen tickpen=gray+0.5*linewidth(currentpen);
pen[] Palette=BWRainbow();

scale(false);

bounds range=image(f, Automatic, a, b, N, Palette);

```

```


xaxis("x",BottomTop,LeftTicks,Above);
yaxis("y",LeftRight,RightTicks,Above);

// Major contours
real[] Cvals;
Cvals=sequence(11)/10 * (range.max-range.min) + range.min;
draw(contour(f,a,b,Cvals,N,operator ..),Tickpen);

// Minor contours
real[] cvals;
real[] sumarr=sequence(1,divs-1)/divs * (range.max-range.min)/Divs;
for (int ival=0; ival < Cvals.length-1; ++ival)
 cvals.append(Cvals[ival]+sumarr);
draw(contour(f,a,b,cvals,N,operator ..),tickpen);

palette("$f(x,y)$",range,point(NW)+(0,0.5),point(NE)+(0,1),Top,Palette,
 PaletteTicks(N=Divs,n=divs,Tickpen,tickpen));

```



Finally, here is an example that illustrates the construction of contours from irregularly spaced data:

```
import contour;
```

```

size(200);

int n=100;
pair[] points=new pair[n];
real[] values=new real[n];

real r() {return 1.1*(rand()/randMax*2-1);}


for(int i=0; i < n; ++i)
 points[i]=(r(),r());

real f(real a, real b) {return a^2+b^2;}

for(int i=0; i < n; ++i)
 values[i]=f(points[i].x,points[i].y);

draw(contour(points,values,new real[]{0.25,0.5,1},operator ..),blue);

```



In the above example, the contours of irregularly spaced data are constructed by first creating a triangular mesh from an array `z` of pairs, using Gilles Dumoulin's C++ port of Paul Bourke's triangulation code:

```

int[][] triangulate(pair[] z);
size(200);
int np=100;
pair[] points;

real r() {return 1.2*(rand()/randMax*2-1);}

for(int i=0; i < np; ++i)

```

```


 points.push((r(),r()));

int[] [] trn=triangulate(points);

for(int i=0; i < trn.length; ++i) {
 draw((points[trn[i][0]]--(points[trn[i][1]])));
 draw((points[trn[i][1]]--(points[trn[i][2]])));
 draw((points[trn[i][2]]--(points[trn[i][0]])));
}

for(int i=0; i < np; ++i)
 dot(points[i],red);

```



The example `Gouraudcontour` illustrates how to produce color density images over such irregular triangular meshes.

## 7.28 slopefield

To draw a slope field of the function  $y' = f(x, y)$  ( $y' = f(x)$  is also accepted), use:

```

picture slopefield(real f(real,real), pair a, pair b,
 int nx=nmesh, int ny=nx,
 real tickfactor=0.5, pen p=currentpen);

```

Here, the points `a` and `b` are the lower left and upper right corners of the rectangle in which the slope field is to be drawn, `nx` and `ny` are the respective number of ticks in the  $x$  and  $y$  directions, `tickfactor` is the fraction of the minimum cell dimension to use for drawing ticks, and `p` is the pen to use for drawing the slope fields. The return value is a picture that can be added to `currentpicture` via the `add(picture)` command.

The function

```

path curve(pair c, real f(real,real), pair a, pair b);

```

takes a point (`c`) and a slope field-defining function `f` (`real f(real)` is also accepted), and returns, as a path, the curve passing through that point. The points `a` and `b` represent the rectangular boundaries over which the curve is interpolated.

## 8 Options

Type `asy -h` to see the full list of command line options supported by Asymptote:

Usage: `asy [options] [file ...]`

Options:

|                                   |                                                         |
|-----------------------------------|---------------------------------------------------------|
| <code>-V,-View</code>             | View output; command-line only                          |
| <code>-a,-align C B T Z</code>    | Center, Bottom, Top, or Zero page alignment [Center]    |
| <code>-autoplain</code>           | Enable automatic importing of plain [true]              |
| <code>-autorotate</code>          | Enable automatic PDF page rotation [false]              |
| <code>-batchMask</code>           | Mask fpu exceptions in batch mode [false]               |
| <code>-batchView</code>           | View output in batch mode [false]                       |
| <code>-bw</code>                  | Convert all colors to black and white [false]           |
| <code>-c,-clearGUI</code>         | Clear GUI operations [false]                            |
| <code>-cmyk</code>                | Convert rgb colors to cmyk [false]                      |
| <code>-d,-debug</code>            | Enable debugging messages [false]                       |
| <code>-x,-deconstruct X</code>    | Deconstruct into transparent GIF objects magnified by X |
| <code>-gray</code>                | Convert all colors to grayscale [false]                 |
| <code>-h,-help</code>             | Show summary of options; command-line only              |
| <code>-historylines n</code>      | Retain n lines of history [1000]                        |
| <code>-i,-ignoreGUI</code>        | Ignore GUI operations [false]                           |
| <code>-inlinetex</code>           | Generate inline tex code [false]                        |
| <code>-interactiveMask</code>     | Mask fpu exceptions in interactive mode [true]          |
| <code>-interactiveView</code>     | View output in interactive mode [true]                  |
| <code>-k,-keep</code>             | Keep intermediate files [false]                         |
| <code>-l,-listvariables</code>    | List available global functions and variables [false]   |
| <code>-localhistory</code>        | Use a local interactive history file [false]            |
| <code>-m,-mask</code>             | Mask fpu exceptions; command-line only                  |
| <code>-multipleView</code>        | View output from multiple batch-mode files [false]      |
| <code>-O,-offset pair</code>      | PostScript offset [(0,0)]                               |
| <code>-f,-outformat format</code> | Convert each output file to specified format [eps]      |
| <code>-o,-outname name</code>     | Alternative output name for first file                  |
| <code>-p,-parseonly</code>        | Parse file [false]                                      |
| <code>-prompt string</code>       | Prompt [ "> " ]                                         |
| <code>-q,-quiet</code>            | Suppress welcome message [false]                        |
| <code>-rgb</code>                 | Convert cmyk colors to rgb [false]                      |
| <code>-safe</code>                | Disable system call; command-line only [true]           |
| <code>-scroll n</code>            | Scroll standard output n lines at a time [0]            |
| <code>-tabcompletion</code>       | Interactive prompt auto-completion [true]               |
| <code>-tex engine</code>          | TeX engine ("latex pdflatex tex pdftex none") ["latex"] |
| <code>-s,-translate</code>        | Show translated virtual machine code [false]            |
| <code>-twice</code>               | Run LaTeX twice (to resolve references) [false]         |
| <code>-unsafe</code>              | Enable system call; command-line only [false]           |
| <code>-u,-user x</code>           | General purpose user string [""]                        |
| <code>-v,-verbose</code>          | Increase verbosity level [0]                            |

All boolean options, except for `-unsafe`, `-safe`, and `help`, can be negated by prepending `no` to the option name.

If no arguments are given, `Asymptote` runs in interactive mode (see [Chapter 9 \[Interactive mode\]](#), page 125). In this case, the default output file is `out.eps`.

If `-` is given as the file argument, `Asymptote` reads from standard input.

If multiple files are specified, they are treated as separate `Asymptote` runs.

Default option values may be entered as `Asymptote` code in a configuration file named `config.asy` (or the file specified by the environment variable `ASYMPTOTE_CONFIG` or `-config` option). `Asymptote` will look for this file in its usual search path. Typically the configuration file is placed in the `.asy` directory in the user's home directory (`%USERPROFILE%\`.`asy` under MSDOS). Configuration variables are accessed using the long form of the option names:

```
import settings;
outformat="pdf";
batchView=false;
interactiveView=true;
batchMask=false;
interactiveMask=true;
```

Command-line options override these defaults. Most configuration variables may also be changed at runtime.

The default output format is EPS for the (default) `latex` tex engine and PDF for the `pdflatex` tex engine. An alternative output format (without antialiasing) may be produced by using the `-f format` option. This supports any format supported by the `ImageMagick convert` program (version 6.2.4 or later recommended; an `Invalid Parameter` error message indicates that the MSDOS utility `convert` is being used instead of the one that comes with `ImageMagick`). To give specific options to the `convert` program, call `convert` manually. This example enables antialiasing and produces a `tiff` format at double the usual size:

```
asy -o - venn | convert -density 288x288 -geometry 100%x eps:- venn.tiff
```

If the option `-unsafe` is given, `Asymptote` runs in unsafe mode. This enables the `int system(string s)` call, allowing one to execute arbitrary shell commands. The default mode, `-safe`, disables this call.

A PostScript offset may be specified as a pair (in bp units) with the `-O` option:

```
asy -O 0,0 file
```

The default offset is zero. The default value of the page alignment setting `align` is `Center`.

The `-u` option may be used to specify arbitrary `Asymptote` code on the command line as a string. It is not necessary to terminate the string with a semicolon. Multiple `-u` options are executed in the order they are given. Command-line code like `-u x=sqrt(2)` can be executed within a module like this:

```
real x;
access settings;
usersetting();
write(x);
```

When the `-l` option is used with file arguments, it only lists global functions and variables defined in the specified file(s).



Additional debugging output is produced with each additional `-v` option:

- `-v`            Display top-level module and final output file names.
- `-vv`          Also display imported and included module names and final `LaTeX` and `dvips` processing information.
- `-vvv`        Also output `LaTeX` bidirectional pipe diagnostics.
- `-vvvv`       Also output knot guide solver diagnostics.
- `-vvvvv`      Also output `Asymptote` traceback diagnostics.

## 9 Interactive mode

Interactive mode is entered by executing the command `asy` with no file arguments. Each line must be a complete `Asymptote` statement; however, it is not necessary to terminate each line with a semicolon.

The following special commands are supported only in interactive mode and must be entered immediately after the prompt:

|                         |                                                                                                                                                                                                                                                                                                                                                                                                                                                                                             |
|-------------------------|---------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| <code>help</code>       | view the manual                                                                                                                                                                                                                                                                                                                                                                                                                                                                             |
| <code>reset</code>      | reset <code>Asymptote</code> to its initial state, except for changes to the settings module (see <a href="#">[settings]</a> , page 123).                                                                                                                                                                                                                                                                                                                                                   |
| <code>input FILE</code> | resets the environment (except for breakpoints; see <a href="#">Chapter 13 [Debugger]</a> , page 129) and does an <code>erase()</code> ; <code>include FILE</code> . If the file name <code>FILE</code> contains nonalphanumeric characters, enclose it with quotation marks. For convenience, a trailing semi-colon followed by optional <code>Asymptote</code> commands may be entered on the same line.                                                                                  |
| <code>quit</code>       | exit interactive mode (abbreviated as <code>q</code> ; <code>exit</code> is a synonym). A history of the most recent 1000 (can be changed with the <code>historylines</code> command-line option) previous commands will be retained in the file <code>.asy/history</code> in the user's home directory (unless the command line option <code>-localhistory</code> was specified, in which case the history will be stored in the file <code>.asy_history</code> in the current directory). |

Typing `ctrl-C` interrupts the execution of `Asymptote` code and returns control to the interactive prompt.

Interactive mode is implemented with the GNU `readline` library, with command history and auto-completion. To customize the key bindings, see: <http://cnswww.cns.cwru.edu/php/chet/readline/readline.html>

The file `asymptote.py` in the `Asymptote` system directory provides an alternative way of entering `Asymptote` commands interactively, coupled with the full power of `Python`. Copy this file to your `Python` path and then execute from within `Python` the commands

```
from asymptote import *
g=asy()
g.size(200)
g.draw("unitcircle")
g.send("draw(unitsquare)")
g.fill("unitsquare, blue")
g.clip("unitcircle")
g.label("\ "0\"", (0,0), SW)
```

## 10 Graphical User Interface

In the event that adjustments to the final figure are required, the Graphical User Interface (GUI) `xasy` included with `Asymptote` allows you to move graphical objects around with mouse `Button-1`.

To use `xasy`, one must first deconstruct `Asymptote` pictures into transparent GIF images with the command `asy -xN`, where `N` denotes the magnification (a positive real number, say 2). The command `asy -VxN` automatically invokes `xasy` once deconstruction is complete. Alternatively, one may turn on the `-xN` option in interactive mode or from within a module using the function `gui()` or `gui(N)`. One can turn GUI mode off again with `gui(0)`. Deconstruction requires that `ImageMagick` (see [\[convert\]](#), [page 123](#)) be installed.

The modified layout can be written to disk with the `w` key in a form readable to `Asymptote`. A wheel mouse is convenient for raising and lowering objects, to expose the object to be moved. If a wheel mouse is not available, mouse `Button-2` (lower) can be used repeatedly instead. Here are the currently defined key mappings:

|                             |        |
|-----------------------------|--------|
| <code>z</code>              | undo   |
| <code>r</code>              | redo   |
| <code>&lt;Delete&gt;</code> | delete |
| <code>w</code>              | write  |
| <code>q</code>              | quit   |

One can also draw connected line segments by holding down the shift key and pressing mouse `Button-1` at each desired node. Releasing the shift key ends the definition of the path. More features will be added to this preliminary GUI soon.

As `xasy` is written in the interactive scripting language `Python/Tk`, it requires that both `Python` and the `tkinter` package be installed (included with `Python` under `MSDOS`). `Fedora Core 4` (and later) users can either install `tkinter` with the command

```
yum install tkinter
```

or manually install the individual packages:

```
rpm -i tkinter-2.4.1-2.i386.rpm
rpm -U --nodeps tix-8.1.4-100.i386.rpm
rpm -U --nodeps tk-8.4.9-3.i386.rpm
```

Deconstruction of compound objects (such as arrows) can be prevented by enclosing them within the commands

```
void begingroup(picture pic=currentpicture);
void endgroup(picture pic=currentpicture);
```

By default, the elements of a picture or frame will be grouped together on adding them to a picture. However, the elements of a frame added to another frame are not grouped together by default: their elements will be individually deconstructed (see [\[add\]](#), [page 41](#)).

## 11 PostScript to Asymptote

The excellent PostScript editor `pstoedit` (version 3.44 or later; available from <http://pstoedit.net>) includes an Asymptote backend. Unlike virtually all other `pstoedit` backends, this driver includes native clipping, even-odd fill rule, PostScript subpath, and full image support.

For example, try:

```
asy -V /usr/share/doc/asymptote/examples/venn.asy
pstoedit -f asy venn.eps test.asy
asy -V test
```

If the line widths aren't quite correct, try giving `pstoedit` the `-dis` option. If the fonts aren't typeset correctly, try giving `pstoedit` the `-dt` option.

## 12 Help

A list of frequently asked questions (FAQ) is maintained at

<http://asymptote.sourceforge.net/FAQ>

Questions on installing and using `Asymptote` that are not addressed in the FAQ should be sent to the `Asymptote` forum:

[http://sourceforge.net/forum/forum.php?forum\\_id=409349](http://sourceforge.net/forum/forum.php?forum_id=409349)

Including an example that illustrates what you are trying to do will help you get useful feedback. `LaTeX` problems can often be diagnosed with the `-vv` or `-vvv` command-line options. Contributions in the form of patches or `Asymptote` modules can be posted here:

[http://sourceforge.net/tracker/?atid=685685&group\\_id=120000](http://sourceforge.net/tracker/?atid=685685&group_id=120000)

To receive announcements of upcoming releases, please subscribe to `Asymptote` at

<http://freshmeat.net/subscribe/50750>

If you find a bug in `Asymptote`, please check (if possible) whether the bug is still present in the latest Subversion developmental code (see [Section 2.8 \[Subversion\]](#), page 6) before submitting a bug report. New bugs can be submitted using the Bug Tracking System at

<http://sourceforge.net/projects/asymptote>

To see if the bug has already been fixed, check bugs with Status `Closed` and recent lines in

<http://asymptote.sourceforge.net/ChangeLog>

`Asymptote` can be configured with the optional GNU library `libsigsegv`, available from <http://libsigsegv.sourceforge.net>, which allows one to distinguish user-generated `Asymptote` stack overflows (see [\[stack overflow\]](#), page 51) from true segmentation faults (due to internal C++ programming errors; please submit the `Asymptote` code that generates such segmentation faults along with your bug report).

## 13 Debugger

Asymptote now includes a line-based (as opposed to code-based) debugger that can assist the user in following flow control. To set a break point in file `file` at line `line`, use the command

```
void stop(string file, int line, code s=quote{ });
```

The optional argument `s` may be used to conditionally set the variable `ignore` in `plain_debugger.asy` to `true`. For example, the first 10 instances of this breakpoint will be ignored (the variable `int count=0` is defined in `plain_debugger.asy`):

```
stop("test",2,quote{ignore=(++count <= 10);});
```

To set a break point in file `file` at the first line containing the string `text`, use

```
void stop(string file, string text, code s=quote{ });
```

To list all breakpoints, use:

```
void breakpoints();
```

To clear a breakpoint, use:

```
void clear(string file, int line);
```

To clear all breakpoints, use:

```
void clear();
```

The following commands may be entered at the debugging prompt:

|                |                                                                |
|----------------|----------------------------------------------------------------|
| <code>h</code> | help;                                                          |
| <code>c</code> | continue execution;                                            |
| <code>i</code> | step to the next instruction;                                  |
| <code>s</code> | step to the next executable line;                              |
| <code>n</code> | step to the next executable line in the current file;          |
| <code>f</code> | step to the next file;                                         |
| <code>r</code> | return to the file associated with the most recent breakpoint; |
| <code>t</code> | toggle tracing ( <code>-vvvvv</code> ) mode;                   |
| <code>q</code> | quit debugging and end execution;                              |
| <code>x</code> | exit the debugger and run to completion.                       |

Arbitrary `Asymptote` code may also be entered at the debugging prompt; however, since the debugger is implemented with `eval`, currently only top-level (global) variables can be displayed or modified.

The debugging prompt may be entered manually with the call

```
void breakpoint(code s=quote{ });
```

## 14 Acknowledgments

Financial support for the development of **Asymptote** was generously provided by the Natural Sciences and Engineering Research Council of Canada, the Pacific Institute for Mathematical Sciences, and the University of Alberta Faculty of Science.

We also would like to acknowledge the previous work of John D. Hobby, author of the program **MetaPost** that inspired the development of **Asymptote**, and Donald E. Knuth, author of **T<sub>E</sub>X** and **MetaFont** (on which **MetaPost** is based).

The authors of **Asymptote** are Andy Hammerlindl, John Bowman, and Tom Prince. Sean Healy designed the **Asymptote** logo. Other contributors include Radoslav Marinov, Chris Savage, Jacques Pienaar, Mark Henning, Steve Melenchuk, Martin Wiebusch, and Stefan Knorr.

# Index

|                       |        |  |
|-----------------------|--------|--|
| <b>!</b>              |        |  |
| ! .....               | 48     |  |
| != .....              | 47     |  |
| <b>%</b>              |        |  |
| % .....               | 47     |  |
| %= .....              | 48     |  |
| <b>&amp;</b>          |        |  |
| & .....               | 10     |  |
| && .....              | 47     |  |
| <b>*</b>              |        |  |
| * .....               | 30, 47 |  |
| ** .....              | 47     |  |
| *= .....              | 48     |  |
| <b>+</b>              |        |  |
| + .....               | 30, 47 |  |
| ++ .....              | 48     |  |
| += .....              | 48     |  |
| <b>-</b>              |        |  |
| - .....               | 47     |  |
| -- .....              | 9, 48  |  |
| --- .....             | 10     |  |
| -= .....              | 48     |  |
| -l .....              | 123    |  |
| -u .....              | 123    |  |
| -v .....              | 3, 7   |  |
| <b>.</b>              |        |  |
| .....                 | 9      |  |
| .asy .....            | 4      |  |
| <b>/</b>              |        |  |
| / .....               | 47     |  |
| /= .....              | 48     |  |
| <b>:</b>              |        |  |
| : .....               | 48     |  |
| :: .....              | 10     |  |
| <b>&lt;</b>           |        |  |
| < .....               | 47     |  |
| <b>&lt;=</b> .....    | 47     |  |
| <b>=</b>              |        |  |
| == .....              | 47     |  |
| <b>&gt;</b>           |        |  |
| > .....               | 47     |  |
| >= .....              | 47     |  |
| <b>?</b>              |        |  |
| ? .....               | 48     |  |
| <b>^</b>              |        |  |
| ^ .....               | 47, 48 |  |
| ^= .....              | 48     |  |
| ^^ .....              | 10     |  |
| <b> </b>              |        |  |
| .....                 | 48     |  |
| <b>2</b>              |        |  |
| 2D graphs .....       | 74     |  |
| <b>3</b>              |        |  |
| 3D graphs .....       | 107    |  |
| <b>A</b>              |        |  |
| a4 .....              | 3      |  |
| abort .....           | 25     |  |
| abs .....             | 21, 54 |  |
| access .....          | 61     |  |
| acknowledgments ..... | 130    |  |
| acos .....            | 54     |  |
| aCos .....            | 54     |  |
| acosh .....           | 54     |  |
| add .....             | 41     |  |
| Ai .....              | 54     |  |
| Ai_deriv .....        | 54     |  |
| Airy .....            | 54     |  |
| alias .....           | 57     |  |
| align .....           | 123    |  |
| Allow .....           | 36     |  |
| and .....             | 10     |  |
| angle .....           | 21     |  |
| animate .....         | 4      |  |
| animation .....       | 20     |  |



annotate..... 74  
 antialiasing..... 123  
 append..... 44, 55  
 ArcArrow..... 13  
 ArcArrows..... 13  
 arclength..... 28, 105  
 arctime..... 28, 105  
 arguments..... 51  
 arithmetic operators..... 47  
 arrays..... 55  
 arrow..... 13, 17  
 Arrow..... 13  
 arrow keys..... 7  
 Arrows..... 13  
 as..... 62  
 asin..... 54  
 aSin..... 54  
 asinh..... 54  
 aspect..... 106  
 Aspect..... 38  
 assignment..... 20  
 asy..... 63  
 asy-mode..... 5  
 asy.vim..... 5  
 asymptote.sty..... 65  
 ASYMPTOTE\_CONFIG..... 123  
 atan..... 54  
 aTan..... 54  
 atan2..... 54  
 atanh..... 54  
 atleast..... 10  
 attach..... 81  
 autolimits..... 107  
 automatic scaling..... 86  
 axial shading..... 15  
 axis..... 92, 94, 109, 110  
 azimuth..... 22

## B

babel..... 73  
 Bar..... 13  
 Bars..... 13  
 base modules..... 70  
 basealign..... 33  
 baseline..... 17  
 batch mode..... 7  
 bbox3..... 106, 107  
 BeginArcArrow..... 13  
 BeginArrow..... 13  
 BeginBar..... 13  
 BeginDotMargin..... 13  
 BeginMargin..... 13  
 BeginPenMargin..... 13  
 BeginPoint..... 17  
 Bessel..... 54  
 bevel..... 114  
 beveljoin..... 32

Bi..... 54  
 Bi\_deriv..... 54  
 binary operators..... 47  
 Blank..... 13  
 block.bottom..... 113  
 block.bottomleft..... 113  
 block.bottomright..... 113  
 block.bound..... 113  
 block.center..... 113  
 block.draw..... 113  
 block.left..... 113  
 block.llcorner..... 113  
 block.position..... 113  
 block.right..... 113  
 block.top..... 113  
 block.topleft..... 113  
 block.topright..... 113  
 bool..... 20  
 boolean operators..... 47  
 Bottom..... 76  
 BottomTop..... 76  
 box..... 39  
 bp..... 7  
 break..... 20  
 breakpoints..... 129  
 brick..... 34  
 broken axis..... 89  
 bug reports..... 128  
 buildcycle..... 29  
 Button-1..... 126  
 Button-2..... 126  
 BWRainbow..... 98  
 BWRainbow2..... 98

## C

C string..... 23  
 CAD..... 74  
 casts..... 60  
 cbrt..... 54  
 cd..... 43  
 ceil..... 54  
 Center..... 17  
 checker..... 34  
 Chinese..... 73  
 Ci..... 54  
 circle..... 26, 114  
 CJK..... 73  
 clear..... 44, 129  
 clip..... 16  
 cm..... 8  
 cmyk..... 30  
 colatitude..... 22  
 color..... 30  
 colorless..... 30  
 colors..... 31  
 comma-separated-value..... 59  
 command-line options..... 4, 122

comment character . . . . . 43  
 Compiling from UNIX source . . . . . 5  
 complement . . . . . 56  
 concat . . . . . 57  
 conditional . . . . . 20, 48  
 config . . . . . 123  
 configuration file . . . . . 3  
 configuring . . . . . 3  
 conj . . . . . 21  
 constructors . . . . . 46  
 continue . . . . . 20, 129  
 contour . . . . . 115  
 controls . . . . . 10, 102  
 convert . . . . . 4, 20, 123  
 copy . . . . . 56, 57  
 cos . . . . . 54  
 Cos . . . . . 54  
 cosh . . . . . 54  
 cputime . . . . . 59  
 crop . . . . . 85  
 cropping graphs . . . . . 85  
 cross . . . . . 22, 84  
 crosshatch . . . . . 35  
 csv . . . . . 59  
 cubicroots . . . . . 59  
 curl . . . . . 10  
 curl3 . . . . . 102  
 currentpen . . . . . 30  
 currentprojection . . . . . 104  
 custom axis types . . . . . 76  
 custom tick locations . . . . . 77  
 cycle . . . . . 7, 9  
 cycle3 . . . . . 102  
 cyclic . . . . . 29, 55, 105  
 cyclicflag . . . . . 55  
 Cyrillic . . . . . 73

## D

dashdotted . . . . . 32  
 dashed . . . . . 32  
 data types . . . . . 20  
 date . . . . . 24  
 Debian . . . . . 2  
 debugger . . . . . 129  
 declaration . . . . . 20  
 default arguments . . . . . 51  
 defaultpen . . . . . 31, 32, 33, 36, 37  
 degrees . . . . . 21, 54  
 Degrees . . . . . 54  
 delete . . . . . 55  
 description . . . . . 1  
 diamond . . . . . 113  
 dimension . . . . . 59  
 dir . . . . . 4, 21, 22, 27, 105  
 direction specifier . . . . . 9  
 directory . . . . . 43  
 dirttime . . . . . 28

display . . . . . 4  
 do . . . . . 20  
 dot . . . . . 14, 21, 22  
 DotMargin . . . . . 13  
 DotMargins . . . . . 13  
 dotted . . . . . 32  
 double . . . . . 44  
 draw . . . . . 13, 14  
 Draw . . . . . 13, 40  
 drawing commands . . . . . 13  
 drawline . . . . . 70  
 drawtree . . . . . 71  
 dvips . . . . . 4

## E

E . . . . . 54  
 Editing modes . . . . . 5  
 Ei . . . . . 54  
 elliptic functions . . . . . 54  
 else . . . . . 20  
 emacs . . . . . 5  
 embed . . . . . 73  
 EndArcArrow . . . . . 13  
 EndArrow . . . . . 13  
 EndBar . . . . . 13  
 EndDotMargin . . . . . 13  
 endl . . . . . 44  
 EndMargin . . . . . 13  
 EndPenMargin . . . . . 13  
 EndPoint . . . . . 17  
 environment variables . . . . . 4  
 eof . . . . . 44, 59  
 eol . . . . . 44, 59  
 EPS . . . . . 17  
 erase . . . . . 23, 42  
 erf . . . . . 54  
 erfc . . . . . 54  
 error . . . . . 43, 44  
 errorbars . . . . . 84  
 eval . . . . . 63  
 evenodd . . . . . 11  
 evenodd . . . . . 33  
 exit . . . . . 125, 129  
 exp . . . . . 54  
 expi . . . . . 21, 22  
 explicit . . . . . 60  
 explicit casts . . . . . 60  
 exponential integral . . . . . 54  
 extendcap . . . . . 32  
 extension . . . . . 70

## F

F . . . . . 54  
 fabs . . . . . 54  
 face . . . . . 105  
 featpost3D . . . . . 113

Fedora Core ..... 2  
 feynman ..... 71  
 fft ..... 58  
 FFTW ..... 5  
 file ..... 43, 129  
 fill ..... 15  
 Fill ..... 13, 40  
 filldraw ..... 15  
 FillDraw ..... 13, 39  
 filloutside ..... 15  
 fillrule ..... 33  
 find ..... 23, 56  
 firstcut ..... 29  
 fit ..... 39  
 fixedscaling ..... 38  
 floor ..... 54  
 flowchart ..... 113  
 flush ..... 44  
 fmod ..... 54  
 font ..... 33  
 font command ..... 33  
 fontcommand ..... 33  
 fontsize ..... 33  
 for ..... 20  
 format ..... 24, 123  
 forum ..... 128  
 frame ..... 38  
 from ..... 61  
 function declarations ..... 51  
 functions ..... 49, 54

## G

gamma ..... 54  
 Gaussrand ..... 54  
 geometry ..... 70  
 getc ..... 43  
 getreal ..... 44  
 getstring ..... 44  
 gifmerge ..... 20  
 GNU Scientific Library ..... 54  
 Gouraud shading ..... 15  
 gradient shading ..... 15  
 graph ..... 74  
 graph3 ..... 107  
 graphic ..... 17  
 graphical user interface ..... 126  
 gray ..... 30  
 grayscale ..... 30  
 Grayscale ..... 98  
 grid ..... 34, 87  
 gs ..... 3  
 gsl ..... 54  
 GSL ..... 5  
 GUI ..... 126  
 guide ..... 25  
 guide3 ..... 102

## H

hatch ..... 35  
 help ..... 125  
 help ..... 128  
 help ..... 129  
 hidden surface removal ..... 105  
 histogram ..... 54  
 hypot ..... 54

## I

I ..... 54  
 i\_scaled ..... 54  
 identity ..... 37, 54  
 if ..... 20  
 IgnoreAspect ..... 38  
 image ..... 98  
 ImageMagick ..... 4, 20, 123  
 implicit casts ..... 60  
 implicit linear solver ..... 70  
 implicit scaling ..... 49  
 import ..... 62  
 inches ..... 8  
 including images ..... 17  
 inheritance ..... 46  
 input ..... 43, 125  
 insert ..... 23, 55  
 inside ..... 29, 105  
 inst ..... 129  
 installation ..... 2  
 int ..... 20  
 integer division ..... 47  
 interactive mode ..... 125  
 international characters ..... 73  
 intersect ..... 28, 70, 105  
 intersection ..... 70  
 intersectionpoint ..... 28  
 intersectpoint ..... 105  
 intMax ..... 20  
 inverse ..... 37, 58  
 invert ..... 104  
 invisible ..... 30

## J

J ..... 54  
 Japanese ..... 73

## K

K ..... 54  
 k\_scaled ..... 54  
 keywords ..... 51  
 Korean ..... 73

## L

label ..... 16

Label ..... 14  
 Label ..... 79  
 labelx ..... 79  
 labely ..... 79  
 Landscape ..... 39  
 lastcut ..... 29  
 LaTeX fonts ..... 33  
 LaTeX usage ..... 65  
 latin1 ..... 73  
 latitude ..... 22  
 lattice shading ..... 15  
 layer ..... 13  
 leastsquares ..... 71, 93  
 Left ..... 78  
 LeftRight ..... 78  
 LeftSide ..... 17  
 LeftTicks ..... 76, 77  
 legend ..... 13  
 legend ..... 82  
 Legendre ..... 54  
 length ..... 21, 22, 23, 27, 55, 105  
 letter ..... 3  
 libm routines ..... 54  
 libsigsegv ..... 51, 128  
 light ..... 107  
 limits ..... 85, 107  
 line ..... 59  
 line mode ..... 59  
 Linear ..... 86  
 linecap ..... 32  
 linejoin ..... 32  
 lineskip ..... 33  
 linewidth ..... 32  
 log ..... 54  
 Log ..... 86  
 log-log graph ..... 86  
 log10 ..... 54  
 log2 graph ..... 89  
 logarithmic graph ..... 86  
 logical operators ..... 47  
 longdashdotted ..... 32  
 longdashed ..... 32  
 longitude ..... 22  
 loop ..... 20

## M

MacOS X binary distributions ..... 2  
 makepen ..... 36  
 map ..... 57  
 Margin ..... 13  
 Margins ..... 13  
 marker ..... 84  
 marknodes ..... 84  
 markuniform ..... 84  
 math ..... 70  
 mathematical functions ..... 54  
 max ..... 29, 57, 58, 105

maxbound ..... 22, 23  
 merge ..... 20  
 MetaPost ..... 73  
 MetaPost ... ..... 10  
 MetaPost cutafter ..... 29  
 MetaPost cutbefore ..... 29  
 MetaPost pickup ..... 30  
 MetaPost whatever ..... 70  
 Microsoft Windows ..... 3  
 MidArrow ..... 13  
 MidPoint ..... 17  
 min ..... 29, 57, 105  
 minbound ..... 21, 23  
 minipage ..... 17  
 miterjoin ..... 32  
 mm ..... 8  
 mouse ..... 126  
 Move ..... 36  
 MoveQuiet ..... 37

## N

named arguments ..... 51  
 new ..... 45, 56  
 newframe ..... 38  
 next ..... 129  
 NFSS ..... 33  
 nobasealign ..... 33  
 NoFill ..... 13, 40  
 NoMargin ..... 13  
 none ..... 44  
 None ..... 13  
 NoTicks ..... 76  
 nullpen ..... 16, 40

## O

oblique ..... 103  
 obliqueX ..... 104  
 obliqueY ..... 104  
 obliqueZ ..... 103  
 offset ..... 123  
 opacity ..... 34  
 open ..... 43  
 operator ..... 48  
 operator -- ..... 74  
 operator ..... 74  
 operator cast ..... 60  
 operator ecast ..... 61  
 operator init ..... 45  
 operators ..... 47  
 options ..... 122  
 orientation ..... 39  
 orthographic ..... 104  
 output ..... 44, 123  
 overloading functions ..... 50  
 overwrite ..... 36

**P**

**P** ..... 54  
 pack ..... 17  
 packing ..... 53  
 pair ..... 7, 21  
 paperheight ..... 3  
 papertype ..... 3  
 paperwidth ..... 3  
 parametrized curve ..... 85  
 path ..... 26, 114  
 path[] ..... 10  
 patterns ..... 34  
 patterns ..... 71  
 pdfviewer ..... 3  
 pen ..... 30  
 PenMargin ..... 13  
 PenMargins ..... 13  
 perpendicular ..... 70  
 perspective ..... 104  
 picture ..... 38  
 picture alignment ..... 41  
 Pl ..... 54  
 plain ..... 70  
 point ..... 27, 105  
 polar ..... 22  
 polargraph ..... 75  
 polygon ..... 84  
 pop ..... 55  
 Portrait ..... 39  
 postcontrol ..... 28, 105  
 postfix operators ..... 48  
 postscript ..... 43  
 PostScript fonts ..... 33  
 PostScript subpath ..... 10  
 pow10 ..... 54  
 precision ..... 44  
 precontrol ..... 28, 105  
 prefix operators ..... 48  
 private ..... 45  
 programming ..... 20  
 psviewer ..... 3  
 pt ..... 8  
 public ..... 45  
 push ..... 55  
 python ..... 3  
 Python usage ..... 125

**Q**

quadraticroots ..... 59  
 quit ..... 125, 129  
 quotient ..... 47

**R**

radial shading ..... 15  
 RadialShade ..... 40  
 radians ..... 54

Rainbow ..... 98  
 rand ..... 54  
 randMax ..... 54  
 read1 ..... 60  
 read2 ..... 60  
 read3 ..... 60  
 reading ..... 43  
 readline ..... 45  
 real ..... 20  
 realDigits ..... 20  
 realEpsilon ..... 20  
 realMax ..... 20  
 realMin ..... 20  
 realmult ..... 21  
 rectangle ..... 113  
 recursion ..... 51  
 reflect ..... 37  
 Relative ..... 17  
 remainder ..... 54  
 replace ..... 24  
 resetdefaultpen ..... 37  
 rest arguments ..... 52  
 restore ..... 43  
 restricted ..... 45  
 return ..... 129  
 reverse ..... 24, 28, 56, 105  
 rewind ..... 44  
 rfind ..... 23  
 rgb ..... 30  
 Riemann zeta function ..... 54  
 Right ..... 78  
 RightSide ..... 17  
 RightTicks ..... 76, 77  
 Rotate ..... 16  
 round ..... 54  
 roundcap ..... 32  
 roundedpath ..... 71  
 roundjoin ..... 32  
 roundrectangle ..... 114  
 RPM ..... 2  
 runtime imports ..... 62  
 Russian ..... 73

**S**

save ..... 43  
 scale ..... 37, 86  
 Scale ..... 16, 86  
 scale3 ..... 104  
 scientific graph ..... 80  
 scroll ..... 44  
 search ..... 56  
 search paths ..... 4  
 Seascape ..... 39  
 secondary axis ..... 90  
 secondaryX ..... 90  
 secondaryY ..... 90  
 seek ..... 44

segmentation fault ..... 128  
 self operators ..... 48  
 sequence ..... 56  
 settings ..... 3, 123  
 sgn ..... 54  
 shading ..... 15  
 shift ..... 37, 104  
 Shift ..... 16  
 shiftless ..... 37  
 shipout ..... 39  
 Si ..... 54  
 simplex ..... 70  
 sin ..... 54  
 Sin ..... 54  
 single ..... 44  
 sinh ..... 54  
 size ..... 27, 38, 105, 123  
 slant ..... 37  
 Slant ..... 16  
 slice ..... 29  
 slide ..... 71  
 slopefield ..... 120  
 solid ..... 32  
 solids ..... 112  
 solve ..... 58  
 sort ..... 57  
 Spline ..... 74  
 sqrt ..... 54  
 squarecap ..... 32  
 srand ..... 54  
 stack overflow ..... 51, 128  
 static ..... 63  
 stats ..... 71  
 stdin ..... 44  
 stdout ..... 44  
 step ..... 129  
 stop ..... 129  
 straight ..... 29, 70, 105  
 Straight ..... 74  
 strftime ..... 24  
 string ..... 23  
 strptime ..... 24  
 struct ..... 45  
 structures ..... 45  
 subpath ..... 28, 105  
 subpictures ..... 39  
 substr ..... 23  
 Subversion ..... 6  
 sum ..... 57  
 superpath ..... 10  
 Suppress ..... 36  
 SuppressQuiet ..... 36  
 SVN ..... 6

## T

tab ..... 44  
 tab completion ..... 7

tan ..... 54  
 Tan ..... 54  
 tanh ..... 54  
 tar.gz ..... 2  
 tell ..... 44  
 tension ..... 10  
 tension3 ..... 102  
 tex ..... 43  
 TeX fonts ..... 33  
 T<sub>E</sub>X string ..... 23  
 texpath ..... 4  
 texpreamble ..... 43  
 texreset ..... 43  
 textbook graph ..... 79  
 this ..... 45  
 three ..... 102  
 tick ..... 79  
 ticks ..... 76  
 Ticks ..... 76, 77  
 tile ..... 34  
 tilings ..... 34  
 time ..... 24  
 Top ..... 76  
 trace ..... 129  
 transform ..... 37, 104  
 transparency ..... 34  
 transpose ..... 57  
 tree ..... 71  
 triangle ..... 70  
 tridiagonal ..... 58  
 trigonometric integrals ..... 54  
 triple ..... 22  
 TrueMargin ..... 13  
 tutorial ..... 7  
 type1cm ..... 33  
 typedef ..... 25, 50

## U

unfill ..... 16  
 UnFill ..... 40  
 unicode ..... 73  
 Uninstall ..... 6  
 unit ..... 21, 22  
 unitcircle ..... 9  
 unitsize ..... 8  
 UNIX binary distributions ..... 2  
 unpacking ..... 53  
 unravel ..... 61  
 UpsideDown ..... 39  
 usepackage ..... 43  
 user coordinates ..... 8  
 user-defined operators ..... 48

## V

vector field ..... 95  
 verbatim ..... 43

vim..... 5  
 virtual functions..... 46  
 void..... 20

## W

wheel mouse ..... 126  
 while..... 20  
 write..... 44, 60

## X

xasy ..... 4, 126  
 xaxis..... 107  
 xequals..... 78  
 XEquals..... 78  
 xinput..... 44  
 xlimits ..... 85, 107  
 xoutput..... 44  
 xpart ..... 21, 22  
 xscale..... 37  
 xscale3..... 104  
 xtick..... 79  
 XZero..... 78

## Y

Y..... 54  
 yaxis..... 107  
 yequals..... 78  
 YEquals..... 76  
 ylimits ..... 85, 107  
 ypart ..... 21, 22  
 yscale..... 37  
 yscale3..... 104  
 ytick..... 79  
 YZero..... 76

## Z

zaxis..... 107  
 zero\_Ai..... 54  
 zero\_Ai\_deriv ..... 54  
 zero\_Bi..... 54  
 zero\_Bi\_deriv ..... 54  
 zero\_J..... 54  
 zerowinding ..... 33  
 zeta..... 54  
 zlimits..... 107  
 zpart..... 22  
 zscale3..... 104